
! 18.  TABLO representation of the MONASH model !

! 18.1 Files !

File
FID # Input-output flows #;
EXTRA # Supplementary updatable data, e.g. capital stocks, rates of return #;
EXTRA3 # Alternative EXTRA file, generated in forecast & used in policy #;
EXTRA4 # Lagged version of EXTRA file #;
EXTRA5 # Lagged version of EXTRA3 file #;
ITER # Iteration number, used in forward-looking expectations #;
PARAMS # Main file of non-updatable data, e.g. substitution elasticities #;
SETINFO # Contents of sets and maps between statisitical categories #;
SRFM # Macro data e.g. balance of payments, government accounts #;
ROREXT # Foward-looking expect. : guess of RORs and params for algorithm #;
(New,Text) WRITFILE # Text output file for examining database values #;

! 18.2 Sets and subsets !

! 18.2a Set declarations listed alphabetically !

Set ABARE # Rural and mining products forecast by ABARE # ! size 15(6) !
  Read Elements from file SETINFO Header "TABA";
Set ABARE_MIN # Mining products forecast by ABARE # ! size 6(3) !
  Read Elements from file SETINFO Header "TMIN";
Set ABARE_RUR # Rural products forecast by ABARE # ! size 10(3) !
  Read Elements from file SETINFO Header "TRUR";
Set ABS_COM # Commodity categories in ABS output data # ! size 110(33) !
  Read Elements from file SETINFO Header "ABSC";
Set AIRTRAN  # Air transport commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "AIRT";
Set CELECSET # Electricity commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "ELEC";
Set COILSET # Oil commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "COIL";
Set COM # Commodities # ! size 115(33) !
  Read Elements from file SETINFO Header "COM ";
Set COMMUNICAT # Communication commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "COMM";
Set COMPCOM # Composite commodities # ! size 6(2) !
  Read Elements from file SETINFO Header "CCOM";
Set COM_JP # Commodities produced by joint-product industries # ! size 9(3) !
  Read Elements from file SETINFO Header "COMJ";
Set COM_UP # Commodities produced by unique-product inds # ! size 106(30) !
  Read Elements from file SETINFO Header "COMU";

Set CONSTRUCTION # Construction commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "CNST";
Set COUNTSET # Used in forming indices in sums # ! size 20(20)! (CS1-CS20);
Set FAC # Primary factors # ! size 3(3) ! (labour,capital,land);
Set IND # Industries # ! size 113(32) !
  Read Elements from file SETINFO Header "IND ";
Set INDSAGGTO20 # Industries aggregated to 20 sectors # ! size 20(20) !
  Read Elements from file SETINFO Header "IN20";
Set INDSAGGTO26 # Industries aggregated to 26 sectors # ! size 26(26) !
  Read Elements from file SETINFO Header "IN26";
Set IND_JP # Joint-production industries # ! size 7(2) !
  Read Elements from file SETINFO Header "INDJ";
Set IND_UP # Unique-product industries # ! size 106(23) !
  Read Elements from file SETINFO Header "INDU";
Set LABCAP # Non-land primary factors # ! size 2(2) ! (labour,capital);
Set LAB_DATA_IND # Inds in employment (hours) data # ! size 92(32) !
  Read Elements from file SETINFO Header "LDAT";
Set MARGCOM # Margin commodities, e.g. retail, road transport # ! size 9(5) !
  Read Elements from file SETINFO Header "MARC";
Set MERCHANDISE # Merchandise commodities #  ! size 85(18) !
  Read Elements from file SETINFO Header "MERC";
Set MOTORIND # MotorVeh Industry # ! size 1(1) !
  Read Elements from file SETINFO Header "MOTI";
Set MOTORVEH # MotorVeh Commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "MOTO";
Set NCOM # Commodities, national accounts consumption classif.# ! size 38(38) !
  Read Elements from File Setinfo Header "NCOM";
Set NONMARGCOM # Non-margin commodities # ! size 106(28) !
  Read Elements from file SETINFO Header "NONM";
Set NTABARE # Meat cattle, Oil and Steel commodities # ! size 3(1) !
  Read Elements from file SETINFO Header "NTAB";
Set NTRADEXP # Non-trad exports: COM less trad, tour & special # ! size 95(24) !
  Read Elements from file SETINFO Header "NONT";
Set OCC # Broad occupations # ! size 8(8) !
  Read Elements from file SETINFO Header "OCCU";
Set PCOSTD # Coms whose dom prices reflect costs: # ! size 104(23) !
  Read Elements from file SETINFO Header "PCST";
Set PWORLDD # Coms whose dom prices reflect world prices # ! size 11(3) !
  Read Elements from file SETINFO Header "PWDD";
Set SOURCE # Domestic/Imported # ! size 2(2) ! (dom,imp);
Set TIME # Sequence of numbers 0,1,2, ...,25 # ! size 26(26) !
  Read Elements from file SETINFO Header "TIME";

Set TRADEXP # Traditional exports, agriculture and mining # ! size 13(4) !
  Read Elements from file SETINFO Header "TRAD";
Set TRANSERV  # Services to transport commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "TSER";
Set TOURISM # Tourism commodities, e.g. air trans, hotels # ! size 4(2) !
  Read Elements from file SETINFO Header "TOUR";
Set WATAIRTRN # Water and air transport industries # ! size 2(2) !
  Read Elements from file SETINFO Header "WATR";
Set WATERTRAN # Water transport commodity # ! size 1(1) !
  Read Elements from file SETINFO Header "WTRA";
Set WOOLGRAIN # Wool and grain commodities # ! size 4(2) !
  Read Elements from file SETINFO Header "WOGR";

! 18.2b Subset declarations listed alphabetically !

Subset
ABARE is subset of COM;  MOTORIND is subset of IND;
ABARE_MIN is subset of ABARE; MOTORVEH is subset of COM;
ABARE_RUR is subset of ABARE;  NONMARGCOM is subset of COM;
AIRTRAN is subset of COM; NTABARE is subset of ABARE;
CELECSET is subset of COM;       NTABARE is subset of NTRADEXP;
COILSET is subset of COM; NTRADEXP is subset of COM;
COMMUNICAT is subset of COM;      PCOSTD is subset of COM;
COM_JP is subset of COM; PWORLDD is subset of COM;
COM_UP is subset of COM;    TOURISM is subset of COM;
CONSTRUCTION is subset of COM; TRADEXP is subset of ABARE;
IND_JP is subset of IND;         TRANSERV is subset of COM;
IND_UP is subset of IND; WATAIRTRN is subset of IND;
LABCAP is subset of FAC;         WATERTRAN is subset of COM;
MARGCOM is subset of COM; WOOLGRAIN is subset of COM;
MERCHANDISE is subset of COM;    

 ! 18.3 Coefficient declarations listed alphabetically !

Coefficient
        (Integer)(All,abs,ABS_COM)
ABSCOMNO(abs) # Gives position of abs in the set ABS_COM R#;
ADJDUMYEAR1 # Adjusts DUM_YEAR1, one in first year, then zero FIu#;
        (All,j,IND)
ADJ_COEFF(j) # Rate of disappearance of disequilibrium in rors R#;
        (All,j,IND)
ADJ_RE(j) # Controls adjustment of expected rors R#;
AE_F_T # $A value of foreign equity held by Austalians, start of year  Fx#;
AE_F_T_1 # $A value of f'gn equity held by Australians, end of yr, Fx#;
AE_F_T_1_B # $A value of f'gn equity held by Australians, end of yr, base FIx#;



148 Dynamic General Equilibrium Modelling for Forecasting and Policy

AE_F_T_B # $A value of f'gn equity held by Australians, start of yr, base FIx#;
AGEBEN # Age pensions RU#;
AGGCAP # Total payments to capital Fi#;
AGGCAPF # Rentals on foreign-owned capital Fx#;
AGGCON # Total purchases by households Fh#;
AGGEXP # Total export earnings Fh#;
AGGIMP # Total foreign currency import costs Fh#;
AGGINV # Aggregate investment, current prices Fh#;
AGGINVENT # Aggregate inventory change, current prices Fh#;
AGGINVG # Value of public sector investment Fv#;
AGGLAB # Total payments to labour Fi#;
AGGLND # Total payments to land Fi#;
AGGNTRADEXP # Total non-traditional export earnings, f.o.b. $A Fh#;
AGGOCT # Total other cost ticket payments Fi#;
AGGOTH # Total value of government consumption Fh#;
AGGTAX # Aggregate indirect tax revenue Fg#;
AGGTAX0 # Aggregate revenue, taxes on production Fg#;
AGGTAX0G # Aggregate genuine revenue, taxes on production Fg#;
AGGTAX1 # Aggregate revenue, indirect taxes on intermediate Fg#;
AGGTAX1G # Genuine agg rev, indirect taxes on intermediate Fg#;
AGGTAX2 # Aggregate revenue, indirect taxes on investment Fg#;
AGGTAX2G # Genuine agg rev, indirect taxes on investment Fg#;
AGGTAX3 # Aggregate revenue, indirect taxes on households Fg#;
AGGTAX3G # Aggregate revenue, genuine indirect taxes on households Fg#;
AGGTAX4 # Aggregate revenue,indirect taxes on flows to export Fg#;
AGGTAX4F # Estimate of agg. ph. taxes on exports used in calc. of CAD FIbU#;
AGGTAX4G # Aggregate genuine revenue,indirect taxes on flows to exports Fg#;
AGGTAX5 # Aggregate revenue,indirect taxes on flows to government Fg#;
AGGTAX5G # Aggregate genuine revenue,indirect taxes on flows to govt. Fg#;
AGGTAXG # Aggregate genuine revenue,indirect taxes on all flows Fg#;
AGGTAXM # Aggregate tariff revenue Ff#;
AGGTOURISM # Foreign currency value of tourism exports Fh#;
AGGTRADEXP # Total traditional export earnings Fh#;
AGGVCAP_T # Aggregate value of industry capital stocks at start of year t Fw#;
ALPHA1 # Controls wage response to gaps between labour demand & supply R#;
ALPHA2 # Slope of long-run labour supply curve R#;
APCGNP # Average propensity to consume out of GNP Fw#;
APSGNP # Average propensity to save out of GNP Fw#;
APSGNP_B # Average propensity to save out of GNP,base RFIw#;

        (All,i,COM)
AVIMPSH(i) # Ave.import share of i across non-margin, non-inventory uses Fp#;
AV_PROP_CON  # Average propensity to consume Fy#;
AV_ROR_SE_B # Average rate of return in year t, base RFIs#;
        (All,i,COM)(All,j,IND)
B0CI(i,j) # Share of industry j in the output of commodity i Fc#;
        (All,i,COM)(All,s,SOURCE)
B3SH(i,s) # Share of (i,s) in household budget RFIuU#;
        (All,i,COM)(All,s,SOURCE)
B3SHB(i,s) # Share of (i,s) in household budget, base FIu#;
        (All,i,COM)(All,s,SOURCE)
B3SHBO(i,s) # Share of (i,s) in household budget in forecast, base FIu#;
        (All,i,COM)(All,s,SOURCE)
B3SHO(i,s) # Share of (i,s) in household budget in forecast RFIuU#;
BAL_ITEM # Current acc. def. less net capital inflow: a statist. discrep. Fw#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
BAS1(i,s,j) # Basic-value flows of commodities: current production RU#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
BAS2(i,s,j) # Basic-value flows of commodities: industry investment RU#;
        (All,i,COM)(All,s,SOURCE)
BAS3(i,s) # Basic-value flows of commodities: private consumption RU#;
        (All,i,COM)
BAS4(i) # Basic-value flows of commodities: exports RU#;
        (All,i,COM)(All,s,SOURCE)
BAS5(i,s) # Basic-value flows of commodities: government demand RU#;
        (All,i,COM)(All,s,SOURCE)
BAS6(i,s) # Basic-value flows of commodities: inventory changes RU#;
BETA_CH # Sens. of rors to cap growth, reset if sim period changes R#;
CADEF # Current account deficit RFIxU#;
        (All,j,IND)
CAPITAL(j) # Returns to capital by industry RU#;
        (All,i,COM_JP)(All,j,IND_JP)(All,cc,COMPCOM)
CCPROD(i,j,cc) # 1 if com. i is part of compcom cc in ind. j, else 0 R#;
        (All,j,IND)
CHKGR1(j) # One if K_GR_MIN(j) >= K_GR(j),else zero Fr#;
        (All,j,IND)
CHKGR2(j) # One if K_GR_MAX(j) <= K_GR(j),else zero Fr#;
COEFF_NYEAR # Zero in the last year of a simulation, else one FIt#;
        (All,j,IND)
COEFF_SL(j) # Coefficient in capital supply curve FIr#;
        (All,t,TIME)
COEFF_TIME(t) # Vector of consecutive numbers 0,1,2,....,NYEARS R#;

        (All,j,IND)
COL_PAYROLL(j) # Collection of payroll tax by industry Fu#;
COL_PAYRTOT # Aggregate collection of payroll taxes Fu#;
        (All,i,COM)
COMMUNICDUM(i) # 1 if and only if com i is communication, else 0 Fa#;
        (Integer)(All,i,COM)
COMTOABSCOM(i) # Gives the element of COM_ABS containing com i R#;
        (All,j,IND)
COSTS(j) # Total cost in each industry Fj#;
        (All,s,COUNTSET)
COUNT(s) # Equals s-1 for s = 1 to TAU, equals 0 for s > TAU Fw#;
        (All,j,IND)
CRESHSUM1(j) # Denominator in CRESH primary-factor shares Fm#;
        (All,j,IND)
CRESHSUM2(j) # Denominator in CRESH occupational-labour shares Fm#;
        (All,j,IND)
C_SPECIMP1(j) # Coef. on j's activity in j's imports of C99 and C100 Fo#;
        (All,j,IND)
C_SPECIMP2(j) # Coef. on j's total imports in j's imports of C99 and C100 Fo#;
        (All,j,IND)
C_SPECIMP3(j) # Coef. on j's exports in j's imports of C99 and C100 Fo#;
        (All,i,COM)
C_TWIST_SRC(i) # Sens. of imp/dom twists to dom output growth rel to GDP R#;
        (All,i,COM)
DELTA(i) # Normalised marginal budget shares RFIkU#;
        (All,i,COM_JP)(All,j,IND_JP)
DENOM(i,j) # Ind j's revenue from comp good to which com i belongs Fc#;
        (All,j,IND)
DEP(j) # Depreciation rates (numbers like 0.07) for physical capital R#;
DIFF # Used in setting maximum feasible capital growth rates R#;
        (All,j,IND)
DISEQRE_B(j) # Diseq. in rational expect. version of ror in base year FIt#;
        (All,j,IND)
DISEQSE_B(j) # Diseq. in static expect. version of the ror in base year FIt#;
DIV_AE_F # $A value of profits on f'gn equity held by Australians RU#;
DIV_FEA_ADD # Preliminary estimate of DIV_FE_A FIx#;
DIV_FE_A # $A value of profits on Australian equity held by f'gners RU#;
        (All,j,IND)
DIV_FE_A_J(j) # $A value of profits on Australian equity held by f'gners FIwFw#;
DUMF # Zero if phantoms are to be removed from data, else one R#;
        (All,j,IND)
DUMMYCAR(j) # One for motor vehicle industry, otherwise zero FIf#;

DUMMY_DEC # Zero in simulations off 1/2-way database, else one R#;
        (All,i,NTRADEXP)
DUMMY_HC(i) # Transformed logistic function of export share in output FIp#;
        (All,i,COM)
DUMMY_NT(i) # One for non-traditional exports, else zero  FIa#;
        (All,j,IND)
DUMMY_WAT(j) # One for water and air transport, else zero FIo#;
      (All,s,COUNTSET)
DUM_COUNT(s) # Equals 1 for s = 1 to TAU, equals 0 for s > TAU Fw#;
DUM_IT1 # One in 1st iter of forecast or policy simulation, else zero FIt#;
        (All,t,TIME)
DUM_TIME(t) # Equals 1 if t  = YEAR, 0 otherwise FIt#;
        (All,t,TIME)
DUM_TIME_LAG(t) # Equals 1 if t= YEAR-1, else 0 FIt#;
        (All,i,COM)
DUM_TW(i) # Equals 10*AVIMPSH(i) if AVIMPSH(i) < 0.1, else 1 Fp#;
DUM_YEAR1 # Zero in first year, then one RU#;
        (All,j,IND)
EEQROR(j) # Equilibrium ror, static expectations Fr#;
EMPLOY #  Aggregate employment: 1 in init. sol. for yr 1 RU#;
EMPLOY_B # Aggregate employment in year t, base FIu#;
EMPLOY_OLD # Aggregate employment, f'cast: 1 in init. sol. for yr 1 RU#;
EMPLOY_O_B # Aggregate employment in year t, forecast value, base FIu#;
EMPLOY_O_L_B # Aggregate employment in year t-1, forecast value,base R#;
EMPLOY_L_B # Aggregate employment in year t-1, base R#;
        (All,i,COM)
EPS(i) # Household expenditure elasticities Fk#;
        (All,j,IND)
EROR_B(j) # Expected ror in initial soln for t, usually exp. ror for t-1 FIt#;
        (All,j,IND)
EROR_F(j) # Expect. ror, imposed in soln. for year t, beyond iter. one FIt#;
        (All,j,IND)(All,t,TIME)
EROR_G(j,t)
  # Matrix for transfer of info on expected rors between iterations RFItU#;
        (All,j,IND)(All,l,TIME)
EROR_G_B(j,l)
  # Supplies current & lag expect. rors to year t beyond iteration one FIt#;
        (All,i,COM)(All,j,COM)
ETA(i,j) # Household price elasticities Fk#;
EXCH # Exchange rate ($ foreign per $ Aus), average in year RU#;
EXCH_B # Exchange rate ($ foreign per $ Aus), average in year, base RFIw#;
EXCH_T # Exchange rate ($ foreign per $ Aus) at the start of year RU#;

EXCH_T_1 # Exchange rate ($ foreign per $ Aus) at the end of year RU#;
EXCH_T_1_B # Exchange rate ($ foreign per $ Aus) at the end of year, base FIx#;
EXCH_T_B # Exchange rate ($ foreign per $ Aus) at the start of year, base FIx#;
        (All,j,IND)
EXPORTS_IND(j) # Basic value of exports from industry j Fo#;
        (All,i,COM)
EXP_ELAST(i) # Export demand elasticities R#;
EXP_ELAST_N # Export demand elasticity for nontraditional composite Fa#;
EXP_ELAST_T # Export demand elasticity for tourism composite Fa#;
FA_T # Foreign assets, start of year Fw#;
FA_T1 # Foreign assets, end of year RU#;
FCDCATT # $A value of f'gn credit given in dom. currency, start of year Fx#;
FCDCATT_1 # $A value of f'gn credit given in dom. currency, end of year Fx#;
FCDCATT_1_B # $A value of f'gn credit given in dom. curr., end of yr, base FIx#;
FCDCATT_B # $A value of f'gn credit given in dom. curr., start of yr, base FIx#;
FCFCATT # $A value of f'gn credit given in f'gn currency, start of year Fx#;
FCFCATT_1 # $A value of f'gn credit given in f'gn currency, end of year Fx#;
FCFCATT_1_B # $A value of f'gn credit given in f'gn curr., end of yr, base FIx#;
FCFCATT_B # $A value of f'gn credit given in f'gn curr., start of yr, base FIx#;
FDDCATT # $A value of f'gn debt incured in dom. currency, start of year Fx#;
FDDCATT_1 # $A value of f'gn debt incured in dom. currency, end of year Fx#;
FDDCATT_1_B # $A value of f'gn debt incured in dom. curr., end of yr, base FIx#;
FDDCATT_B # $A value of f'gn debt incur. in dom. curr., start of yr, base FIx#;
FDFCATT # $A value of f'gn debt incured in f'gn currency, start of yr Fx#;
FDFCATT_1 # $A value of f'gn debt incured in f'gn currency, end of yr Fx#;
FDFCATT_1_B # $A value of f'gn debt incur. in f'gn curr., end of yr, base FIx#;
FDFCATT_B # $A value of f'gn debt incur. in f'gn curr., start of yr, base FIx#;
FEMPADJ # Level of the shift variable in E_d_f_empadj RU#;
FEMPADJ_B # Level of the shift variable in E_d_f_empadj in t-1, FIu#;
FEMPADJ_O # Level of the shift variable in E_d_f_empadj, forecast value R#;
FE_A_T # $A value of equity in Australia held by f'gners, start of year Fx#;
FE_A_T1_ADD # Preliminary estimate of FE_A_T_1 FIx#;
        (All,j,IND)
FE_A_T1_J(j) # $A val. of f'gn-owned j equity, end of yr FIxFx#;
        (All,j,IND)
FE_A_T1_J_B(j) # $A val. of f'gn-owned j equity, end of yr, base FIx#;
FE_A_T_1 # $A value of equity in Australia held by f'gners, end of yr Fx#;
FE_A_T_ADD # Preliminary estimate of FE_A_T FIx#;
        (All,j,IND)
FE_A_T_J(j) # $A val. of f'gn-owned j equity , start of yr, FIxFx#;
        (All,j,IND)
FE_A_T_J_B(j) # $A val. of f'gn-owned j equity , start of yr, base FIx#;

FGN_SHD # Scalar shifter in formula for dividends on f'gn equity in j RFIxU#;
        (All,j,IND)
FGN_SHD_J(j) # Vector shifter in formula for dividends on f'gn equity in j RU#;
FL_T # Foreign labilities, start of year RU#;
FL_T1 # Foreign labilities, end of year RU#;
FRISCH # Frisch coeff: neg of ratio of exp to luxury exp, about -1.8  RU#;
F_EEQROR # Scalar shifter, equilib. rate of return RU#;
        (All,j,IND)
F_EEQROR_J(j) # Vector shifter, equilib. rate of return RU#;
GDPEX # Nominal GDP from expenditure side Fh#;
GDPIN # Nominal GDP from income side Fi#;
GNP # Gross national product Fw#;
GNP_B # Gross national product, base RFIw#;
GOV_DEF # Government deficit or financing transation Fv#;
GOV_SAV # Government saving Fy#;
GRANT # Grants & transf. other than unemploy, age, & oth person. bens. RU#;
        (All,j,IND)
G_VINVEST(j) # Value of government investment in industry j RU#;
        (All,cc,COMPCOM)(All,j,IND_JP)
H0CC(cc,j) # Share of composite commodity cc in industry j's revenue Fc#;
        (All,i,COM)(All,j,IND)
H0CI(i,j) # Share of commodity i in total revenue of industry j Fc#;
        (All,m,OCC)(All,j,IND)
HOURS(m,j) # Labour input by occupation and industry, hours Fi#;
HOURSTOT # Aggregate labour input, hours Fi#;
HOUS_DIS_INC # Household disposable income Fy#;
HOUS_SAV # Household saving Fy#;
        (All,i,COM)
IMPCOST(i) # Total ex-duty imports of good i Ff#;
        (All,i,COM)
IMPORTS(i) # Total basic-value imports of good i Ff#;
        (All,j,IND)
IMPORTS_IND(j) # Basic value of imported inputs to industry j Fo#;
INCTAX # Income tax Fu#;
        (Integer)(All,j,IND)
INDTOIA20(j) # INDTOIA20(j) gives position of j in the set INDSAGGTO20 R#;
        (Integer)(All,j,IND)
INDTOIA26(j) # INDTOIA26(j) gives position of j in the set INDSAGGTO26 R#;
        (Integer)(All,i,IND)
INDTOLABDIND(i) # Gives the element of LAB_DATA_IND containing ind i R#;
INF # Rate of inflation Fs#;
INF_L # Lagged rate of inflation Ft#;

INT # Nominal Rate of Interest Fs#;
INTFCDC # $A value of interest on f'gn credits given in dom. currency RU#;
INTFCFC # $A value of interest on f'gn credits given in f'gn currency RU#;
INTFDDC # $A value of interest on f'gn debt incured in dom. currency RU#;
INTFDFC # $A value of interest on f'gn debt incured in f'gn currency RU#;
INT_L # Lagged nominal interest rate Ft#;
INT_PSD # Nominal rate of interest on public sector debt Fs#;
ITER_ADJUST # Used in adjusting the iteration number FIt#;
ITER_NUM # Iteration number, used when expectations are forward-looking RU#;
ITER_NUM_B # Iter. number, used when expectations are forward-looking FIt#;
        (All,i,COM)(All,j,COM)
KD(i,j) # Kroneker's delta: 1 if i=j, else 0 Fk#;
        (All,j,IND)
K_GR(j) # Capital growth rate in year t-1 FqFr#;
        (All,j,IND)
K_GR_MAX(j)
    # Capital supply curve: upper bound on normal growth range for capital FIr#;
        (All,j,IND)
K_GR_MIN(j) # Minimum possible capital growth FIr#;
        (All,j,IND)
LABIND_J(j) # Total wage bill in industry j Fi#;
        (All,m,OCC)
LABOCC(m) # Total hours worked in occupation m Fi#;
        (All,m,OCC)(All,j,IND)
LABOCCIND(m,j) # Wage bill for occupation m in industry j RU#;
LAB_SUP # Labour supply RU#;
LAB_SUP_O # Labour supply, forecast run RU#;
        (All,j,IND)
LAND(j) # Total rental value of land used in industry j RU#;
      (All,i,COM)(All,s,SOURCE)
LEVX3(i,s) # Level of consumption by commodity/source  RU#;
      (All,i,COM)(All,s,SOURCE)
LEVX3B(i,s) # Level of consumption by commodity/source, base  FIu#;
      (All,i,COM)(All,s,SOURCE)
LEVX3B_O(i,s) # Level of consumption by com/source in f'cast sim., base FIu#;
      (All,i,COM)(All,s,SOURCE)
LEVX3_O(i,s) # Level of consumption by commodity/source in f'cast sim. RU#;
LEV_CPI # Level of the CPI RU#;
LEV_CPI_B # Level of the CPI, base, usually CPI for t-1 FIs#;
LEV_CPI_2L # CPI double lagged, that is for t-2 RU#;
LEV_CPI_2L_B # CPI double lagged, base FIt#;
LEV_CPI_L # Lagged level of the CPI, that is level in t-1 RU#;

LEV_CPI_L_B # Lagged level of the CPI, base FIs#;
        (All,j,IND)
LEV_F1OCT(j) # Level of F1OCT: shift in price of other cost tickets Fl#;
        (All,i,COM)
LEV_PTAXM(i) # Level of power of tariff RU#;
LEV_XI2T # Level of capital goods price index at start of year t RU#;
        (All,i,COM)
LOST_GOODS(i) # Cost of making good i less sales of good i: should be zero Fj#;
LS_COEFF # Labour supply adjustment coefficient R#;
        (All,i,COM)(All,j,IND)
MAKE(i,j) # Basic value of production of commodity i by industy j RU#;
        (All,j,IND)
MAKE_COSTS(j) # Basic value of output by ind j: column sums of make Fc#;
        (All,i,COM)
MAKE_SALES(i) # Basic value of outputs of good i: row sums of make matrix Fc#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
MAR1(i,s,j,r) # Basic values of margins on intermediate inputs RU#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
MAR2(i,s,j,r) # Basic values of margins on inputs to capital creation RU#;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
MAR3(i,s,r) # Basic values of margins on flows to consumption RU#;
        (All,i,COM)(All,r,MARGCOM)
MAR4(i,r) # Basic values of margins on exports RU#;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
MAR5(i,s,r) # Basic values of margins on flows to government RU#;
        (All,cc,COMPCOM)(All,j,IND_JP)
MH0CC(cc,j) # CRETH share of composite cc in revenue of industry j Fc#;
        (All,i,COM)(All,na,NCOM)
MM(i,na) # Consumption of MONASH good i in National Acc. Cat. na RFInU#;
        (All,i,COM)
MMI(i) # Row sum, MM: Nat Acc's estimate of consum. of MONASH good i FIn#;
        (All,na,NCOM)
MMN(na) # Column sum, MM: consumption of National Accounts good na FIn#;
NAT_SAV # National saving: household saving plus public sector surplus Fy#;
NETINT_G # Net interest paid by government Fw#;
NET_FLT # $A value of net foreign liabilities RU#;
NET_FLTF # $F value of net foreign liabilities Fw#;
NET_FLTF_B # $F value of net foreign liabilities, base RFIw#;
NET_TAXTOTG # Net genuine income and indirect taxes collected Fv#;
NEWFCDC # $A value of new f'gn credit given in dom. currency in t Fx#;
NEWFCFC # $A value of new f'gn credit given in f'gn currency in t Fx#;
NEWFDDC # $A value of new f'gn debt in dom. currency in t Fx#;

NEWFDFC # $A value of new f'gn debt incurred in f'gn currency in t Fx#;
NEW_AE_F # $A value of new foreign equity held by Austalians in t Fx#;
NEW_FE_A # $A value of new equity in Australia held by f'gners in t Fx#;
        (All,j,IND)
NEW_FE_A_J(j) # $A value of new f'gn-owned j equity in t Fw#;
NFLATT # $A value of net f'gn liabiliities, start of year Fw#;
NOFITERS # Number of forecast iterations in rational expectation sims R#;
NUMCOM # Number of commodities Fa#;
NUMCOMP # Number of composite commodities Fa#;
NUMFAC # Number of primary factors Fa#;
NUMIND # Number of industries Fa#;
NUMOCC # Number of occupations Fa#;
NYEARS # Length in years of the simulation horizon R#;
ONE_IT1_REP # Used in rat. exp.: one in 1st iter of policy, else zero FIt#;
ONE_ITER1 # Used in rat. exp.: one in 1st iter of forecast, else zero FIt#;
OTHBEN # Benefits paid by government, except age and unemployment RU#;
OTHCAPGOV # Other gov. cap. expend., eg purch & sale of 2nd hand assets RU#;
        (All,j,IND)
OTHCOST(j) # Cost of other cost tickets to industry j RFIbU#;
OTHGOVREV # Gov. revenue apart from taxes & interest, e.g. profits  RU#;
        (All,i,COM)(All,s,SOURCE)
P0LEV(i,s) # Levels of basic prices RU#;
        (All,j,IND)
P1OCT(j) # Price of other cost tickets RU#;
        (All,j,IND)
PCAP_AT_T(j) # Asset price of capital stocks, start of forecast year RU#;
        (All,j,IND)
PCAP_AT_T1(j) # Asset price of capital by industry, end of year  RU#;
        (All,j,IND)
PCAP_AT_T1_B(j) # Asset price of capital by industry, end of year, base FIq#;
        (All,j,IND)
PCAP_AT_T_B(j) # Asset price of capital, start of year, base FIq#;
        (All,j,IND)
PCAP_J(j) # Asset price of capital by industry, average in year RU#;
        (All,j,IND)
PCAP_J_B(j) # Asset price of capital by industry, average in year, base RFIq#;
        (All,j,IND)
PCAP_J_L(j) # Lagged asset prices of capital, average in year RU#;
        (All,j,IND)
PCAP_J_L_B(j) # Init. sol for lagged asset price of capital,ave. in yr FIq#;
        (All,j,IND)
POW_PAYROLL(j) # Power of payroll tax RU#;

        (All,j,IND)
POW_TAX0(j) # Power of production taxes Fd#;
PSDATT # Public sector debt, start of year RU#;
PSDATTPLUS1 # Public sector debt, end of year Fv#;
PSDATT_1_B # Public sector debt, end of year, base FIv#;
PSDATT_B # Public sector debt, start of year, base FIv#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
PURCHVAL1(i,s,j) # Purchasers' values of intermediate inputs Fd#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
PURCHVAL2(i,s,j) # Purchasers' values of inputs to capital creation Fd#;
        (All,i,COM)(All,s,SOURCE)
PURCHVAL3(i,s) # Purchasers' values of flows to households Fd#;
        (All,i,COM)
PURCHVAL4(i) # Purchasers' values of exports Fd#;
        (All,i,COM)(All,s,SOURCE)
PURCHVAL5(i,s) # Purchasers' values of flows to government Fd#;
        (All,j,IND)
PURE_PROFITS(j) # Sales by industry j minus j's costs: should be zero Fj#;
        (All,j,IND)
QCAPATT(j) # Quantity of capital stocks, start of forecast year, ie K(t) Fq#;
        (All,j,IND)
QCAPATTPLUS1(j) # Qty of capital stocks, end of forecast year, ie K(t+1) Fq#;
        (All,j,IND)
QCAPATT_B(j) # Qty cap stock, start of data year, year 0 or t-1, base RFIq#;
        (All,j,IND)
QCAPPLUS1_B(j) # Qty of cap stock, end of data year, base  FIq#;
        (All,j,IND)
QINVEST(j) # Quantity of investment by industry j Fq#;
        (All,j,IND)
QINV_BASE(j) # Quantity of investment by industry j, base solution FIq#;
RALPH # Share of depreciation that is tax deductible R#;
RINT # Real Rate of Interest RU#;
RINT_B # Initial solution for real interest rate FIt#;
RINT_L # Lagged real interest rate RU#;
RINT_L_B # Initial solution for lagged real interest rate FIt#;
RINT_PSD # Real rate of interest, on public sector debt RU#;
RINT_PT_SE # Post-tax real interest rate, static expectations Fs#;
ROIFOREIGN # Foreign rate of interest R#;
        (All,j,IND)
RORN(j) # Normal rates of return, given as data R#;
        (All,j,IND)
ROR_ACT_L(j) # Lagged rate of return, ror in t-1 Ft#;

        (All,j,IND)
ROR_ACT_L_B(j) # Lagged rate of return, init. soln., usually ror in t-2 FIt#;
        (All,j,IND)
ROR_SE(j) # Rates of return in year t: static expectations Fs#;
        (All,j,IND)
ROR_SE_BASE(j) # Static expect. of rors in init. soln. for year t RFIs#;
        (All,j,IND)
ROTHCOST(j) # Quantity of other costs Fl#;
RWAGE # Real wage in year t, CPI deflated RU#;
RWAGE_B # Real wage in year t, CPI deflated, base FIu#;
RWAGE_L_B # Real wage in year t-1, CPI deflated,base  R#;
RWAGE_OLD # Real wage in year t, CPI deflated, forecast RU#;
RWAGE_OLD_B # Real wage in year t, CPI deflated, forecast, base FIu#;
RWAGE_O_L_B # Real wage in year t-1, CPI deflated, forecast, base R#;
RWAGE_PT # Real post-tax wage in year t, CPI deflated Fu#;
RWAGE_PT_B # Real post-tax wage in year t, CPI deflated, base FIu#;
RWAGE_PT_L_B # Real post-tax wage in year t-1, CPI deflated, base  FIu#;
RWAGE_PT_OLD # Real post-tax wage in year t, CPI deflated, forecast Fu#;
RWAGE_PT_O_B # Real post-tax wage in year t, CPI deflated, forecast, base FIu#;
RWEAL_T # Real start-of-year national wealth in t RFIwU#;
RWEAL_T_B # Real start-of-year national wealth in t, base FIw#;
RWEAL_T_L # Real start-of-year national wealth in t-1, base R#;
RWEAL_T_L_O # Real start-of-year national wealth in t-1, forecast, base R#;
RWEAL_T_O # Real start-of-year national wealth in t, forecast, base RFIw#;
RW_PT_O_L_B # Real post-tax wage in t-1, CPI deflated, forecast, base FIu#;
R_CADGDP # Ratio of current account deficit to GDP Fx#;
R_CADGDP_B # Ratio of current account deficit to GDP, base FIx#;
R_DEFGDP_B # Ratio public sector deficit to GDP in base year FIv#;
R_NFL_GDP # Ratio of net foreign liabilities to GDP Fx#;
R_NFL_GDP_B # Ratio of net foreign liabilities to GDP, base FIx#;
R_NFL_VCAP # Ratio of net foreign liabilities to capital stock Fx#;
R_NFL_VCAP_B # Ratio of net foreign liabilities to capital stock, base FIx#;
R_PSDGDP # Ratio of st-of-yr public sector debt to GDP Fv#;
R_PSDGDP_B # Ratio of st-of-yr public sector debt to GDP FIv#;
        (All,i,COM_JP)(All,j,IND_JP)
S0C(i,j) # Share of i in j's rev from the comp good of which i is part Fc#;
        (All,i,COM)
S3COM(i) # Share of good i in total household expenditure Fk#;
        (All,i,COM)
SALES(i) # Total sales of domestic good i Fj#;
        (All,j,IND)
SHFEA_T1_J(j) # F'gn shares in the ownership of capital at end of yr RFIxU#;

        (All,j,IND)
SHFEA_T_J(j) # F'gn shares in the ownership of capital at start of yr RFIxU#;
SHT1_FCDC # Share in f'gn assets of f'gn credits in dom curr., end of yr RU#;
SHT1_FCFC # Share in f'gn assets of f'gn credits in f'gn curr., end of yr RU#;
SHT1_FDDC # Share in f'gn liab of f'gn debt in dom curency, end of yr RU#;
SHT1_FDFC # Share in f'gn liab of f'gn debt in f'gn curency, end of yr RU#;
SHT_FCDC # Share in f'gn assets of f'gn credits in dom curr., start of yr RU#;
SHT_FCFC # Share in f'gn assets of f'gn credits in f'gn curr., start of yr RU#;
SHT_FDDC # Share in f'gn liab of f'gn debt in dom curency, start of yr RU#;
SHT_FDFC # Share in f'gn liab of f'gn debt in f'gn curency, start of yr RU#;
        (All,cc,COMPCOM)(All,j,IND_JP)
SIG0CC(cc,j) # CRETH params giving ease of transform. btwn composite coms R#;
        (All,i,COM)
SIGMA1(i) # Armington imp/dom substitution elasticities for intermediate R#;
        (All,j,IND)
SIGMA1CAP(j) # CRESH substitution parameter for capital R#;
        (All,j,IND)
SIGMA1LAB(j) # CRESH substitution parameter for labour R#;
        (All,m,OCC)(All,j,IND)
SIGMA1LABOCC(m,j) # CRESH substit param for labour of type m in ind j R#;
        (All,j,IND)
SIGMA1LND(j) # CRESH substitution parameter for land R#;
        (All,i,COM)
SIGMA2(i) # Armington imp/dom substitution elasticities for investment R#;
        (All,i,COM)
SIGMA3(i) # Armington imp/dom substitution elasticities for consumption R#;
        (All,cc,COMPCOM)(All,j,IND_JP)
SIGMACC(cc,j) # Within composite transformation elasticities R#;
SMURF # Recip. of slopes of cap. supply curves at K_GR(j) = TREND_K(j) R#;
        (All,s,SOURCE)
SOURCEDOM(s) # Equals 1 if s="dom", else 0 Fa#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
SOURCE_SHR1(i,s,j) # Share of good (i,s) in j's intermediate use of good i Fe#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
SOURCE_SHR2(i,s,j) # Share of good (i,s) in j's investment use of good i Fe#;
        (All,i,COM)(All,s,SOURCE)
SOURCE_SHR3(i,s) # Share of good (i,s) in household  use of good i Fe#;
        (All,i,COM)(All,s,SOURCE)
SOURCE_SHR5(i,s) # Share of good (i,s) in government use of good i Fe#;
        (All,s,FAC)(All,j,IND)
SOURCE_SHRLK(s,j) # Labour and capital shares in labour-capital inputs Fm#;

        (All,i,COM)
SS3COM(i) # Subsistence expenditure on i as share of budget Fk#;
        (All,v,FAC)(All,j,IND)
SSTAR1FAC(v,j) # CRESH primary-factor shares Fm#;
        (All,m,OCC)(All,j,IND)
SSTAR1LABOCC(m,j) # CRESH occupational-labour shares Fm#;
SUMDELTA # Initial sum of the marginal budget shares FIk#;
        (All,j,IND_JP)
SUMSIG(j) # Denominator in CRETH revenue shares Fc#;
        (All,i,COM)
TARIFF(i) # Tariff revenue Ff#;
TAU # Years in hist/decomp sim. eg for 1987 to 1994, TAU = 7 R#;
        (All,j,IND)
TAX0(j) # Collection of production taxes Fd#;
        (All,j,IND)
TAX0F(j) # Collection of phantom production taxes RFIbU #;
        (All,j,IND)
TAX0G(j) # Collection of genuine production taxes RU #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX1(i,s,j) # Total tax collection on intermediate use of (i,s) by ind j Fd#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX1F(i,s,j) # Phantom tax collection on intermed. use of (i,s) by ind j RFIbU#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX1G(i,s,j) # Genuine tax collection on intermediate use of (i,s) by ind j RU#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX2(i,s,j) # Total tax collection on investment use of (i,s) by ind j Fd#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX2F(i,s,j) # Phantom tax collection on invest. use of (i,s) by ind j RFIbU#;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX2G(i,s,j) # Genuine tax collection on investment use of (i,s) by ind j RU#;
        (All,i,COM)(All,s,SOURCE)
TAX3(i,s) # Total tax collection on household use of good (i,s) Fd#;
        (All,i,COM)(All,s,SOURCE)
TAX3F(i,s) # Phantom tax collection on household use of good (i,s) RFIbU#;
        (All,i,COM)(All,s,SOURCE)
TAX3G(i,s) # Genuine tax collection on household use of good (i,s) RU#;
        (All,i,COM)
TAX4(i) # Total tax collection on exports of good i Fd#;
        (All,i,COM)
TAX4F(i) # Phantom tax collection on exports of good i RFIbU#;
        (All,i,COM)
TAX4G(i) # Genuine tax collection on exports of good i RU#;

        (All,i,COM)(All,s,SOURCE)
TAX5(i,s) # Total tax collection on usage of good (i,s) by government Fd#;
        (All,i,COM)(All,s,SOURCE)
TAX5F(i,s) # Phantom tax collection on usage of good (i,s) by govt. RFIbU#;
        (All,i,COM)(All,s,SOURCE)
TAX5G(i,s) # Genuine tax collection on usage of good (i,s) by government RU#;
TAXFGNGOV # $A val of income taxes paid by f'gners to Australia RU#;
        (All,i,COM)
TAXF_COM(i) # Total phantom taxes on commodity i Fg#;
TAX_CAP # Tax collected from capital income Fu#;
TAX_K_RATE # Rate of tax on capital and land income RU#;
TAX_LAB # Tax collected from labour income Fu#;
TAX_LND # Tax collected from land income Fu#;
TAX_L_RATE # Rate of tax on labour income, year t RU#;
TAX_L_RATE_L # Rate of tax on labour income in year t-1, base R#;
TAX_L_RATE_O # Rate of tax on labour income, forecast, year t RU#;
TAX_L_R_O_L # Rate of tax on labour income in year t-1, forecast, base R#;
TINY # Tiny number often used to avoid zero divides Fa#;
TOTFAC # Total primary factor payments Fi#;
        (All,i,COM)
TOTFACCOM(i) # Primary factor input to production of commodity i Fi#;
        (All,j,IND)
TOTFACIND(j) # Total primary factor payments in industry j Fi#;
        (All,i,COM)
TOTMARGINS(i) # Total usage of good i for margins purposes Fj#;
        (All,i,COM)
TOT_T(i) # Total trade, exports plus imports of good i Fh#;
        (All,i,COM)(All,j,IND)
TPURCHVAL1(i,j) # Purchasers' value of intermediate use of good i by ind j Fd#;
        (All,i,COM)(All,j,IND)
TPURCHVAL2(i,j) # Purchasers' value of investment use of good i by ind j Fd#;
        (All,i,COM)
TPURCHVAL3(i) # Purchasers' value of household use of good i Fd#;
        (All,i,COM)
TPURCHVAL5(i) # Purchasers' value of government use of good i Fd#;
TRANS # Transfers from the government Fv#;
        (All,i,COM)
TRANSERVDUM(i) # 1 if and only if com. i is transport services, else 0 Fa#;
        (All,j,IND)
TREND_K(j) # Trend growth in capital R#;
TRNFGNAUS # Net transfers from foreigners to Australians Fw#;
TRNFGNAUS_B # Net transfers from foreigners to Australians, base RFIw#;

TRNFGNHH # $A val of trans, f'gners to Aust. h'hlds: eg migrant assets RU#;
TRNOTHFGN # $A value of other trans to f'gners: eg gifts RU#;
        (All,i,COM)
TTPURCHVAL(i) # Purchasers' value of sales of i in Australia Fd#;
UNEMPBEN # Unemployment benefits RU#;
        (All,j,IND)
VALUEADD(j) # Value added by industry Fi#;
        (All,j,IND)
VCAP_AT_T(j) # Start of year capital stocks valued at start of yr prices RFIqU#;
        (All,j,IND)
VCAP_AT_T1(j) # End of year capital valued at end of yr prices Fx#;
        (All,j,IND)
VCAP_AT_TM(j) # Val of capital, start of f'cast year in mid-yr prices Fq#;
        (All,j,IND)
VINVEST(j) # Value of investment by industry j, current prices Fh#;
        (All,m,OCC)(All,j,IND)
WAGE(m,j) # Wage rate excluding payroll tax by occupation and industry RU#;
WEALTH_T # National wealth at start of year t Fw#;
WP # World prices in $f'gn, average in year RU#;
WPT # World prices in $f'gn, start of year RU#;
WPTPLUS1 # World prices in $f'gn, end of year RU#;
WPT_1_B # World prices in $f'gn, end of year, base FIx#;
WPT_B # World prices in $f'gn, start of year, base FIx#;
WP_B # World prices in $f'gn, average in year, base FIx#;
YEAR # Year of current solution RU#;
YEAR_B # Year of current solution FIt#;
YR_POLICY # Year in which policy shock is first anticipated R#;
ZERO_PYR1 # Zero in 1st yr of a rational-expect. policy sim, else 1 FIt#;

! 18.4 Read instructions for coefficients listed alphabetically !

Read
ABSCOMNO from file SETINFO Header"ABSN";
ADJ_COEFF from file PARAMS Header"ADJC";
ADJ_RE from file ROREXT Header"ADRE";
AGEBEN from file SRFM Header"0040";
ALPHA1 from file PARAMS  Header"ALPH";
ALPHA2 from file PARAMS  Header"ALP2";
APSGNP_B from file SRFM Header"APSB";
AV_ROR_SE_B from file EXTRA Header"ARSB";
B3SH from file EXTRA Header"B3SH";
B3SHO from file EXTRA3 Header"B3SH";
(All,i,COM)(All,j,IND) BAS1(i,"dom",j) from file FID Header"F001";

(All,i,COM)(All,j,IND) BAS1(i,"imp",j) from file FID Header"F002";
(All,i,COM)(All,j,IND) BAS2(i,"dom",j) from file FID Header"F003";
(All,i,COM)(All,j,IND) BAS2(i,"imp",j) from file FID Header"F004";
(All,i,COM) BAS3(i,"dom") from file FID Header"F005";
(All,i,COM)  BAS3(i,"imp") from file FID Header"F006";
BAS4 from file FID Header"F007";
(All,i,COM) BAS5(i,"dom") from file FID Header"F008";
(All,i,COM) BAS5(i,"imp") from file FID Header"F009";
(All,i,COM) BAS6(i,"dom") from file FID Header"BS6D";
(All,i,COM) BAS6(i,"imp") from file FID Header"BS6I";
BETA_CH from file PARAMS Header"BETA";
CADEF from file SRFM Header"CADF";
CAPITAL from file FID Header"F011";
CCPROD from file PARAMS Header"CCPR";
COEFF_TIME from file PARAMS Header"TIME";
COMTOABSCOM from file SETINFO Header"CTAB";
C_TWIST_SRC from file SETINFO Header"CTSR";
DELTA from file EXTRA Header"MBSH";
DEP from file PARAMS Header"DPRC";
DIFF from file PARAMS Header"DIFF";
DIV_AE_F from file SRFM Header"0031";
DIV_FE_A from file SRFM Header"0028";
DUMF from file PARAMS Header"DMPH";
DUMMY_DEC from file PARAMS Header"DMDC";
DUM_YEAR1 from file SRFM Header"0045";
EMPLOY from file EXTRA Header"EMPT";
EMPLOY_L_B from file EXTRA4  Header"EMPT";
EMPLOY_OLD from file EXTRA3 Header"EMPT";
EMPLOY_O_L_B from file EXTRA5 Header"EMPT";
EROR_G from file ROREXT Header"RORG";
EXCH from file SRFM Header"0008";
EXCH_B from file EXTRA Header"EXCB";
EXCH_T from file SRFM Header"0007";
EXCH_T_1 from file srfm Header"0047";
EXP_ELAST from file PARAMS Header"P018";
FA_T1 from file SRFM Header"0018";
FEMPADJ from file EXTRA  Header"EADJ";
FEMPADJ_O from file EXTRA3 Header"EADJ";
FGN_SHD from file SRFM Header"0033";
FGN_SHD_J from file SRFM Header"0032";
FL_T from file SRFM Header"0005";
FL_T1 from file SRFM Header"0006";

FRISCH from file EXTRA Header"P021";
F_EEQROR from file EXTRA Header"FCSE";
F_EEQROR_J from file EXTRA Header"FSTA";
GNP_B from file SRFM Header"GNPB";
GRANT from file SRFM Header"0042";
G_VINVEST from file SRFM Header"0044";
INDTOIA20 from file SETINFO Header"TI20";
INDTOIA26 from file SETINFO Header"TI26";
INDTOLABDIND from file SETINFO Header"ITLI";
INTFCDC from file SRFM Header"0030";
INTFCFC from file SRFM Header"0029";
INTFDDC from file SRFM Header"0027";
INTFDFC from file SRFM Header"0026";
ITER_NUM from file ITER Header"ITNO";
LABOCCIND from file FID Header"F010";
LAB_SUP from file EXTRA Header"LSUP";
LAB_SUP_O from file EXTRA3 Header"LSUP";
LAND from file FID Header"F012";
LEVX3 from file EXTRA Header"LVX3";
LEVX3_O from file EXTRA3 Header"LVX3";
LEV_CPI from file EXTRA Header"LCPI";
LEV_CPI_2L from file EXTRA Header"LCP2";
LEV_CPI_L from file EXTRA Header"LCPL";
LEV_PTAXM from file FID Header"LPTR";
LEV_XI2T from file EXTRA Header"LXI2";
LS_COEFF from file PARAMS Header"LSCO";
MAKE from file FID Header"Y ";
        (All,i,COM)(All,j,IND)(All,r,MARGCOM)
MAR1(i,"dom",j,r) from file FID Header"FC55";
        (All,i,COM)(All,j,IND)(All,r,MARGCOM)
MAR1(i,"imp",j,r) from file FID Header"FC56";
        (All,i,COM)(All,j,IND)(All,r,MARGCOM)
MAR2(i,"dom",j,r) from file FID Header"FC57";
        (All,i,COM)(All,j,IND)(All,r,MARGCOM)
MAR2(i,"imp",j,r) from file FID Header"FC58";
(All,i,COM)(All,r,MARGCOM) MAR3(i,"dom",r) from file FID Header"F019";
(All,i,COM)(All,r,MARGCOM) MAR3(i,"imp",r) from file FID Header"F020";
MAR4 from file FID Header"F023";
(All,i,COM)(All,r,MARGCOM) MAR5(i,"dom",r) from file FID Header"F025";
(All,i,COM)(All,r,MARGCOM) MAR5(i,"imp",r) from file FID Header"F026";
MM from file EXTRA Header"MM02";
NET_FLT from file SRFM Header"NFLT";

NET_FLTF_B from file SRFM Header"FLFB";
NOFITERS from file ROREXT Header"NFIT";
NYEARS from file ROREXT Header"HORZ";
OTHBEN from file SRFM Header"0041";
OTHCAPGOV from file SRFM Header"0035";
OTHCOST from file FID Header"F013";
OTHGOVREV from file SRFM Header"0043";
P0LEV from file EXTRA Header"P0LV";
P1OCT from file EXTRA Header"POTH";
PCAP_AT_T from file EXTRA Header"PCAP";
PCAP_AT_T1 from file EXTRA Header"PCP1";
PCAP_J from file EXTRA Header"PCPM";
PCAP_J_B from file SRFM Header"0060";
PCAP_J_L from file EXTRA Header"PCPL";
POW_PAYROLL from file SRFM Header"PPAY";
PSDATT from file SRFM Header"0037";
QCAPATT_B from file EXTRA Header"QCTB";
RALPH from file PARAMS Header"RALF";
RINT from file EXTRA Header"RINT";
RINT_L from file EXTRA Header"RNTL";
RINT_PSD from file SRFM Header"0036";
ROIFOREIGN from file setinfo Header"ROIF";
RORN from file EXTRA Header"RORN";
ROR_SE_BASE from file EXTRA Header"RRSB";
RWAGE from file EXTRA  Header"RWAG";
RWAGE_L_B from file EXTRA4 Header"RWAG";
RWAGE_OLD from file EXTRA3 Header"RWAG";
RWAGE_O_L_B from file EXTRA5 Header"RWAG";
RWEAL_T from file EXTRA Header"RWET";
RWEAL_T_L from file EXTRA4 Header"RWET";
RWEAL_T_L_O from file EXTRA5 Header"RWET";
RWEAL_T_O from file EXTRA3 Header"RWET";
SHFEA_T1_J from file SRFM Header"0010";
SHFEA_T_J from file SRFM Header"0009";
SHT1_FCDC from file SRFM Header"0016";
SHT1_FCFC from file SRFM Header"0015";
SHT1_FDDC from file SRFM Header"0004";
SHT1_FDFC from file SRFM Header"0003";
SHT_FCDC from file SRFM Header"0014";
SHT_FCFC from file SRFM Header"0013";
SHT_FDDC from file SRFM Header"0002";
SHT_FDFC from file SRFM Header"0001";

SIG0CC from file PARAMS Header"CRTH";
SIGMA1 from file PARAMS Header"P015";
SIGMA1CAP from file PARAMS Header"P029";
SIGMA1LAB from file PARAMS Header"P028";
SIGMA1LABOCC from file PARAMS Header"P031";
SIGMA1LND from file PARAMS Header"P030";
SIGMA2 from file PARAMS Header"P016";
SIGMA3 from file PARAMS Header"P017";
SIGMACC from file PARAMS Header"SGCC";
SMURF from file PARAMS Header"MURF";
TAU from file SETINFO Header"TAU ";
TAX0G from file FID Header"FT0G";
TAX0F from file FID Header"FT0F";
(All,i,COM)(All,j,IND) TAX1F(i,"dom",j) from file FID Header"F51F";
(All,i,COM)(All,j,IND) TAX1F(i,"imp",j) from file FID Header"F52F";
(All,i,COM)(All,j,IND) TAX1G(i,"dom",j) from file FID Header"F51G";
(All,i,COM)(All,j,IND) TAX1G(i,"imp",j) from file FID Header"F52G";
(All,i,COM)(All,j,IND) TAX2F(i,"dom",j) from file FID Header"F53F";
(All,i,COM)(All,j,IND) TAX2F(i,"imp",j) from file FID Header"F54F";
(All,i,COM)(All,j,IND) TAX2G(i,"dom",j) from file FID Header"F53G";
(All,i,COM)(All,j,IND) TAX2G(i,"imp",j) from file FID Header"F54G";
(All,i,COM) TAX3F(i,"dom") from file FID Header"F21F";
(All,i,COM) TAX3F(i,"imp") from file FID Header"F22F";
(All,i,COM) TAX3G(i,"dom") from file FID Header"F21G";
(All,i,COM) TAX3G(i,"imp") from file FID Header"F22G";
TAX4F from file FID Header"F24F";
TAX4G from file FID Header"F24G";
(All,i,COM) TAX5F(i,"dom") from file FID Header"F27F";
(All,i,COM) TAX5F(i,"imp") from file FID Header"F28F";
(All,i,COM) TAX5G(i,"dom") from file FID Header"F27G";
(All,i,COM) TAX5G(i,"imp") from file FID Header"F28G";
TAXFGNGOV from file SRFM Header"0023";
TAX_K_RATE from file EXTRA Header"TAXK";
TAX_L_RATE from file EXTRA Header"TAXL";
TAX_L_RATE_L from file EXTRA4 Header"TAXL";
TAX_L_RATE_O from file EXTRA3 Header"TAXL";
TAX_L_R_O_L from file EXTRA5 Header"TAXL";
TREND_K from file EXTRA Header"GRSE";
TRNFGNAUS_B from file SRFM Header"TFAB";
TRNFGNHH from file SRFM Header"0021";
TRNOTHFGN from file SRFM Header"0025";
UNEMPBEN from file SRFM Header"UBEN";

VCAP_AT_T from file EXTRA Header"VCAP";
WAGE from file EXTRA Header"WAGE";
WP from file SRFM Header"0020";
WPT from file SRFM Header"0019";
WPTPLUS1 from file SRFM Header"0046";
YEAR from file EXTRA Header"YEAR";
YR_POLICY from file ROREXT Header"YRPL";

! 18.5 Formulas in thematic groups !

! 18.5a Default shares, binary dummies and miscellaneous scalar parameters !

Formula NUMCOM = 1.0/Sum(i,COM,1);
NUMIND = 1.0/Sum(j,IND,1);
NUMFAC = 1.0/Sum(f,FAC,1);
NUMOCC = 1.0/Sum(o,OCC,1);
NUMCOMP = 1.0/Sum(cc,COMPCOM,1);
SOURCEDOM("dom") = 1.0;
SOURCEDOM("imp") = 0.0;
(All,i,COM) TRANSERVDUM(i) = 0;
(All,i,TRANSERV) TRANSERVDUM(i) = 1;
(All,i,COM) COMMUNICDUM(i) = 0;
(All,i,COMMUNICAT) COMMUNICDUM(i) = 1;
TINY    = 0.000000000001;
EXP_ELAST_N = -4.0;
EXP_ELAST_T = -3.0;
(Initial)(All,i,COM) DUMMY_NT(i) = 0.0;
(Initial)(All,i,NTRADEXP) DUMMY_NT(i) = 1.0;

! 18.5b Correction to initial IO data to eliminate phantom taxes !

(Initial) AGGTAX4F = Sum(i,COM,TAX4F(i));
(Initial) (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX1F(i,s,j) = DUMF*TAX1F(i,s,j);
(Initial) (All,i,COM)(All,s,SOURCE)(All,j,IND)
TAX2F(i,s,j) = DUMF*TAX2F(i,s,j);
(Initial)(All,i,COM)(All,s,SOURCE) TAX3F(i,s) = DUMF*TAX3F(i,s);
(Initial)(All,i,COM) TAX4F(i) = DUMF*TAX4F(i);
(Initial)(All,i,COM)(All,s,SOURCE) TAX5F(i,s) = DUMF*TAX5F(i,s);
(Initial)(All,j,IND) TAX0F(j) = DUMF*TAX0F(j);
(Initial)(All,j,IND) OTHCOST(j) = Sum(i,COM,MAKE(i,j)) - TAX0G(j)-TAX0F(j)
    - {  Sum[i,COM,Sum{s,SOURCE,BAS1(i,s,j)+TAX1G(i,s,j) +TAX1F(i,s,j)
                          +Sum(r,MARGCOM,MAR1(i,s,j,r))}]
        + Sum(m,OCC,LABOCCIND(m,j)) + CAPITAL(j) +LAND(j)  };

! 18.5c  MAKE shares and joint production coefficients !

(All,i,COM) MAKE_SALES(i) = Sum(k,IND,MAKE(i,k));
(All,j,IND) MAKE_COSTS(j) = Sum(i,COM,MAKE(i,j));
Zerodivide Default NUMCOM;
Formula (All,i,COM)(All,j,IND) H0CI(i,j) = MAKE(i,j)/MAKE_COSTS(j);
Zerodivide off;
Zerodivide Default NUMIND;
Formula (All,i,COM)(All,j,IND) B0CI(i,j) = MAKE(i,j)/MAKE_SALES(i);
Zerodivide off;
Formula (All,i,COM_JP)(All,j,IND_JP) DENOM(i,j) =
         Sum(qq,COM_JP:MAKE(qq,j) ne 0,
   Sum(cc,COMPCOM:CCPROD(i,j,cc) ne 0 and CCPROD(qq,j,cc) ne 0, MAKE(qq,j)));
Zerodivide Default NUMCOMP;
Formula (All,i,COM_JP)(All,j,IND_JP) S0C(i,j) = MAKE(i,j)/DENOM(i,j);
Zerodivide off;
Formula (All,cc,COMPCOM)(All,j,IND_JP)
  H0CC(cc,j) = Sum(i,COM_JP,CCPROD(i,j,cc) *H0CI(i,j));
(All,j,IND_JP) SUMSIG(j) = Sum(t,COMPCOM,SIG0CC(t,j) *H0CC(t,j));
(All,cc,COMPCOM)(All,j,IND_JP) MH0CC(cc,j) = SIG0CC(cc,j)*H0CC(cc,j)/SUMSIG(j);

! 18.5d Tax flows and purchasers' values of commodity flows !

(All,i,COM)(All,s,SOURCE)(All,j,IND) TAX1(i,s,j) = TAX1G(i,s,j) + TAX1F(i,s,j);
(All,i,COM)(All,s,SOURCE)(All,j,IND) TAX2(i,s,j) = TAX2G(i,s,j) + TAX2F(i,s,j);
(All,i,COM)(All,s,SOURCE) TAX3(i,s) = TAX3G(i,s) + TAX3F(i,s);
(All,i,COM) TAX4(i) = TAX4G(i) + TAX4F(i);
(All,i,COM)(All,s,SOURCE) TAX5(i,s) = TAX5G(i,s) + TAX5F(i,s);
(All,j,IND) TAX0(j) = TAX0G(j) + TAX0F(j);
(All,j,IND) POW_TAX0(j) = 1 +  TAX0(j)/Sum(i,COM,MAKE(i,j));
(All,i,COM)(All,s,SOURCE)(All,j,IND) PURCHVAL1(i,s,j) =
                             BAS1(i,s,j) + TAX1(i,s,j) + Sum(r,MARGCOM,MAR1(i,s,j,r));
(All,i,COM)(All,s,SOURCE)(All,j,IND) PURCHVAL2(i,s,j) =
                              BAS2(i,s,j) + TAX2(i,s,j) + Sum(r,MARGCOM,MAR2(i,s,j,r));
(All,i,COM)(All,s,SOURCE)
  PURCHVAL3(i,s) = BAS3(i,s) + TAX3(i,s) + Sum(r,MARGCOM,MAR3(i,s,r));
(All,i,COM) PURCHVAL4(i) =
                                            BAS4(i) + TAX4(i) + Sum(r,MARGCOM,MAR4(i,r));
(All,i,COM)(All,s,SOURCE)
  PURCHVAL5(i,s) = BAS5(i,s) + TAX5(i,s) + Sum(r,MARGCOM,MAR5(i,s,r));
(All,i,COM)(All,j,IND) TPURCHVAL1(i,j) = Sum(s,SOURCE,PURCHVAL1(i,s,j));
(All,i,COM)(All,j,IND) TPURCHVAL2(i,j) = Sum(s,SOURCE,PURCHVAL2(i,s,j));
(All,i,COM) TPURCHVAL3(i) = Sum(s,SOURCE,PURCHVAL3(i,s));
(All,i,COM) TPURCHVAL5(i) = Sum(s,SOURCE,PURCHVAL5(i,s));



MONASH  represented in TABLO language 169

(All,i,COM) TTPURCHVAL(i) = Sum(j,IND, TPURCHVAL1(i,j) + TPURCHVAL2(i,j))
                  + TPURCHVAL3(i) + TPURCHVAL5(i) + Sum(s,SOURCE,BAS6(i,s));

! 18.5e Import/domestic shares !

Zerodivide Default 0.5;
Formula (All,i,COM)(All,s,SOURCE)(All,j,IND)
  SOURCE_SHR1(i,s,j) = PURCHVAL1(i,s,j)/ TPURCHVAL1(i,j);
(All,i,COM)(All,s,SOURCE)(All,j,IND)
  SOURCE_SHR2(i,s,j) = PURCHVAL2(i,s,j)/TPURCHVAL2(i,j);
(All,i,COM)(All,s,SOURCE) SOURCE_SHR3(i,s) = PURCHVAL3(i,s)/TPURCHVAL3(i);
(All,i,COM)(All,s,SOURCE) SOURCE_SHR5(i,s) = PURCHVAL5(i,s)/TPURCHVAL5(i);
Zerodivide off;

! 18.5f Import flows and tariffs !

Formula (All,i,COM) IMPORTS(i) = Sum(j,IND,BAS1(i,"imp",j) +
             BAS2(i,"imp",j)) + BAS3(i,"imp") + BAS5(i,"imp") + BAS6(i,"imp");
(All,i,COM) TARIFF(i) = (LEV_PTAXM(i)-1)*IMPORTS(i)/LEV_PTAXM(i);
(All,i,COM) IMPCOST(i) = IMPORTS(i) - TARIFF(i);
AGGTAXM = Sum(i,COM,TARIFF(i));
(Initial)(All,j,IND) DUMMYCAR(j) = 0.0;
(Initial)(All,j,MOTORIND) DUMMYCAR(j) = 1.0;

! 18.5g Tax aggregates  !

AGGTAX1 = Sum(i,COM,Sum(s,SOURCE,Sum(j,IND,TAX1(i,s,j))));
AGGTAX2 = Sum(i,COM,Sum(s,SOURCE,Sum(j,IND,TAX2(i,s,j))));
AGGTAX3 = Sum(i,COM,Sum(s,SOURCE,TAX3(i,s)));
AGGTAX4 = Sum(i,COM,TAX4(i));
AGGTAX5 = Sum(i,COM,Sum(s,SOURCE,TAX5(i,s)));
AGGTAX0 = Sum(j,IND,TAX0(j));
AGGTAX1G = Sum(i,COM,Sum(s,SOURCE,Sum(j,IND,TAX1G(i,s,j))));
AGGTAX2G = Sum(i,COM,Sum(s,SOURCE,Sum(j,IND,TAX2G(i,s,j))));
AGGTAX3G = Sum(i,COM,Sum(s,SOURCE,TAX3G(i,s)));
AGGTAX4G = Sum(i,COM,TAX4G(i));
AGGTAX5G = Sum(i,COM,Sum(s,SOURCE,TAX5G(i,s)));
AGGTAX0G = Sum(j,IND,TAX0G(j));
(All,i,COM) TAXF_COM(i) = Sum(j,IND,Sum(s,SOURCE,TAX1F(i,s,j) +
         TAX2F(i,s,j))) + TAX4F(i) + Sum(s,SOURCE, TAX3F(i,s) + TAX5F(i,s));
AGGTAX = AGGTAX1 + AGGTAX2 + AGGTAX3 + AGGTAX4 + AGGTAX5
                               +AGGTAX0 + AGGTAXM;
AGGTAXG = AGGTAX1G + AGGTAX2G + AGGTAX3G + AGGTAX4G
                                                         + AGGTAX5G + AGGTAX0G + AGGTAXM;

! 18.5h Expenditure-side of GDP and related aggregates !

(All,j,IND) VINVEST(j) = Sum(i,COM,TPURCHVAL2(i,j)) + TINY;
AGGCON = Sum(i,COM,TPURCHVAL3(i));
AGGINV = Sum(j,IND,VINVEST(j));
AGGOTH = Sum(i,COM,Sum(s,SOURCE,PURCHVAL5(i,s)));
AGGINVENT = Sum(i,COM,Sum(s,SOURCE,BAS6(i,s)));
AGGEXP = Sum(i,COM,PURCHVAL4(i));
AGGIMP = Sum(i,COM,IMPCOST(i));
GDPEX = AGGCON + AGGINV + AGGOTH + AGGINVENT + AGGEXP - AGGIMP;
AGGTRADEXP  = Sum(i,TRADEXP,PURCHVAL4(i));
AGGNTRADEXP = Sum(i,NTRADEXP,PURCHVAL4(i));
AGGTOURISM = Sum(i,TOURISM,PURCHVAL4(i));
(All,i,COM) TOT_T(i) = IMPCOST(i) + PURCHVAL4(i);

! 18.5i Primary factor aggregates and income-side GDP !

(All,j,IND) LABIND_J(j) = Sum(m,OCC,LABOCCIND(m,j));
(All,m,OCC)(All,j,IND)
  HOURS(m,j) = LABOCCIND(m,j)/[POW_PAYROLL(j)*WAGE(m,j)];
(All,m,OCC) LABOCC(m) = Sum(j,IND,HOURS(m,j));
HOURSTOT = Sum(m,OCC,Sum(j,IND, HOURS(m,j)));
(All,j,IND) TOTFACIND(j) = LABIND_J(j) + CAPITAL(j) + LAND(j);
(All,j,IND) VALUEADD(j) = TOTFACIND(j)+OTHCOST(j) +TAX0G(j);
(All,i,COM) TOTFACCOM(i) = Sum(j,IND,H0CI(i,j)*TOTFACIND(j));
AGGLAB = Sum(j,IND,LABIND_J(j));
AGGCAP = Sum(j,IND,CAPITAL(j));
AGGLND = Sum(j,IND,LAND(j));
AGGOCT = Sum(j,IND,OTHCOST(j));
TOTFAC = AGGLAB + AGGCAP + AGGLND;
GDPIN = TOTFAC + AGGOCT + AGGTAX;

! 18.5j Total margins, sales and costs, and checks on IO balances !

(All,r,MARGCOM) TOTMARGINS(r) = Sum(i,COM, MAR4(i,r) +
      Sum(s,SOURCE,MAR3(i,s,r) + MAR5(i,s,r)
          + Sum(j,IND, MAR1(i,s,j,r) + MAR2(i,s,j,r))));
(All,i,NONMARGCOM) TOTMARGINS(i) = 0.0;
(All,i,COM) SALES(i) = Sum(j,IND,BAS1(i,"dom",j) + BAS2(i,"dom",j))
+ BAS3(i,"dom") + BAS4(i) + BAS5(i,"dom") + BAS6(i,"dom") + TOTMARGINS(i);
(All,j,IND) COSTS(j) = Sum(i,COM, TPURCHVAL1(i,j)) + TOTFACIND(j)
                                                   + OTHCOST(j)  + TAX0(j);
(All,j,IND) PURE_PROFITS(j) = Sum(i,COM,B0CI(i,j) *SALES(i)) - COSTS(j);
(All,i,COM) LOST_GOODS(i)   = Sum(j,IND,H0CI(i,j) *COSTS(j)) - SALES(i);

! 18.5k Coefficients for linear expenditure system !

(Initial) SUMDELTA = Sum(i,COM, DELTA(i));
(Initial)(All,i,COM) DELTA(i) = DELTA(i)/SUMDELTA;
(All,i,COM) S3COM(i) = TPURCHVAL3(i)/AGGCON;
Zerodivide Default 1.0;
Formula (All,i,COM) EPS(i) = DELTA(i)/S3COM(i);
Zerodivide Off;
Formula (All,i,COM)
SS3COM(i) = S3COM(i) + DELTA(i)/FRISCH;
(All,i,COM)(All,j,COM) KD(i,j) = 0.0;
(All,i,COM) KD(i,i) = 1.0;
(All,i,COM)(All,j,COM)
  ETA(i,j) = KD(i,j)*EPS(i)/FRISCH - EPS(i)*[S3COM(j) + DELTA(j)/FRISCH];

! 18.5l Other cost tickets !

(All,j,IND) ROTHCOST(j) = OTHCOST(j)/P1OCT(j);
(All,j,IND)LEV_F1OCT(j) = P1OCT(j)/LEV_CPI;

! 18.5m  CRESH factor demands !

(All,j,IND) CRESHSUM1(j) = SIGMA1LAB(j)*LABIND_J(j)
              + SIGMA1CAP(j)*CAPITAL(j) + SIGMA1LND(j)*LAND(j);
Zerodivide Default NUMFAC;
Formula (All,j,IND)
  SSTAR1FAC("labour",j) = SIGMA1LAB(j)*LABIND_J(j)/CRESHSUM1(j);
(All,j,IND) SSTAR1FAC("capital",j)= SIGMA1CAP(j)*CAPITAL(j)/CRESHSUM1(j);
(All,j,IND) SSTAR1FAC("land",j) = SIGMA1LND(j)*LAND(j)/CRESHSUM1(j);
Zerodivide Off;
Formula (All,j,IND)
SOURCE_SHRLK("labour",j) = LABIND_J(j)/[LABIND_J(j) + CAPITAL(j)];
(All,j,IND) SOURCE_SHRLK("capital",j) = CAPITAL(j) /[LABIND_J(j) + CAPITAL(j)];
(All,j,IND) SOURCE_SHRLK("land",j) = 0.0;
(All,j,IND) CRESHSUM2(j) = Sum(m,OCC, SIGMA1LABOCC(m,j)*LABOCCIND(m,j));
Zerodivide Default NUMOCC;
Formula (All,m,OCC)(All,j,IND)
SSTAR1LABOCC(m,j) = SIGMA1LABOCC(m,j)*LABOCCIND(m,j)/CRESHSUM2(j);
Zerodivide Off;

! 18.5n Mapping between consumption of MONASH and national accounts
  commodities !
Formula (Initial)(All,i,COM) MMI(i) = Sum(na,NCOM,MM(i,na));
(Initial)(All,i,COM)(All,na,NCOM) MM(i,na) = MM(i,na) + If(MMI(i) EQ 0, TINY);
(Initial)(All,i,COM) MMI(i) = Sum(na,NCOM,MM(i,na));

(Initial)(All,i,COM)(All,na,NCOM)
                                   MM(i,na) = MM(i,na)*{(TINY + TPURCHVAL3(i))/MMI(i)};
(Initial)(All,i,COM) MMI(i) = Sum(na,NCOM, MM(i,na));
(Initial)(All,na,NCOM) MMN(na) = Sum(i,COM, MM(i,na)) + TINY;

! 18.5o Calculations of coefficients used in special treatments of imports !

(All,j,IND) IMPORTS_IND(j) = Sum(i,COM, BAS1(i,"imp",j)) + TINY;
(All,j,IND) EXPORTS_IND(j) =
                        Sum(i,COM:MAKE(i,j)>0,MAKE(i,j)*BAS4(i)/SALES(i)) + TINY;
(All,j,IND) C_SPECIMP1(j) =
    MAKE_COSTS(j)/[MAKE_COSTS(j)+EXPORTS_IND(j)+IMPORTS_IND(j)];
(All,j,IND) C_SPECIMP2(j) =
    IMPORTS_IND(j)/[MAKE_COSTS(j)+EXPORTS_IND(j)+IMPORTS_IND(j)];
(All,j,IND) C_SPECIMP3(j) =
    EXPORTS_IND(j)/[MAKE_COSTS(j)+EXPORTS_IND(j)+IMPORTS_IND(j)];
(Initial)(All,j,IND) DUMMY_WAT(j) =0.0;
(Initial)(All,j,WATAIRTRN) DUMMY_WAT(j) =1.0;

! 18.5p Calculations of coefficients for special forecasting equations !

(Initial)(All,i,NTRADEXP) DUMMY_HC(i) = {EXP{58.889*(BAS4(i)/SALES(i)-0.05)}
                                       / [1+EXP{58.889*(BAS4(i)/SALES(i)-0.05)}] -0.05}/0.95;
(All,i,COM) AVIMPSH(i) = { Sum(j,IND, SOURCE_SHR1(i,"imp",j)*BAS1(i,"dom",j)
  + SOURCE_SHR2(i,"imp",j)*BAS2(i,"dom",j))
  + SOURCE_SHR3(i,"imp")*BAS3(i,"dom")
  + SOURCE_SHR5(i,"imp")*BAS5(i,"dom") } / { Sum(j,IND, BAS1(i,"dom",j)
                                        + BAS2(i,"dom",j))+ BAS3(i,"dom") + BAS5(i,"dom") + TINY };
(All,i,COM) DUM_TW(i) =
                              IF( AVIMPSH(i) <= 0.1, 10*AVIMPSH(i))+IF( AVIMPSH(i) > 0.1, 1);

! 18.5q Capital quantities and prices, and investment quantities !

(Initial)(All,j,IND) VCAP_AT_T(j) = VCAP_AT_T(j) + TINY;
(All,j,IND) VCAP_AT_TM(j) = VCAP_AT_T(j)*PCAP_J(j)/PCAP_AT_T(j);
(All,j,IND) QCAPATT(j) = VCAP_AT_T(j)/PCAP_AT_T(j);
(All,j,IND) QINVEST(j) = VINVEST(j)/PCAP_J(j);
(Initial)(All,j,IND)
  QCAPATT_B(j) = DUMMY_DEC*QCAPATT(j) + (1-DUMMY_DEC)*QCAPATT_B(j);
(Initial)(All,j,IND) QINV_BASE(j) = QINVEST(j);
(All,j,IND) QCAPATTPLUS1(j) = QCAPATT(j)*(1 - DEP(j)) + QINVEST(j);
(Initial)(All,j,IND) QCAPPLUS1_B(j) = QCAPATTPLUS1(j);
(All,j,IND) K_GR(j) = [QCAPATTPLUS1(j)/QCAPATT(j)] - 1;
(Initial)(All,j,IND) PCAP_AT_T_B(j) = PCAP_AT_T(j);
(Initial)(All,j,IND) PCAP_J_L_B(j) = PCAP_J_L(j);
(Initial)(All,j,IND) PCAP_AT_T1_B(j)  = PCAP_AT_T1(j);

(Initial)(All,j,IND)
  PCAP_J_B(j) = DUMMY_DEC*PCAP_J(j) + (1-DUMMY_DEC)*PCAP_J_B(j);

! 18.5r The inverse logistic relationships between expected rates of return
 and rates of capital growth !

(Initial)(All,j,IND) K_GR_MIN(j) = - DEP(j);
(Initial)(All,j,IND) K_GR_MAX(j) =
                                    TREND_K(j) + DIFF + If(QCAPATT(j) <= 0.00001, 1.0);
(Initial)(All,j,IND) COEFF_SL(j) =SMURF*(K_GR_MAX(j)-K_GR_MIN(j))
                          /[(K_GR_MAX(j)-TREND_K(j))*(TREND_K(j)-K_GR_MIN(j))];
(All,j,IND) CHKGR1(j)= 0.0 + If(K_GR_MIN(j) >= K_GR(j),1.0);
(All,j,IND) CHKGR2(j)= 0.0 + If(K_GR_MAX(j) <= K_GR(j),1.0);
(All,j,IND) K_GR(j) = K_GR(j)
                + If(K_GR_MIN(j) >= K_GR(j), K_GR_MIN(j) - K_GR(j) + 0.005)
                + If(K_GR_MAX(j) <= K_GR(j),K_GR_MAX(j) - K_GR(j) - 0.005);
(All,j,IND) EEQROR(j)=RORN(j) + F_EEQROR + F_EEQROR_J(j)
+ (1/COEFF_SL(j))*{ [ Loge(K_GR(j)-K_GR_MIN(j)) - Loge(K_GR_MAX(j) - K_GR(j))]
   - [Loge(TREND_K(j)-K_GR_MIN(j)) - Loge(K_GR_MAX(j) - TREND_K(j))] };

! 18.5s The expected rate of return under static expectations !

(Initial) LEV_CPI_B = LEV_CPI;
(Initial) LEV_CPI_L_B = LEV_CPI_L;
INF = LEV_CPI/LEV_CPI_L -1;
INT = (1 + RINT)*(1 + INF) - 1;
INT_PSD = (1 + RINT_PSD)*(1 + INF) - 1;
RINT_PT_SE = (1+INT*(1-TAX_K_RATE))/(1+ INF) -1;
(All,j,IND) ROR_SE(j) = (1/[1 + RINT_PT_SE])*
  {  [CAPITAL(j)*(1 - TAX_K_RATE)]/VCAP_AT_TM(j)
             + (1 - DEP(j)) + RALPH*TAX_K_RATE*DEP(j)  } - 1;
(Initial)(All,j,IND) ROR_SE_BASE(j) =
       DUMMY_DEC*ROR_SE(j) + (1-DUMMY_DEC)*ROR_SE_BASE(j);
(Initial)  AV_ROR_SE_B = DUMMY_DEC
*Sum(j,IND,ROR_SE_BASE(j)*VCAP_AT_TM(j))/Sum(j,IND,VCAP_AT_TM(j))
                                                      + (1-DUMMY_DEC)*AV_ROR_SE_B;

! 18.5t Expected and actual rates of return, and the algorithm for
  imposing forward-looking expectations !

(Initial) YEAR_B = YEAR;
(Initial) ITER_NUM_B = ITER_NUM;
(Initial) ONE_ITER1 =  If(ITER_NUM_B =1, 1);
(Initial) ONE_IT1_REP =  If(ITER_NUM_B =NOFITERS+2, 1);
(Initial) ZERO_PYR1 =
                   1 + If(ITER_NUM_B >= NOFITERS+2 and YEAR_B =YR_POLICY,-1);

(Initial) ITER_ADJUST = 0 + If(YEAR_B = NYEARS , 1);
(Initial)(All,t,TIME) DUM_TIME(t) = 0 + If(YEAR_B = COEFF_TIME(t) ,1);
(Initial)(All,t,TIME) DUM_TIME_LAG(t) =
                                                                0 + If(YEAR_B = COEFF_TIME(t) + 1 ,1);
(Initial) DUM_IT1 =
                             0 + If(ITER_NUM_B=1 OR ITER_NUM_B = NOFITERS+2, 1);
(Initial) COEFF_NYEAR  = 0 + If(YEAR_B < NYEARS,1);
INF_L = LEV_CPI_L/LEV_CPI_2L -1;
INT_L = (1+INF_L)*(1+RINT_L) -1;
(Initial) RINT_B  = RINT;
(Initial) RINT_L_B  = RINT_L;
(Initial) LEV_CPI_2L_B = LEV_CPI_2L;
(All,j,IND) ROR_ACT_L(j) ={1/[PCAP_J_L(j) *(1+INT_L*(1-TAX_K_RATE))]}*
     [   (1-TAX_K_RATE)*(CAPITAL(j)/QCAPATT(j))
         + (1-DEP(j))*PCAP_J(j) + RALPH*TAX_K_RATE*DEP(j)*PCAP_J(j)  ] -1;
(Initial)(All,j,IND) ROR_ACT_L_B(j) =ROR_ACT_L(j);
(Initial) (All,j,IND)(All,t,TIME) EROR_G_B(j,t) =EROR_G(j,t)
  + ZERO_PYR1*DUM_IT1*DUM_TIME_LAG(t)*(ROR_ACT_L_B(j)-EROR_G(j,t));
(Initial)(All,j,IND) EROR_F(j) =  Sum(t,TIME,DUM_TIME(t)*EROR_G_B(j,t));
(Initial)(All,j,IND) EROR_B(j) = Sum(t,TIME,DUM_TIME_LAG(t)*EROR_G_B(j,t));
(Initial)(All,j,IND) DISEQRE_B(j) = EROR_B(j) - EEQROR(j);
(Initial)(All,j,IND) DISEQSE_B(j) = ROR_SE_BASE(j) - EEQROR(j);
(Initial)(All,j,IND)(All,t,TIME) EROR_G(j,t) = EROR_G_B(j,t) + ZERO_PYR1
     *{If[DUM_TIME_LAG(t) NE 0, ADJ_RE(j)*(ROR_ACT_L_B(j)-EROR_G_B(j,t))]};

! 18.5u Creation of useful coefficients for policy simulations !

(Initial) RWAGE_B = RWAGE;
(Initial) RWAGE_OLD_B = RWAGE_OLD;
RWAGE_PT = RWAGE*(1 - TAX_L_RATE);
(Initial) RWAGE_PT_B = RWAGE_PT;
RWAGE_PT_OLD = RWAGE_OLD*(1 - TAX_L_RATE_O);
(Initial) RWAGE_PT_O_B = RWAGE_PT_OLD;
(Initial) RWAGE_PT_L_B   = RWAGE_L_B*(1 - TAX_L_RATE_L);
(Initial) RW_PT_O_L_B = RWAGE_O_L_B*(1 - TAX_L_R_O_L);
(Initial) EMPLOY_B =EMPLOY;
(Initial) EMPLOY_O_B = EMPLOY_OLD;
(Initial) FEMPADJ_B = FEMPADJ;
(Initial)(All,i,COM)(All,s,SOURCE) LEVX3B(i,s) =LEVX3(i,s);
(Initial)(All,i,COM)(All,s,SOURCE) LEVX3B_O(i,s) = LEVX3_O(i,s);
(Initial)(All,i,COM)(All,s,SOURCE) B3SH(i,s) =
  (1-DUM_YEAR1)*{PURCHVAL3(i,s)/AGGCON} + DUM_YEAR1*B3SH(i,s);
(Initial)(All,i,COM)(All,s,SOURCE) B3SHO(i,s) =

 (1-DUM_YEAR1)*{PURCHVAL3(i,s)/AGGCON} + DUM_YEAR1*B3SHO(i,s);
(Initial)(All,i,COM)(All,s,SOURCE) B3SHB(i,s) = B3SH(i,s);
(Initial)(All,i,COM)(All,s,SOURCE) B3SHBO(i,s) = B3SHO(i,s);
(All,j,IND) COL_PAYROLL(j) =
                              (LABIND_J(j)/POW_PAYROLL(j))*(POW_PAYROLL(j) -1);
COL_PAYRTOT = Sum(j,IND,COL_PAYROLL(j));
TAX_LAB = TAX_L_RATE*(AGGLAB-COL_PAYRTOT);
TAX_CAP = TAX_K_RATE*AGGCAP;
TAX_LND = TAX_K_RATE*AGGLND;
INCTAX = TAX_LAB + TAX_CAP + TAX_LND;
(Initial) ADJDUMYEAR1 = 0.0 + If( DUM_YEAR1 LT 0.1, 1.0);

! 18.5v Calculations of coefficients used in government accounts !

AGGINVG =  Sum(j,IND,G_VINVEST(j));
NET_TAXTOTG = AGGTAXG + INCTAX + COL_PAYRTOT;
PSDATTPLUS1 ={  PSDATT*(1+INT_PSD/2) + AGGOTH + AGGINVG
            + OTHCAPGOV - NET_TAXTOTG - OTHGOVREV + UNEMPBEN
            + AGEBEN + OTHBEN + GRANT   }/(1-INT_PSD/2);
NETINT_G = INT_PSD*(PSDATT+PSDATTPLUS1)/2;
TRANS = UNEMPBEN + AGEBEN + OTHBEN + GRANT + NETINT_G;
GOV_DEF = AGGOTH +  AGGINVG + OTHCAPGOV - NET_TAXTOTG
                                                       - OTHGOVREV + TRANS;
(Initial) R_DEFGDP_B = GOV_DEF/GDPEX;
(Initial) PSDATT_1_B = PSDATTPLUS1;
(Initial) PSDATT_B = PSDATT;
R_PSDGDP = PSDATT/GDPEX;
(Initial) R_PSDGDP_B = R_PSDGDP;

! 18.5w Calculation of GNP, net foreign liabilities and related coefficients used
mainly in historical/decomposition simulations !

FA_T= FL_T - NET_FLT;
NET_FLTF= NET_FLT*EXCH_T;
(Initial) NET_FLTF_B =
             DUMMY_DEC*NET_FLTF +(1-DUMMY_DEC)*NET_FLTF_B;
GNP = GDPEX - ROIFOREIGN*NET_FLTF/EXCH;
(Initial) GNP_B = DUMMY_DEC*GNP+(1-DUMMY_DEC)*GNP_B;
APCGNP = (AGGCON+AGGOTH)/GNP;
APSGNP = 1 - APCGNP;
(Initial) APSGNP_B = DUMMY_DEC*APSGNP+(1-DUMMY_DEC)*APSGNP_B;
TRNFGNAUS = TRNFGNHH  + TAXFGNGOV  - TRNOTHFGN;
(Initial) TRNFGNAUS_B =
                          DUMMY_DEC*TRNFGNAUS+(1-DUMMY_DEC)*TRNFGNAUS_B;
(All,s,COUNTSET) COUNT(s) = $POS(s)-1;

(All,s,COUNTSET) DUM_COUNT(s) = 0.0 + IF($POS(s) < TAU+0.5 ,1.0);
(Initial) EXCH_B = DUMMY_DEC*EXCH +(1-DUMMY_DEC)*EXCH_B;
AGGVCAP_T = Sum(j,IND, VCAP_AT_T(j));
WEALTH_T = AGGVCAP_T - NET_FLT;
(Initial) RWEAL_T =
                         (1-DUM_YEAR1)*WEALTH_T/LEV_XI2T+DUM_YEAR1*RWEAL_T;
(Initial) RWEAL_T_B = RWEAL_T;
(Initial) RWEAL_T_O =
                                     (1-DUM_YEAR1)*RWEAL_T_B + DUM_YEAR1*RWEAL_T_O;

! 18.5x Calculation of coefficients used in balance of payments and related
equations !

FDFCATT = SHT_FDFC*FL_T;
FDFCATT_1 = SHT1_FDFC*FL_T1;
FDDCATT = SHT_FDDC*FL_T;
FDDCATT_1 = SHT1_FDDC*FL_T1;
FE_A_T = FL_T - FDFCATT - FDDCATT;
FE_A_T_1 = FL_T1 - FDFCATT_1 - FDDCATT_1;
(Initial) FDFCATT_B = FDFCATT;
(Initial) FDDCATT_B = FDDCATT;
(Initial) FDFCATT_1_B = FDFCATT_1;
(Initial) FDDCATT_1_B = FDDCATT_1;
(All,j,IND) VCAP_AT_T1(j) =
                    VCAP_AT_T(j)*(1-DEP(j))*PCAP_AT_T1(j)/PCAP_AT_T(j)
                         + VINVEST(j)*PCAP_AT_T1(j)/PCAP_J(j);
(Initial)(All,j,IND) FE_A_T_J(j) = SHFEA_T_J(j)*VCAP_AT_T(j);
(Initial) FE_A_T_ADD = Sum(j,IND,FE_A_T_J(j));
(Initial)(All,j,IND) SHFEA_T_J(j) = (FE_A_T/FE_A_T_ADD)*SHFEA_T_J(j);
(All,j,IND) FE_A_T_J(j) = SHFEA_T_J(j)*VCAP_AT_T(j);
(Initial)(All,j,IND) FE_A_T1_J(j) = SHFEA_T1_J(j)*VCAP_AT_T1(j);
(Initial) FE_A_T1_ADD = Sum(j,IND,FE_A_T1_J(j));
(Initial)(All,j,IND) SHFEA_T1_J(j) = (FE_A_T_1/FE_A_T1_ADD)*SHFEA_T1_J(j);
(All,j,IND) FE_A_T1_J(j) = SHFEA_T1_J(j)*VCAP_AT_T1(j);
(Initial)(All,j,IND) FE_A_T_J_B(j) = FE_A_T_J(j);
(Initial)(All,j,IND) FE_A_T1_J_B(j) =FE_A_T1_J(j);
(Initial)(All,j,IND) DIV_FE_A_J(j) =  CAPITAL(j)*
  (1-TAX_K_RATE)*(FE_A_T_J(j)/VCAP_AT_T(j))*FGN_SHD_J(j)*FGN_SHD;
(Initial) DIV_FEA_ADD =Sum(j,IND,DIV_FE_A_J(j));
(Initial) FGN_SHD = (DIV_FE_A/DIV_FEA_ADD)*FGN_SHD;
(All,j,IND) DIV_FE_A_J(j) =  CAPITAL(j)*
  (1-TAX_K_RATE)*(FE_A_T_J(j)/VCAP_AT_T(j))*FGN_SHD_J(j)*FGN_SHD;
AGGCAPF = Sum(j,IND, SHFEA_T_J(j)*CAPITAL(j));

NEWFDFC ={FDFCATT_1 - FDFCATT*(EXCH_T/EXCH_T_1)}*(EXCH_T_1/EXCH);
NEWFDDC =FDDCATT_1 - FDDCATT;
(All,j,IND) NEW_FE_A_J(j) =
{FE_A_T1_J(j) - FE_A_T_J(j)*(1-DEP(j))*(PCAP_AT_T1(j)/PCAP_AT_T(j))}*
                                                                                 (PCAP_J(j)/PCAP_AT_T1(j));
NEW_FE_A = Sum(j,IND, NEW_FE_A_J(j));
FCFCATT = SHT_FCFC*FA_T;
FCFCATT_1 = SHT1_FCFC*FA_T1;
FCDCATT = SHT_FCDC*FA_T;
FCDCATT_1 = SHT1_FCDC*FA_T1;
AE_F_T = FA_T - FCFCATT - FCDCATT;
AE_F_T_1 = FA_T1 - FCFCATT_1 - FCDCATT_1;
(Initial) FCFCATT_B = FCFCATT;
(Initial) FCDCATT_B = FCDCATT;
(Initial) AE_F_T_B = AE_F_T;
(Initial) FCFCATT_1_B = FCFCATT_1;
(Initial) FCDCATT_1_B = FCDCATT_1;
(Initial) AE_F_T_1_B = AE_F_T_1;
NEWFCFC =
    {FCFCATT_1 - FCFCATT*(EXCH_T/EXCH_T_1)}*(EXCH_T_1/EXCH);
NEWFCDC =FCDCATT_1 - FCDCATT;
NEW_AE_F = {AE_F_T_1 - AE_F_T*(WPTPLUS1/WPT)*(EXCH_T/EXCH_T_1)}
                                                       /{(WPTPLUS1/WP)*(EXCH/EXCH_T_1)};
(Initial) CADEF = (1-DUM_YEAR1)*{AGGIMP - (AGGEXP + [1-DUMF]*AGGTAX4F)
  + INTFDFC + INTFDDC - INTFCFC -INTFCDC +DIV_FE_A - DIV_AE_F
   - TRNFGNAUS } + DUM_YEAR1*CADEF;
BAL_ITEM = CADEF - (NEWFDFC + NEWFDDC - NEWFCFC – NEWFCDC
                                                      + NEW_FE_A - NEW_AE_F);
R_CADGDP = CADEF/GDPEX;
(Initial) R_CADGDP_B = R_CADGDP;
NFLATT = FDFCATT + FDDCATT - FCFCATT - FCDCATT + FE_A_T  - AE_F_T;
R_NFL_VCAP = NFLATT/Sum(j,IND,VCAP_AT_T(j));
R_NFL_GDP = NFLATT/GDPEX;
(Initial) R_NFL_VCAP_B = R_NFL_VCAP;
(Initial) R_NFL_GDP_B = R_NFL_GDP;
(Initial) EXCH_T_1_B = EXCH_T_1;
(Initial) EXCH_T_B =EXCH_T;
(Initial) WPT_1_B = WPTPLUS1;
(Initial) WPT_B =WPT;
(Initial) WP_B =WP;

! 18.5y Coefficients related to household income and saving !

HOUS_DIS_INC = GDPEX + TRANS - NET_TAXTOTG - OTHGOVREV
  + TAXFGNGOV + TRNFGNHH + INTFCFC + INTFCDC + DIV_AE_F
  - INTFDFC - INTFDDC - DIV_FE_A;
HOUS_SAV = HOUS_DIS_INC - AGGCON;
NAT_SAV = HOUS_SAV - GOV_DEF +  AGGINVG + OTHCAPGOV;
GOV_SAV = NAT_SAV - HOUS_SAV;
AV_PROP_CON = AGGCON/HOUS_DIS_INC;

! 18.6 Variable declarations in alphabetical order !

Variable
        (All,j,IND)
a(j)  # Average of technical change terms, production #;
        (All,cc,COMPCOM)(All,j,IND_JP)
a0ccom(cc,j) # Composite-good-augmenting technical change in industry j #;
        (All,i,COM_JP)(All,j,IND_JP)
a0ci(i,j) # Technical change - commodity i produced by industry j #;
        (All,i,COM)
a0com(i) # All-industry commodity-specific technical change #;
        (All,j,IND)
a0ind(j) # Output augmenting technical change #;
        (All,j,IND)
a1(j) # All-input augmenting technical change #;
        (All,j,IND)
a1cap(j) # Capital augmenting technical change #;
         (All,i,COM)(All,j,IND)
a1ci(i,j) # Input-i-saving technical change in current production #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
a1csi(i,s,j) # Input-(i,s)-saving technical change in current production #;
        (All,j,IND)
a1lab(j) # Labor augmenting technical change #;
a1labgen # All-industry labor augmenting technical change #;
        (All,j,IND)(All,m,OCC)
a1laboi(m,j) # Labor-by-skill-type m augmenting technical change in ind.j #;
        (All,j,IND)
a1land(j) # Land augmenting technical change #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
a1marg(i,s,j,r) # Tech. change in margin use, deliveries to current prod. #;
        (All,j,IND)
a1oct(j) # "Other cost" ticket augmenting techncal change #;
        (All,j,IND)
a1prim(j) # All-factor augmenting technical change #;

a1primgen # All-industry all-factor augmenting technical change #;
        (All,i,COM)(All,j,IND)
a2ci(i,j) # Input-i-saving technical change in investment #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
a2csi(i,s,j) # Input-(i,s)-saving technical change in capital creation #;
        (All,j,IND)
a2ind(j) # Neutral technical change - capital creation #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
a2marg(i,s,j,r) # Technical change in margins use, deliveries to investment #;
        (All,i,COM)
a3com(i) # Combined change in household tastes #;
        (All,i,COM)(All,s,SOURCE)
a3cs(i,s) # Commodity-(i,s)-augmenting change in household preferences #;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
a3marg(i,s,r) # Technical change in margins use, deliveries to consumption #;
        (All,na,NCOM)
a3ncom(na) # Taste change, national accounts commodities #;
        (All,i,COM)
a3shift(i) # Consumption shifter for MONASH commodities #;
        (All,i,COM)(All,r,MARGCOM)
a4marg(i,r) # Technical change in margins use, deliveries to export #;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
a5marg(i,s,r) # Technical change in margins use, deliveries to government #;
        (All,i,COM)
aa3com(i) # Cont. of change in hhld pref for i to sales of domestic i #;
        (All,i,COM)
ac(i) # Commodity-using technical and taste change #;
        (All,i,COM)
ac12mar_tot(i) # Average i-aug tech change in inter, capital & marg uses #;
        (All,i,COM)
ac1_tot(i) # Average i-augmenting tech change in intermediate usage #;
        (All,i,COM)
ac2_tot(i) # Average i-augmenting tech change in capital formation #;
        (All,i,COM)
adj(i) # Adjusts quantities to reconcile demands and supplies #;
        (All,abs,ABS_COM)
adj_abs(abs) # Adjusts supplies of ABS coms. to reconcile demands & supplies #;
aef_t  # $A value of f'gn equity held by Australians, start of year #;
aef_t1 # Australian equity in f'gn countries, end of year #;
age_ben # Age pensions #;
agginv_g # Value of public sector investment #;
agginv_rg # Real public investment #;

aggvcap # Aggregate value of capital #;
agg_tour # Quantity of tourism exports, often set by BTR forecasts #;
apc # Average propensity to consume #;
apc_gnp # Ratio of consumption (public and private) to GNP #;
aps_gnp # Average propensity to save out of GNP #;
        (All,t,ABS_COM)
aq(t) # t-saving tech change: t is a commodity in the ABS historical class'n #;
        (All,j,IND_JP)
avea0ci_j(j) # Average output-contracting tech change in industry j #;
ave_aprim # Average primary-factor-saving technical change across industries #;
ave_a3com # Average value of a3com #;
ave_wage # Average nominal wage rate, excludes payroll tax #;
a_gdp # Contribution of technical change to GDP growth #;
      (All,i,COM)(All,s,SOURCE)
b3shr(i,s) # Share of (i,s) in consumer budget #;
      (All,i,COM)(All,s,SOURCE)
b3sh_o(i,s) # Share of (i,s) in consumer budget in forecast simulation #;
c # Nominal total household consumption #;
        (All,j,IND)
capprod(j) # Capital productivity in industries #;
caprev # Aggregate payments to capital #;
        (All,j,IND)
cap_at_t(j)  # Capital stock at t (start of forecast year) #;
        (All,j,IND)
cap_at_tplus1(j) # Capital stock at t+1 (end of forecast year) #;
        (All,s,INDSAGGTO26)
cap_at_t_s(s) # Current capital stocks - sector #;
        (All,j,IND)
ch_kgr1(j) # Values of CHKGR1: checks if K_GR_MIN(j) >= K_GR(j) #;
        (All,j,IND)
ch_kgr2(j) # Values of CHKGR2: checks if K_GR_MAX(j) <= K_GR(j) #;
cr # Real household consumption #;
cv_under # Lower bound on compensating variation as % of h'hold expenditure #;
        (All,i,COM)
deltapc(i) # Change in household tastes - luxury part #;
        (Change)
del_b # Ordinary change in balance of trade, in $A billion #;
        (Change)
del_bt_gdp # Change in BT/GDP ratio (ratio is number like -2.0 pc) #;
        (Change)(All,j,IND)
del_f1oct(j) # Change in shift in price of other cost tickets #;

        (Change)(All,j,IND)
del_ff1oct(j) # Shifter for del_f1oct #;
        (Change)(All,j,IND)
del_ff_rate(j) # Shifter in change form of ror-capital-growth equation #;
        (Change)(All,qq,INDSAGGTO20)
del_ff_rate_s(qq) # shift in rate of return for sector qq, 20 sectors #;
        (Change)(All,j,IND)
del_f_a1(j) # Allows for cost-changing, com-saving tech. change, curr. prod. #;
        (Change)(All,j,IND)
del_f_a2(j) # Allows for cost-changing, com-saving tech. change, investment #;
        (Change)(All,j,IND)
del_f_ac_p_y(j) # Shifter in cap. accum. eq'n for year t-1 in yr-to-yr f'cast #;
        (Change)(All,j,IND)
del_f_rate(j) # Allows shifts in j's ror away from the economy-wide shift #;
        (Change)
del_f_wage_c # Shift in labour supply, pre-tax #;
        (Change)
del_f_wage_pt # Shifter in post-tax stick-wage equation #;
        (Change)(All,j,IND)
del_k_gr(j) # Capital growth thru forecast year #;
        (Change)(All,j,IND)
del_p1oct(j) # Change in price of other cost tickets #;
        (Change)
del_r # Makes rors sensitive to base rors in hist./decomp sims: often 0.5 #;
        (Change)(All,j,IND)
del_ror_se(j) # Percentage point changes in rates of return: static expect. #;
        (Change)(All,j,IND)
del_ror_se_o(j) # Static exp. rate of return for industry j in forecast #;
        (Change)
del_r_tot # Uniform shifter in changes in rates of return #;
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
del_tax1(i,s,j) # Revenue from taxes on intermediate input flows #;
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
del_tax1g(i,s,j) # Revenue from genuine taxes on intermediate input flows #;
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
del_tax1ph(i,s,j) # Revenue from phantom taxes on intermediate input flows #;
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
del_tax2(i,s,j) # Revenue from taxes on investmet flows #;
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
del_tax2g(i,s,j) # Revenue from genuine taxes on investment flows #;
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
del_tax2ph(i,s,j) # Revenue from phantom taxes on investment flows #;

        (Change)(All,i,COM)(All,s,SOURCE)
del_tax3(i,s) # Revenue from taxes on flows to households #;
        (Change)(All,i,COM)(All,s,SOURCE)
del_tax3g(i,s) # Revenue from genuine taxes on flows to households #;
        (Change)(All,i,COM)(All,s,SOURCE)
del_tax3ph(i,s) # Revenue from phantom taxes on flows to households #;
        (Change)(All,i,COM)
del_tax4(i) # Change in collection of taxes on exports #;
        (Change)(All,i,COM)
del_tax4g(i) # Change in revenue from genuine indirect taxes on exports #;
        (Change)(All,i,COM)
del_tax4ph(i) # Change in collection of phantom taxes on exports #;
        (Change)(All,i,COM)(All,s,SOURCE)
del_tax5(i,s) # Revenue from taxes on flows to government #;
        (Change)(All,i,COM)(All,s,SOURCE)
del_tax5g(i,s) # Revenue from genuine taxes on flows to government #;
        (Change)(All,i,COM)(All,s,SOURCE)
del_tax5ph(i,s) # Revenue from phantom taxes on flows to government #;
        (Change)(All,j,IND)
del_tax0(j) # Revenue from production taxes #;
        (Change)(All,j,IND)
del_tax0ph(j) # Revenue from phantom production taxes #;
        (Change)(All,j,IND)
del_tax0g(j) # Rev from genuine production taxes #;
        (Change)(All,i,COM)
del_taxm(i) # Change in collection of taxes on imports #;
        (Change)
del_tot_tax4 # Change in aggregate revenue from indirect taxes on exports #;
        (Change)
del_tot_tax4g # Genuine aggregate revenue from indirect taxes on exports #;
        (Change)
del_tot_tax4ph # Change in agg. revenue from Ph. indirect taxes on exports #;
        (Change)
del_tot_tax5 # Aggregate revenue from indirect taxes on government #;
        (Change)
del_tot_tax5g # Genuine aggregate revenue from indirect taxes on government #;
        (Change)
del_tot_tax0 # Aggregate revenue from taxes on production #;
        (Change)
del_tot_tax0g # Genuine aggregate revenue from taxes on production #;
        (Change)
del_unity # Normally shocked to unity #;

        (Change)
del_wealth # 1 in wealth-sens. revenue-replac. car tariff cut sim, else 0 #;
div_aef # $A value of profits on Australian equity held by f'gners #;
div_fea # $A value of profits on f'gn equity held by Australians #;
        (All,j,IND)
div_fea_j(j) # $A value of profits in j on equity held by f'gners #;
        (Change)
d_add_rev # Used in calculating revenue replacement tax in car tariff cut sim #;
        (Change)
d_balitem # Balancing item in the balance of payments #;
        (Change)
d_cad # Current account deficit #;
        (Change)
d_col_payr # Aggregate collection of payroll taxes #;
        (Change)
d_def_gdp_r # Change in the ratio of the government deficit to GDP #;
        (Change)(All,j,IND)
d_diseq(j) # Disequilibrium in expected rate of return, static expect. #;
        (Change)
d_dum_year1 # One in year one, zero in later years #;
        (Change)(All,j,IND)
d_eeqror(j) # Equilibrium expected rate of return, static expectations #;
        (Change)
d_empadj # Determines the speed of direct adjustment of employment #;
        (Change)
d_emp_sh # Set at one to zero out shift variable in E_d_f_empadj #;
        (Change)(All,j,IND)
d_eror(j) # Percentage point changes in expected rates of return #;
d_eror_ave # Average expected rate of return #;
        (Change)(All,j,IND)
d_eror_o(j) # Expected rate of return for industry j in forecast #;
        (Change)(All,j,IND)
d_f(j) # Exog. in all iters with rational expect., endo for static exp #;
        (Change)(All,j,IND)
d_ff(j) # Exogenous in simulations  with static exp., endo for rat. exp. #;
        (Change)
d_ff_empadj # Exogenized to zero out shift variable in E_d_f_empadj #;
        (Change)
d_ff_excht # Endogenous if initial solution is from t-1 #;
        (Change)
d_ff_excht1 # Endogenous if initial solution is not from t-1 #;

        (Change)(All,j,IND)
d_ff_pcapatt(j) # Endogenous  if initial solution is from t-1 #;
        (Change)(All,j,IND)
d_ff_pcapatt1(j) # Endogenous  if initial solution is not from t-1 #;
        (Change)
d_ff_xiwldt # Endogenous if initial solution is from t-1 #;
        (Change)
d_ff_xiwldt1 # Endogenous if initial solution is not from t-1 #;
        (Change)(All,j,IND)
d_fpowtax0g(j) # Shift in power of genuine production tax #;
        (Change)
d_f_aef_t # Turns off aef_t eqn., if initial solution is not from t-1 #;
        (Change)(All,j,IND)
d_f_diseq(j) # Shifter in determination of d_diseq in static expectations #;
        (Change)(All,j,IND)
d_f_diseqre(j) # Shifter in determination of d_diseq in rational expectations #;
        (Change)(All,j,IND)
d_f_eror_o(j) # Shift in equation that records forecast of expected ror #;
        (Change)
d_f_eeqror # General capital growth shifter, in yr-to-yr #;
        (Change) (All,j,IND)
d_f_eeqror_j(j) # Industry-specific cap. growth shifter, yr-to-yr, rational #;
        (Change)
d_f_excht # Endogenous if initial solution is not from t-1 #;
        (Change)
d_f_excht1 # Endogenous if initial solution is from t-1 #;
        (Change)
d_f_fcdc_t # Turns off fcdc_t eqn., if initial solution is not from t-1 #;
        (Change)
d_f_fcfc_t  # Turns off fcfc_t eqn., if initial solution is not from t-1 #;
        (Change)
d_f_fddc_t # Endogenous if initial soln is not from t-1 #;
        (Change)
d_f_fdfc_t # Endogenous if initial soln is not from t-1 #;
        (Change)
d_f_empadj # Exogenized to cause direct adjustment of aggregate employment #;
        (Change)(All,j,IND)
d_f_fea_t_j(j) # Endogenous if initial soln is not from t-1 #;
        (Change)
d_f_labsup # Shift in labour supply #;
        (Change)
d_f_netfltf # Shifter, exog. in historical/decomp sims, endog. in yr-to-yr #;

        (Change)
d_f_othcapgov # Ratio of purchases or sales of 2nd hand assets by gov to GDP #;
        (Change)(All,j,IND)
d_f_pcapatt(j) # Endogenous  if initial solution is not from t-1 #;
        (Change)(All,j,IND)
d_f_pcapatt1(j) # Endogenous  if initial solution is from t-1 #;
        (All,j,IND)
d_f_pi_l(j) # Turns off lag cap. price equ, if initial soln is not from t-1 #;
        (Change)
d_f_psd_t # Turns of eqn. for pub. sect. def. if initial soln. is not for t-1 #;
        (Change)
d_f_psd_t1 # Shifter in equation for end of year public sector debt #;
        (Change)
d_f_rint_l # Turns off lag real int. rate equ, if init. soln is not from t-1 #;
        (Change)
d_f_rint_psd # Shift, diff btn real rates of interest on PSD and bus borrow #;
        (Change)(All,j,IND)
d_f_ror_se_o(j) # Shift in eqn. that records static exp. rate of return #;
        (Change)(All,i,COM)
d_f_tw_forc(i) # Shift in impact of twist trends equation, forecast #;
        (Change)(All,i,COM)
d_f_tw_hist(i) # Shift in impact of twist trends equation, historical #;
        (Change)
d_f_xi3_2l # Turns off 2 lag inflation equ, if init. soln is not from t-1 #;
        (Change)
d_f_xi3_l # Shift for xi3_l, should be endog. if initial sol is not year t-1 #;
        (Change)
d_f_xiwldt # Endogenous if initial solution is not from t-1 #;
        (Change)
d_f_xiwldt1 # Endogenous if initial solution is from t-1 #;
        (Change)(All,i,COM)
d_gamma(i) # Change in subsistence cons. of i expressed as % of cons. of i #;
        (Change)
d_gov_def # Public sector deficit, or public sector financing transactions #;
        (Change)
d_inf # Rate of inflation #;
        (Change)
d_inf_l # Lagged rate of inflation, proport. change in CPI from t-2 to t-1 #;
        (Change)
d_int # Change in nominal rate of interest #;
        (Change)
d_int_l # Lagged rate of interest, i.e. rate of interest in t-1 #;

        (Change)
d_int_psd # Rate of interest on public sector debt #;
        (Change)
d_inventreal # Aggregate real inventory changes #;
        (Change)
d_netflt # $A value of net foreign liabilities #;
        (Change)
d_netfltf # $F value of net foreign liabilities #;
        (Change)
d_net_int_g # Net interest paid by government #;
        (Change)
d_net_tax_gdp # Change in the net tax to GDP ratio #;
        (Change)
d_newfa # New foreign assets #;
        (Change)
d_newfl # $A value of f'gn borrow. & f'gn equity invest. in Aust. during yr #;
        (Change)
d_new_aef # $A value of new f'gn equity held by Australians #;
        (Change)
d_new_fcdc # $A value of new f'gn credit given in dom. currency #;
        (Change)
d_new_fcfc # $A value of new f'gn credit given in f'gn. currency #;
        (Change)
d_new_fddc # $A value of new f'gn borrowing incurred in dom. currency #;
        (Change)
d_new_fdfc # $A value of new f'gn borrowing incurred in f'gn currency #;
        (Change)
d_new_fea # $A value of new f'gn-owned equity in Australia #;
        (Change)(All,j,IND)
d_new_fea_j(j) # $A value of new f'gn-owned equity in industry j #;
        (Change)
d_nfl_gdp # Change in ratio of st-of-yr net foreign liabilities to GDP #;
        (Change)
d_nfl_vcap # Change in ratio of st-of-yr NFL to aggregate capital stock #;
        (Change)
d_othcapgov # Purchase or sale of 2nd hand assets by government #;
        (Change)
d_psd_t # Public sector debt, start of year #;
        (Change)
d_psd_t1 # Public sector debt, end of year #;
        (Change)
d_rint # Real interest rate #;

        (Change)
d_rint_l # Lag real rate of interest, i.e. real rate of interest in year t-1 #;
        (Change)
d_rint_psd # Real rate of interest on public sector debt #;
        (Change)
d_rint_pt_se # Real post-tax interest rate, static expectations #;
        (Change)(All,j,IND)
d_ror_act_l(j) # Lagged actual rate of return #;
        (Change)
d_r_balgdp # Ratio of balancing item in the CAD to GDP #;
        (Change)
d_r_cadgdp # Change in the ratio of CAD to GDP #;
        (Change)
d_r_psdgdp # Ratio of st-of-yr public sector debt to GDP #;
        (Change)
d_trn_fgn_a # $A value of net transfers from foreigners to Australians #;
        (Change) (All,i,COM)(All,s,SOURCE)
d_x6cs(i,s) # Inventory changes #;
        (Change)
d_xi6 # Inventories price index #;
eligsh # Share of unemployed receiving unemployment benefits #;
emp_c_wgts # Aggregate employment, labour cost weights #;
emp_hours # Aggregate employment, hours #;
emp_hours_o # Aggregate employment in hours in forecast simulation #;
ev_over # Upper bound on equivalent variation as % of h'hold expenditure #;
excht # Start of year exchange rate, $f'gn per $Aust. #;
excht1 # End of year exchange rate, $ f'gn per $ Aust. #;
expvald # Domestic currency value of exports #;
expvalf # Foreign currency value of exports #;
expvol # Export volume index #;
        (All,j,IND)
f1_commun(j) # Shifts in industry import demands for communication #;
        (All,j,IND)
f1_commun_a(j) # Shifts in altern. specific., imports of communication #;
f1_commun_u # Allows uniform shift in int. demands for communication #;
        (All,j,IND)
f1_trans(j) # Shifts in industry import demands for transport services #;
        (All,j,IND)
f1_trans_a(j) # Shifts in altern. specification, imports of transport serv. #;
f1_trans_u # Allows uniform shift in int. demands for trans. #;
        (All,i,COM)
f5dom(i) # Shift terms for government demands domestic #;

f5gen # Overall shift term for government demands #;
        (All,i,COM)
f5imp(i) # Shift terms for government demands imported #;
        (All,i,COM)
fa1c(i) # Shifter: input-i-saving technical change in current production #;
        (All,i,COM)(All,j,IND)
fa1ci(i,j) # Shifter: input-i-saving tech. change in production, industry j #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
fa1marg(i,s,j,r)
  # Shifter: tech change in margins use, deliv. to current production #;
        (All,r,MARGCOM)
fa1margc(r) # Shifter: technical change in intermediate margin use of r #;
        (All,i,COM)
fa2c(i) # Shifter: input-i-saving technical change in investment #;
        (All,i,COM)(All,j,IND)
fa2ci(i,j) # Shifter: input-i-saving tech. change in industry j's investment #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
fa2marg(i,s,j,r)
  # Shifter: tech change in margins use, delivery to investment #;
        (All,r,MARGCOM)
fa2margc(r) # Shifter: technical change investment margin use of r #;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
fa3marg(i,s,r) # Shifter: tech change in margins use, deliv. to cons. #;
        (All,r,MARGCOM)
fa3margc(r) # Shifter: technical change in household margins use of r #;
        (All,i,COM)(All,r,MARGCOM)
fa4marg(i,r) # Shifter: tech. change in margins use, deliveries to export #;
        (All,r,MARGCOM)
fa4margc(r) # Shifter: technical change in export margin use of r #;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
fa5marg(i,s,r) # Shifter: tech. change in margins use, deliveries to govt. #;
        (All,r,MARGCOM)
fa5margc(r) # Shifter: technical change in government margins use of r #;
fat # Total foreign assets, start of year #;
fat1 # Total foreign assets, end of year #;
fcdc_t # $A value of f'gn credit given in dom. currency, start of year #;
fcdc_t1 # $A value of f'gn credit given in dom. currency, end of year #;
fcfc_t   # $A value of f'gn credit given in f'gn currency, start of year #;
fcfc_t1  # $A value of f'gn credit given in f'gn currency, end of year #;
fddc_t # $A value of f'gn debt incured in dom. currency, start of year #;
fddc_t1 # $A value of f'gn debt incured in dom. currency, end of year #;
fdfc_t # $A value of f'gn debt incured in f'gn currency, start of year #;

fdfc_t1 # $A value of f'gn debt incured in f'gn currency, end of year #;
fea_t1 # $A value of f'gn-owned equity in Aust.,end of year #;
fea_t # $A value of f'gn-owned equity in Aust., start of year #;
        (All,j,IND)
fea_t1_j(j) # $A value of f'gn-owned equity in j in Aust., end of year #;
        (All,j,IND)
fea_t_j(j) # $A value of f'gn-owned equity in j in Aust., start of year #;
(All,i,TRADEXP)
fep(i) # Price (vertical) shifts in traditional export demand schedules #;
fep_ntrad # Price shifter (vertical) for demand for non-traditional exports #;
        (All,i,NTRADEXP)
fep_ntradi(i) # Shifter for composition of nontraditional exports #;
fep_tour # Price shifter (vertical) in the demand for the tourism composite #;
        (All,i,TOURISM)
fep_touri(i) # Shift in composition of tourism exports #;
        (All,i,TRADEXP)
feq(i) # Quantity (horizontal) shifts in traditional export demand schedules #;
feq_general # General export shifter #;
feq_ntrad # Quantity shifter (horizontal) for demand for non-trad exports #;
feq_tour # Quantity shifter (horizontal) in demand for tourism composite #;
        (All,i,COM_UP)
ffac(i) # Allows diffs in i-saving tc for i's in same ABS historical class'n #;
ffpowtaxm # Uniform shift, power of tariffs #;
        (All,qq,INDSAGGTO20)
ff_p_va_s(qq) # Shift in capital/other cost adjustment equation #;
ff_shfeat # General shift, start-of-year foreign-ownership share #;
ff_shfeat1 # General shift, end-of-year foreign-ownership share #;
        (All,j,IND)
ff_twistlk(j) # Frees up twistlk(j) #;
ff_y_g # Scalar shifter in public sector investment by industry equation #;
fgnshd # Scalar shifter in formula for dividends on f'gn equity in j #;
        (All,j,IND)
fgnshd_j(j) # Vector shifter in formula for dividends on f'gn equity in j #;
fhist_cont # Allows f’cast exog. ntradexpvol while hist_cont(i)s imposed #;
        (All,i,IND)
flabprod(i) # Allows labour productivity to vary within labour-data inds #;
flt # Total foreign liabilities, start of year #;
flt1 # Total foreign liabilities, end of year #;
        (All,i,COM)
fpdm(i) # Ratio of basic prices: domestic to import #;
        (All,i,COM)(All,s,SOURCE)
fpowtax1gg(i,s) # Genuine power tax shift: uniform across inter usage #;

        (All,i,COM)(All,s,SOURCE)
fpowtax1phph(i,s) # Phantom power tax shift: uniform across inter usage #;
        (All,i,COM)(All,s,SOURCE)
fpowtax2gg(i,s) # Genuine power tax shift: uniform across invest usage #;
        (All,i,COM)(All,s,SOURCE)
fpowtax2phph(i,s) # Phantom power tax shift: uniform across invest use #;
        (All,i,COM)(All,s,SOURCE)
fpowtax3g(i,s) # Genuine power tax shift: consumption #;
fpowtax3gu # Genuine uniform power tax shift: consumption #;
        (All,i,COM)(All,s,SOURCE)
fpowtax3ph(i,s) # Phantom power tax shift: consumption #;
        (All,i,COM)(All,s,SOURCE)
fpowtax5g(i,s) # Genuine power tax shift: government #;
        (All,i,COM)(All,s,SOURCE)
fpowtax5ph(i,s) # Phantom power tax shift: government #;
        (All,i,COM)
fpowtaxm(i) # Shift, power of tariffs #;
        (All,i,COM)
fpowtaxmo(i) # Allows exogenization of powtaxmo in deviation sims #;
        (All,i,COM)
ft2(i) # Allows commodity-specific changes in tariff rates #;
ftax_fgn_gov # Shift in eqn for income taxes paid by f’gners to Australia #;
ftax_l_r_o # Shift in equation that sets the value of tax_l_r_o #;
ftrnoth_fgn # Shift in eqn for other net transfers to foreigners #;
ftrn_fgn_hh # Shift in eqn for unrequit. trans. to Aust. resid. from f'gners #;
        (All,i,COM)
ftwist_eff(i) # Twist caused by strong growth, shifter #;
        (All,i,COM)
ftwist_src(i) # Twist shift - see 'shift twist' #;
fvas # Shifter in the price of value added by sector equation #;
fwage # Overall wage shifter #;
        (All,j,IND)
fwagei(j) # Industry-specific wage shifter #;
        (All,m,OCC)
fwageo(m) # Occupation-specific wage shifter #;
        (All,j,IND)(All,m,OCC)
fwageoi(m,j) # Occupation and industry-specific wage shifter #;
        (All,jj,INDSAGGTO20)
fwage_s(jj) # Sectoral shifter in the price of labour#;
fx4_abare_mi # General shift from ABARE forecasts for mining products #;
fx4_abare_ru # General shift from ABARE forecasts for rural products #;
fx4_commun # Ratio of exports to imports of communication services #;

fx4_transerv # Ratio of transport service exports to activity variable #;
fx4_wattran # Ratio of exports of water transport to trad. export volumes #;
        (All,qq,INDSAGGTO20)
f_1octss(qq) # Average f1oct for sector qq, 20 sector aggregation #;
f_age_ben # Shifter in equation for age pensions #;
        (All,i,COM)(All,s,SOURCE)
f_b3sh_o(i,s) # Shift in the equation that sets the value of b3sh_o(i,s) #;
f_emp_o # Shift in equation that sets the value of emp_hours_o #;
f_fat # Ratio of start-of-year f'gn assets to GDP #;
f_fat1 # Ratio of end-of-year f'gn assets to GDP #;
        (All,i,NTRADEXP)
f_forc(i) # allows imposition of hist_cont(i)s from historical sims #;
f_g # Common movement in public sector invest. in main pub. sect. inds #;
f_gdpreal # Ratio:(exogenous GDP)/(sum of exog. f'csts of C,I,G,X-M) #;
        (All,j,IND)
f_gj(j) # Allows non-uniform moves in govt. invest. in main govt. inds #;
f_grants # Shifter on equation linking grants to GDP #;
(All,i,NTRADEXP)
f_hist(i) # Shifter in the historical hist_cont equation #;
f_labsup_o # Shift term in equation that sets the value of labsup_o #;
        (All,i,NTRADEXP)
f_ntrad(i) # Shifter for composition of non-traditional exports #;
f_oil # Ratio of domestic output of oil to imports of oil #;
f_oth_ben # Shifter in eqn. for benefits except age and unemployment #;
f_oth_g_rev # Ratio of other government revenue to GDP #;
f_pe_u # Allows uniform shift from observed or f'cast values for pe #;
f_pe_u_nt # Uniform shift from obs. or f'cast values for pe for non-trads #;
f_pm # Generalised shifter for pm(i) #;
f_rwage_o # Shift in equation that sets the value of real_wage_c_o #;
f_rwage_pt_o # Shift in equation that sets the value of real_wage_pt_o #;
        (All,j,IND)
f_r_inv_cap(j) # Allows freeing up of r_inv_cap(j) #;
        (All,j,IND)
f_shfeat1_j(j) # Ind-specific shift in end-of-year f'gn-ownership share #;
f_tax_r # Ratio of capital to labour tax rates #;
        (All,i,TOURISM)
f_tour(i) # Shifter for composition of tourism exports #;
f_twistlk # General labour/capital twist #;
f_unempben # Ratio of unemployment benefit rate to CPI #;
        (All,i,COM_JP)
f_x0ci(i) # Allows uniform diff across j between x0ci(i,j) & obs. x0ci(i,j) #;

        (All,i,COM)(All,s,SOURCE)
f_x3cs_o(i,s) # Shift in equation that sets the value of x3cs_o #;
f_xiworld # Shift variable in equation for world inflation #;
        (All,j,IND)
f_y_g(j) # Vector shifter in public sector investment by industry equation #;
gdpexp # Nominal GDP from expenditure side #;
gdpinc # Nominal GDP from income side #;
gdpreal # Real GDP from the expenditure side #;
gdprealinc # Real GDP from the income side #;
gdpreal_true # Can be exogenous forecast of real GDP #;
gne # Nominal GNE #;
gner # Real GNE #;
gnpnom # Gross national product #;
govsav # Government saving: public sector surplus #;
govsav_r # Real government saving #;
grants # Grants & transf. other than unemp, age & oth person. bens. #;
gx4_abare_mi # Total mining exports #;
gx4_abare_ru # Total rural exports #;
hdy # Household disposable income #;
        (All,i,NTRADEXP)
hist_cont(i) # Contrib.to output growth of export growth above non-trad ave #;
housav # Household saving #;
hs_ntrad # Horizontal shift in demand for non-traditional exports #;
        (All,i,COM)
iacrate(i) # 'IAC' ad-valorem tariff rates #;
        (All,i,COM)
impftwist(i) # Twist trends impact on non-margin, non-invent. domestic demand #;
impvald # Domestic currency value of imports #;
impvalf # Foreign currency value of imports #;
impvol # Import volume index #;
in # Aggregate nominal investment #;
int_fcdc # $A value of interest on f'gn credits given in dom. currency #;
int_fcfc # $A value of interest on f'gn credits given in f'gn currency #;
int_fddc # $A value of interest on f'gn debt incured in dom. currency #;
int_fdfc # $A value of interest on f'gn debt incured in f'gn currency #;
        (All,j,IND)
int_input(j) # Total use of intermediate inputs by industry #;
ir # Aggregate real investment expenditure #;
irtrue # National Accounts concept of real investment expenditure #;
k_a_wgts # Aggregate capital: asset value weights #;
k_r_wgts # Aggregate capital: rental weights #;

        (All,j,IND)
labind(j) # Employment by industry #;
        (All,d,LAB_DATA_IND)
labind_d(d) # Employment: hours by 82 labour-data industries #;
        (All,jj,INDSAGGTO20)
labind_s(jj) # Employment by sector #;
        (All,j,IND)
labprod(j) # Labour productivity #;
        (All,d,LAB_DATA_IND)
labprod_d(d) # Used to impose assumptions for lab productiv. in ind. groups #;
labrev # Aggregate costs of labour #;
labrevinc # Labour income net of payroll taxes #;
        (All,j,IND)
labrev_j(j) # Costs of labour by industry #;
labsup # Labour supply #;
labsup_o # Labour supply, forecast sim #;
        (All,m,OCC)
lambda(m) # Employment in occupation m #;
        (Change)
lev_def_gdp_r # Level of ratio of public sector deficit to GDP #;
        (Change)(All,j,IND)
lev_eror(j) # Levels of expected rors in year t #;
        (Change)(All,j,IND)
lev_eror_l(j) # Lagged levels of expect. rors, usually expect. rors for t-1 #;
        (Change)
lev_nfl_gdp  # Level of ratio of net foreign liabilities to GDP #;
        (Change)
lev_nfl_vcap # Level of ratio of net foreign liabilities to capital stock #;
        (Change)(All,j,IND)
lev_ror_act_l(j) # Level of actual ror in year t-1 #;
        (Change)
lev_r_cadgdp # Level of the ratio of CAD to GDP #;
        (Change)
lev_r_psdgdp # Level of ratio of st-of-yr public sector debt to GDP #;
lndr # Aggregate quantity of land #;
lndrev # Aggregate payments to land #;
merch_trad # Volume of merchandise trade #;
        (All,j,IND)
n(j) # Use of land #;
natsav # National saving: household saving plus public sector surplus #;
natsav_r # Real national saving #;
net_tax_tot # Net collection of income taxes and genuine indirect taxes #;

ntradexpvalf # Foreign currency value of non-traditional exports #;
ntradexpvol # Non-traditional exports, volume #;
octrev # Aggregate other cost ticket payments #;
othnom # Aggregate nominal value of government demands #;
othreal # Aggregate real government demands #;
oth_ben # Benefits paid by government, except age and unemployment #;
oth_gov_rev # Government revenue apart from taxes & interest, e.g. profits #;
        (All,i,COM)(All,s,SOURCE)
p0(i,s) # Basic price of good i,source s #;
        (All,cc,COMPCOM)(All,j,IND_JP)
p0ccom(cc,j) # Price of composite commodity cc produced by ind j #;
        (All,i,COM)
p0dom(i) # Basic price of domestic goods #;
        (All,i,COM)
p0imp(i) # Basic price of imported goods #;
        (All,j,IND)
p0ind(j) # Industry basic price of output #;
        (All,j,IND)
p1cap(j) # Rental price of capital #;
p1cap_ave # Economy-wide average rental rate of capital #;
p1cap_ave_f # Average rental rate of capital owned by foreigners #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
p1csi(i,s,j) # Prices for current production #;
        (All,j,IND)
p1lab(j) # Cost of a unit of labour to industry j #;
        (All,j,IND)(All,m,OCC)
p1laboi(m,j) # Cost of a unit of labour of occupation m in industry j #;
        (All,j,IND)
p1land(j) # Rental price of land #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
p2csi(i,s,j) # Prices for capital creation #;
        (All,i,COM)
p3(i) # Price of household composites #;
        (All,i,COM)(All,s,SOURCE)
p3cs(i,s) # Purchasers' prices by commodities and source to households #;
p3_adj # Adjustment to observered consumer prices #;
        (All,i,COM)
p3_obs(i) # Observed consumer prices #;
        (All,i,COM)(All,s,SOURCE)
p5cs(i,s) # Purchasers' prices paid for commodities (by source) by government #;
        (All,j,IND)
pcapatt(j) # Asset price of capital by industry, start of year #;

        (All,j,IND)
pcapatt1(j) # Asset price of capital by industry, end of year #;
        (All,i,COM)
pe(i) # F.o.b. foreign currency export prices #;
        (All,i,COM)
peobs(i) # Observed foreign currency export prices #;
phi # Exchange rate (dollar foreign per dollar Aus) #;
        (All,j,IND)
pi(j) # Costs of units of capital #;
        (All,j,IND)
pi_l(j) # Lagged price of capital, price in t-1 #;
        (All,i,COM)
pm(i) # C.i.f. foreign currency import prices #;
        (All,i,COM)
pmobs(i) # Observed import prices, foreign currency #;
        (All,i,COM)
pmrel(i) # Historical distribution of foreign currency import prices #;
pm_f # Foreign currency price index for aggregate imports #;
pm_ff # Allows hist. distrib. of f'gn curr. import prices to affect forecast #;
        (All,j,IND)
powtax0(j) # Power of production tax #;
        (All,j,IND)
powtax0g(j) # Power of genuine production tax #;
pop # Australian population #;
pop_aged # Population over 65 #;
        (All,j,IND)
powpayroll(j) # Power of payroll tax #;
        (All,j,IND)
powtax0ph(j) # Power of phantom production tax #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
powtax1(i,s,j) # Power of tax on sales to intermediate #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
powtax2(i,s,j) # Power of tax on sales to capital creation #;
        (All,i,COM)(All,s,SOURCE)
powtax3(i,s) # Power of tax on flows to households #;
        (All,i,COM)
powtax3vg(i) # Allows powtax on imported and dom cons. goods to move as one #;
        (All,i,COM)
powtax4(i) # Power of export tax #;
        (All,i,COM)
powtax4g(i) # Power of genuine export tax #;

        (All,i,COM)
powtax4ph(i) # Power of phantom export tax #;
        (All,i,COM)
powtax4sph(i) # Power of spreading phantom export tax #;
        (All,i,COM)(All,s,SOURCE)
powtax5(i,s) # Power of tax on sales to government #;
        (All,i,COM)(All,s,SOURCE)
powtaxgg(i,s) # Power of common genuine tax on non-export sales #;
        (All,i,COM)
powtaxm(i) # Power of tariffs #;
        (All,i,COM)
powtaxmo(i) # Forecast of power of tariff, used in deviation sims #;
        (All,i,COM)(All,s,SOURCE)
powtaxphph(i,s) # Power of common phantom tax on non-export sales #;
pow_phph3_ave # Average powtaxphph applying to consumption #;
        (All,j,IND)
p_va(j) # Price of value added by industry #;
        (All,j,INDSAGGTO20)
p_va_s(j) # Price of value added by sector #;
q # Number of households #;
realdev # Real devaluation index #;
real_housav # Real household saving #;
real_wage_c # Real wage for consumers #;
real_wage_c_o # Real wage in forecast sim #;
real_wage_e # Wage rate deflated by the price for GDP #;
real_wage_e2 # Wage rate deflated by the price for primary factors #;
real_wage_pt # Real post-tax wage to consumers #;
real_wage_pt_o # Real post-tax wage to consumers, forecast sim #;
rodiv_aef # Rate of profit on f'gn equity held by Australians #;
roi_fcdc # Rate of interest on f'gn credit given in dom. currency #;
roi_fcfc # Rate of interest on f'gn credit given in f'gn currency #;
roi_fddc # Rate of interest on f'gn debt incurred in dom. currency #;
roi_fdfc # Rate of interest on f'gn debt incurred in f'gn currency #;
rwealth_at_t # Real national wealth at start of year t #;
r_cr_othreal # Ratio of real private comsumption to real gov't consumption #;
        (All,s,SOURCE)
r_elec_cpi(s) # Ratio of consumer price of electricity to CPI #;
r_cr_gner # Ratio of real consumption to real GNE #;
r_cr_ir # Ratio of real consumption to real investment #;
r_c_gdpinc # Ratio of nominal consumption to nominal GDPINC #;
r_c_inc # Ratio of nominal consumption to income from lab, cap and land #;

        (All,j,IND)
r_inv_cap(j) # Investment/capital ratio shifters by industry #;
        (All,s,INDSAGGTO26)
r_inv_cap_s(s) # Investment/capital shifter, 18 sectors #;
r_inv_cap_u # Uniform shifter in investment/capital ratios #;
r_ir_gner # Ratio of real investment to real GNE #;
r_othr_gner # Ratio of real government demands to real GNE #;
        (All,j,IND)
shfeat1_j(j) # Industry-specific shift, end-of-year foreign-ownership share #;
        (All,j,IND)
shfeat_j(j) # Industry-specific shift, start-of-year foreign-ownership share #;
sht1fcdc # Share of dom-currency credit in total f'gn assets, end of year #;
sht1fcfc # Share of f'gn-currency credit in total f'gn assets, end of year #;
sht1fddc # Share of dom-currency debt in total liabilities, end of year #;
sht1fdfc # Share of f'gn-currency debt in total liabilities, end of year #;
shtfcdc # Share of dom-currency credit in total f'gn assets, start of year #;
shtfcfc # Share of f'gn-currency credit in total f'gn assets, start of year #;
shtfddc # Share of dom-currency debt in total liabilities, start of year #;
shtfdfc # Share of f'gn-currency debt in total liabilities, start of year #;
t2_star # Allows uniform shift in tariff rates #;
taxind # Aggregate revenue from all indirect taxes #;
taxindg # Aggregate revenue from all genuine indirect taxes #;
taxrev1 # Aggregate revenue from indirect taxes on intermediate #;
taxrev1g # Genuine aggregate revenue from indirect taxes on intermediate #;
taxrev2 # Aggregate revenue from indirect taxes on capital creation #;
taxrev2g # Genuine aggregate revenue from indirect taxes on capital creation #;
taxrev3 # Aggregate revenue from indirect taxes on households #;
taxrev3g # Genuine aggregate revenue from indirect taxes on households #;
        (All,i,COM)
taxrev3gc(i) # Genuine tax collections, consumption of (i) #;
taxrevm # Aggregate tariff revenue #;
taxrev_inc # Income tax revenue #;
tax_fgn_gov # $A value of income taxes paid by foreigners to Australia #;
tax_k_r # Rate of tax on capital income #;
tax_l_r # Rate of tax on labour income #;
tax_l_r_o # Rate of tax on wages in the forecast run #;
toft # Terms of trade #;
tradexpvalf # Foreign currency value of traditional exports #;
tradexpvol # Traditional exports, volume #;
transfers # Transfers from the government #;
trnoth_fgn # $A value of other net transfers to foreigners #;
trn_fgn_hh # $A value of unrequited transfers to Aust. residents from f'gners #;

        (All,i,COM)
tt(i) # Total trade #;
        (All,j,IND)
twistlk(j) # Labour capital twist by industry #;
        (All,s,INDSAGGTO26)
twistlk_s(s) # Labour capital twist by sector #;
        (All,i,COM)
twist_eff(i) # Twist caused by strong growth #;
        (All,i,COM)
twist_src(i) # Import/domestic twist by commodity #;
twist_src_bar # Common twist #;
unempben_rat # Unemployment benefit rate #;
unemp_ben # Unemployment benefits received #;
        (All,jj,INDSAGGTO20)
wagebill_s(jj) # Wagebill by sector, excludes payroll tax #;
wealth_at_t # National wealth at start of year t #;
        (All,j,IND)(All,m,OCC)
wlaboi(m,j) # Wage of occupation type m in industry j, excludes payroll tax #;
        (All,jj,INDSAGGTO20)
wlab_s(jj) # Wage rate of labour by sector, excludes payroll tax #;
world_gdp # World GDP, real #;
        (All,i,COM)
x0(i) # Quantity of sales of i (domestic and imported) in Australia #;
        (All,cc,COMPCOM)(All,j,IND_JP)
x0ccom(cc,j) # Supply of composite commodity cc produced by ind j #;
        (All,i,COM_JP)(All,j,IND_JP)
x0ci(i,j) # Supply of commodity i produced by industry j #;
        (All,i,COM_JP)(All,j,IND_JP)
x0ci_obs(i,j) # Supply of commodity i produced by industry j, observed #;
        (All,i,COM)
x0dom(i) # Total supplies of domestic goods #;
        (All,t,ABS_COM)
x0dom_abs(t) # Commodity outputs: ABS categories in historical data #;
        (All,abs,ABS_COM)
x0dom_absobs(abs) # Commodity outputs: ABS cat. in historical data, observed #;
        (All,i,COM)
x0dom_dom(i) # Quantity of sales of domestically produced i in Australia #;
        (All,i,COM)
x0imp(i) # Total supplies of imported goods #;
        (All,i,COM)
x0imp_obs(i) # Total supplies of imported goods, observed #;

        (All,j,IND)
x0ind(j) # Output of all industries #;
        (All,s,INDSAGGTO26)
x0ind_26s(s) # Output of 26 sectors #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
x1csi(i,s,j) # Demands for inputs for current production #;
        (All,j,IND)(All,m,OCC)
x1laboi(m,j) # Employment of occupation type m in industry j #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
x1marg(i,s,j,r) # Margins, production #;
        (All,j,IND)
x1oct(j) # Demand for other cost tickets #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)
x2csi(i,s,j) # Demands for inputs for capital creation #;
        (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
x2marg(i,s,j,r) # Margins, capital creation #;
        (All,i,COM)
x3(i) # Household demands undifferentiated by source #;
        (All,i,COM)(All,s,SOURCE)
x3cs(i,s) # Household demand for goods #;
        (All,i,COM)(All,s,SOURCE)
x3cs_o(i,s) # Consumption by commodity/source in forecast simulation #;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
x3marg(i,s,r) # Margins, households #;
        (All,na,NCOM)
x3ncom(na) # Consumption composite, national accounts level #;
        (All,na,NCOM)
x3ncom_obs(na) # Consumption composite, national accounts level, observed #;
        (All,i,COM)
x3_imputed(i) # Consumption of i in the absence of taste changes #;
        (All,i,COM)(All,na,NCOM)
x3_m_na(i,na) # Cons. of cross-classified MONASH/national accounts commods. #;
        (All,i,COM)
x3_obs(i) # Consumption of commodity i, observed #;
        (All,i,COM)
x4(i) # Export volumes #;
        (All,i,COM)(All,r,MARGCOM)
x4marg(i,r) # Margins, exports #;
        (All,i,ABARE)
x4_abare(i) # Usually ABARE export forecasts #;
        (All,i,COM)
x4_obs(i) # Export volumes, observed #;

        (All,i,COM)(All,s,SOURCE)
x5cs(i,s) # Demands for inputs for government demands #;
        (All,i,COM)(All,s,SOURCE)
x5cs_obs(i,s) # Demands for inputs for government demands, observed #;
        (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
x5marg(i,s,r) # Margins, government #;
xi2 # Aggregate investment price index #;
xi2g # Price index for government investment expenditure #;
xi2_t # Index of start-of-year prices of investment goods #;
xi3 # Consumer price index #;
xi3_2l # Double lagged rate of inflation, usually from t-3 to t-2 #;
xi3_l # Lagged CPI, usually CPI in year t-1 #;
xi4 # Exports price index #;
xi4ntrad  # Non-traditional exports price index, foreign currency #;
xi4tour # Price index for the tourism composite #;
xi4trad # Traditional exports price index, foreign currency #;
xi5 # "Other" demands price index #;
xifac # Price of primary factors (labour, capital, land) #;
xigdp # GDP price index, expenditure side #;
xigne # GNE price deflator #;
xim # Index of c.i.f. imports prices, domestic currency #;
ximf # Index of cif prices of imports, foreign currency #;
ximp0 # Duty-paid imports price index #;
xiworld # World price level, average in year #;
xiworld_t # Start-of-year world price level #;
xiworld_t1 # End-of-year world price level #;
        (All,j,IND)
y(j) # Capital creation by using industry #;
        (All,j,IND)
y_g(j) # Public sector investment by industry #;
        (All,s,INDSAGGTO26)
y_s(s) # Investment for sectors #;
        (All,j,IND)
z(j) # Activity level or value added #;

! 18.7 Updates in alphabetical order !

Update
AGEBEN = age_ben;
AGGTAX4F = expvald;
(All,i,COM)(All,s,SOURCE) B3SH(i,s) =b3shr(i,s);
(All,i,COM)(All,s,SOURCE) B3SHO(i,s) =b3sh_o(i,s);
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND: BAS1(i,s,j) ne 0)

BAS1(i,s,j) = [BAS1(i,s,j)/100]*[p0(i,s) + x1csi(i,s,j)];
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND: BAS2(i,s,j) ne 0)
BAS2(i,s,j) = [BAS2(i,s,j)/100]*[p0(i,s) + x2csi(i,s,j)];
        (Change)(All,i,COM)(All,s,SOURCE)
BAS3(i,s) = [BAS3(i,s)/100]*[p0(i,s) + x3cs(i,s)];
        (Change)(All,i,COM: BAS4(i) ne 0)
BAS4(i) = [BAS4(i)/100]*[p0(i,"dom") + x4(i)];
        (Change)(All,i,COM)(All,s,SOURCE: BAS5(i,s) ne 0)
BAS5(i,s) = [BAS5(i,s)/100]*[p0(i,s) + x5cs(i,s)];
        (Change)(All,i,COM)(All,s,SOURCE)
BAS6(i,s) = [BAS6(i,s)/100]*p0(i,s) + P0LEV(i,s)*d_x6cs(i,s);
(Change) CADEF = d_cad;
(All,j,IND) CAPITAL(j) = p1cap(j)*cap_at_t(j);
(All,i,COM) DELTA(i) = deltapc(i);
DIV_AE_F = div_aef;
DIV_FE_A = div_fea;
(Change) DUM_YEAR1 = d_dum_year1;
EMPLOY = emp_hours;
EMPLOY_OLD = emp_hours_o;
(Change)(All,j,IND)(All,t,TIME) EROR_G(j,t) = If(DUM_TIME_LAG(t) NE 0,
                 ZERO_PYR1*[DUM_IT1+(1-DUM_IT1)*ADJ_RE(j)]*d_ror_act_l(j));
EXCH = phi;
EXCH_T= excht;
EXCH_T_1 = excht1;
FA_T1 = fat1;
(Change) FEMPADJ = d_f_empadj;
FGN_SHD = fgnshd;
(All,j,IND) FGN_SHD_J(j) = fgnshd_j(j);
FL_T = flt;
FL_T1 = flt1;
(Explicit) FRISCH = -100/
      [100 - Sum(i,COM, SS3COM(i)*(100+p3(i)-c+q) + S3COM(i)*d_gamma(i))];
(Change) F_EEQROR = d_f_eeqror;
(Change)(All,j,IND) F_EEQROR_J(j) = d_f_eeqror_j(j);
GRANT = grants;
(All,j,IND) G_VINVEST(j) = y_g(j)*pi(j);
INTFCDC = int_fcdc;
INTFCFC = int_fcfc;
INTFDDC = int_fddc;
INTFDFC = int_fdfc;
(Change) ITER_NUM = ITER_ADJUST*del_unity;
(All,m,OCC)(All,j,IND) LABOCCIND(m,j) = p1laboi(m,j)*x1laboi(m,j);

LAB_SUP = labsup;
LAB_SUP_O = labsup_o;
(All,j,IND) LAND(j) = p1land(j)*n(j);
(All,i,COM)(All,s,SOURCE) LEVX3(i,s) =x3cs(i,s);
(All,i,COM)(All,s,SOURCE) LEVX3_O(i,s) =x3cs_o(i,s);
LEV_CPI = xi3;
LEV_CPI_2L = xi3_2l;
LEV_CPI_L = xi3_l;
(All,i,COM) LEV_PTAXM(i) = powtaxm(i);
LEV_XI2T = xi2_t;
        (Change)(All,i,COM_JP)(All,j,IND_JP : MAKE(i,j) ne 0)
MAKE(i,j) = [MAKE(i,j)/100]*[p0dom(i) + x0ci(i,j)];
        (Change)(All,i,COM_UP)(All,j,IND_UP: MAKE(i,j) ne 0)
MAKE(i,j) = [MAKE(i,j)/100]*[p0dom(i) + x0dom(i)];
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
        (All,r,MARGCOM: MAR1(i,s,j,r) ne 0)
MAR1(i,s,j,r) = [MAR1(i,s,j,r)/100]*[p0dom(r) + x1marg(i,s,j,r)];
        (Change)(All,i,COM)(All,s,SOURCE)(All,j,IND)
        (All,r,MARGCOM: MAR2(i,s,j,r) ne 0)
MAR2(i,s,j,r) = [MAR2(i,s,j,r)/100]*[p0dom(r) + x2marg(i,s,j,r)];
        (Change)(All,i,COM)(All,s,SOURCE)(All,r,MARGCOM: MAR3(i,s,r) ne 0)
MAR3(i,s,r) = [MAR3(i,s,r)/100]*[p0dom(r) + x3marg(i,s,r)];
        (Change)(All,i,COM)(All,r,MARGCOM: MAR4(i,r) ne 0)
MAR4(i,r) = [MAR4(i,r)/100]*[p0dom(r) + x4marg(i,r)];
        (Change)(All,i,COM)(All,s,SOURCE)(All,r,MARGCOM: MAR5(i,s,r) ne 0)
MAR5(i,s,r) = [MAR5(i,s,r)/100]*[p0dom(r) + x5marg(i,s,r)];
(Change)(All,i,COM)(All,na,NCOM) MM(i,na) = MM(i,na)*[x3_m_na(i,na)+p3(i)]/100;
(Change) NET_FLT = d_netflt;
OTHBEN = oth_ben;
(Change) OTHCAPGOV = d_othcapgov;
        (Change)(All,j,IND)
OTHCOST(j) = [OTHCOST(j)/100]*x1oct(j) + ROTHCOST(j)*del_p1oct(j);
OTHGOVREV = oth_gov_rev;
(All,i,COM)(All,s,SOURCE) P0LEV(i,s) = p0(i,s);
(Change)(All,j,IND) P1OCT(j) = del_p1oct(j);
(All,j,IND) PCAP_AT_T(j) = pcapatt(j);
(All,j,IND) PCAP_AT_T1(j) = pcapatt1(j);
(All,j,IND) PCAP_J(j) = pi(j);
(All,j,IND) PCAP_J_L(j) = pi_l(j);
(All,j,IND) POW_PAYROLL(j) =powpayroll(j);
(Change) PSDATT = d_psd_t;
(Change) RINT = d_rint;



MONASH  represented in TABLO language 203

(Change) RINT_L =  d_rint_l;
(Change) RINT_PSD = d_rint_psd;
RWAGE = real_wage_c;
RWAGE_OLD = real_wage_c_o;
RWEAL_T = rwealth_at_t;
(All,j,IND) SHFEA_T1_J(j) = shfeat1_j(j);
(All,j,IND) SHFEA_T_J(j) = shfeat_j(j)*ff_shfeat;
SHT1_FCDC = sht1fcdc;
SHT1_FCFC = sht1fcfc;
SHT1_FDDC = sht1fddc;
SHT1_FDFC = sht1fdfc;
SHT_FCDC = shtfcdc;
SHT_FCFC = shtfcfc;
SHT_FDDC = shtfddc;
SHT_FDFC = shtfdfc;
(Change)(All,j,IND) TAX0F(j) = del_tax0ph(j);
(Change)(All,j,IND) TAX0G(j) = del_tax0g(j);
(Change)(All,i,COM)(All,s,SOURCE)(All,j,IND) TAX1F(i,s,j) = del_tax1ph(i,s,j);
(Change)(All,i,COM)(All,s,SOURCE)(All,j,IND) TAX1G(i,s,j) = del_tax1g(i,s,j);
(Change)(All,i,COM)(All,s,SOURCE)(All,j,IND) TAX2F(i,s,j) = del_tax2ph(i,s,j);
(Change)(All,i,COM)(All,s,SOURCE)(All,j,IND) TAX2G(i,s,j) = del_tax2g(i,s,j);
(Change)(All,i,COM)(All,s,SOURCE) TAX3F(i,s) = del_tax3ph(i,s);
(Change)(All,i,COM)(All,s,SOURCE) TAX3G(i,s) = del_tax3g(i,s);
(Change)(All,i,COM) TAX4F(i) = del_tax4ph(i);
(Change)(All,i,COM) TAX4G(i) = del_tax4g(i);
(Change)(All,i,COM)(All,s,SOURCE) TAX5F(i,s) = del_tax5ph(i,s);
(Change)(All,i,COM)(All,s,SOURCE) TAX5G(i,s) = del_tax5g(i,s);
TAXFGNGOV = tax_fgn_gov;
TRNFGNHH = trn_fgn_hh;
TRNOTHFGN = trnoth_fgn;
TAX_K_RATE = tax_k_r;
TAX_L_RATE = tax_l_r;
TAX_L_RATE_O = tax_l_r_o;
UNEMPBEN = unemp_ben;
(All,j,IND) VCAP_AT_T(j) = cap_at_t(j)*pcapatt(j);
(All,m,OCC)(All,j,IND) WAGE(m,j) = wlaboi(m,j);
WP = xiworld;
WPT = xiworld_t;
WPTPLUS1 = xiworld_t1;
(Change) YEAR =  del_unity;

! 18.8 Equations in thematic order !

! 18.8a Outputs of commodities !

Equation E_x0ccom # Supplies of composite commodities by joint-product inds #
  (All,cc,COMPCOM)(All,j,IND_JP)
  x0ccom(cc,j) = z(j)
   + SIG0CC(cc,j)*[p0ccom(cc,j) - Sum(t,COMPCOM, MH0CC(t,j)*p0ccom(t,j))]
   - a0ind(j) - a0ccom(cc,j)
   - SIG0CC(cc,j)*[a0ccom(cc,j) - Sum(t,COMPCOM, MH0CC(t,j)*a0ccom(t,j))];

Equation E_x0ci   # Supplies of commodities by joint-product industries #
  (All,i,COM_JP)(All,j,IND_JP)
  x0ci(i,j) = Sum(cc,COMPCOM, CCPROD(i,j,cc)*x0ccom(cc,j))
    + Sum(cc,COMPCOM, CCPROD(i,j,cc)*SIGMACC(cc,j)*[p0dom(i)
       - Sum(ii,COM_JP, CCPROD(i,j,cc)*CCPROD(ii,j,cc)*S0C(ii,j)*p0dom(ii))])
    - a0com(i) - a0ci(i,j)
  + Sum(cc,COMPCOM, CCPROD(i,j,cc)*SIGMACC(cc,j)*[- a0com(i) - a0ci(i,j)
  - Sum(ii,COM_JP,CCPROD(i,j,cc)*CCPROD(ii,j,cc)*S0C(ii,j)*
                                    (-a0com(ii) - a0ci(ii,j)))]);

Equation E_p0ccom # Prices of composite commodities #
  (All,cc,COMPCOM)(All,j,IND_JP)  p0ccom(cc,j) =
     Sum(i,COM_JP, CCPROD(i,j,cc)*S0C(i,j)*[p0dom(i) - a0com(i) - a0ci(i,j)]);

Equation E_x0dom_JP # Outputs of domestic coms from joint-product ind.s #
  (All,i,COM_JP) x0dom(i) = Sum(j,IND_JP, B0CI(i,j)*x0ci(i,j));

Equation E_x0dom_UP # Outputs of domestic commodities from unique-product inds #
  (All,i,COM_UP) x0dom(i) = Sum(j,IND_UP,B0CI(i,j)*[z(j) - a0ind(j) -a0com(i)]);

! 18.8b Demands by industries for intermediate inputs !

Equation E_x1csi # Demands for intermediate inputs #
  (All,i,COM)(All,s,SOURCE)(All,j,IND) x1csi(i,s,j) = z(j)
  - SIGMA1(i) *{ p1csi(i,s,j) -Sum(t,SOURCE,SOURCE_SHR1(i,t,j)*p1csi(i,t,j)) }
  + a1(j) + a1ci(i,j) +  a1csi(i,s,j)
  - SIGMA1(i) *{ a1csi(i,s,j) -Sum(t,SOURCE,SOURCE_SHR1(i,t,j)*a1csi(i,t,j)) }
  - { SOURCEDOM(s) - SOURCE_SHR1(i,"dom",j) }*twist_src(i)
  + { 1 - SOURCEDOM(s) }*TRANSERVDUM(i)*f1_trans(j)
  + { 1 - SOURCEDOM(s) }*COMMUNICDUM(i)*f1_commun(j);

Equation E_f1_trans
  # Trans. serv. imports related to industry activity, imports and exports #
  (All,j,IND)(All,r,TRANSERV) {x1csi(r,"imp",j) - f1_trans_a(j) - f1_trans_u }

  = (1-DUMMY_WAT(j))*C_SPECIMP1(j)*z(j)
  + (1-DUMMY_WAT(j))*C_SPECIMP2(j)*(1/IMPORTS_IND(j))
                        *Sum(i,COM,BAS1(i,"imp",j)*x1csi(i,"imp",j))
  + [DUMMY_WAT(j)+(1-DUMMY_WAT(j))*C_SPECIMP3(j)]
       *(1/EXPORTS_IND(j))*Sum(i,COM,[MAKE(i,j)*BAS4(i)/SALES(i)]*x4(i));

Equation E_f1_commun
  # Communications imports related to industry activity, imports and exports #
  (All,j,IND)(All,r,COMMUNICAT)
  {x1csi(r,"imp",j) - f1_commun_a(j) - f1_commun_u } = C_SPECIMP1(j)*z(j)
  + C_SPECIMP2(j)*(1/IMPORTS_IND(j))
                                                     *Sum(i,COM,BAS1(i,"imp",j)*x1csi(i,"imp",j))
  +(C_SPECIMP3(j)/EXPORTS_IND(j))
                                            *Sum(i,COM,[MAKE(i,j)*BAS4(i)/SALES(i)]*x4(i));

Equation E_x1oct # Industry demands for other cost tickets #
  (All,j,IND) x1oct(j) = z(j) + a1(j) + a1oct(j);

Equation E_del_p1oct
  # Allows P1OCT to move with the CPI: del form allows P1OCT to pass thru 0 #
  (All,j,IND)
  del_p1oct(j) = LEV_CPI*del_f1oct(j) + {(LEV_CPI*LEV_F1OCT(j))/100}*xi3;

! 18.8c Demands by industries for primary factors !

Equation E_labind # Industry demands for composite units of labour #
  (All,j,IND) labind(j) = z(j)
  - SIGMA1LAB(j)*{ p1lab(j) - SSTAR1FAC("labour",j) *p1lab(j)
  -  SSTAR1FAC("capital",j)*p1cap(j) - SSTAR1FAC("land",j)*p1land(j)}
  + a1(j) + (a1prim(j) + a1primgen) + (a1lab(j) +a1labgen)
  - SIGMA1LAB(j)*{  [a1lab(j)+a1labgen]
             - SSTAR1FAC("labour",j)*[a1lab(j)+a1labgen]
             - SSTAR1FAC("capital",j)*a1cap(j) - SSTAR1FAC("land",j)*a1land(j)  }
                                                                + SOURCE_SHRLK("capital",j)*twistlk(j);

Equation E_cap_at_t # Industry demands for capital #
  (All,j,IND) cap_at_t(j) = z(j)
  - SIGMA1CAP(j)*{p1cap(j) - SSTAR1FAC("labour",j)*p1lab(j)
    -  SSTAR1FAC("capital",j)*p1cap(j) - SSTAR1FAC("land",j)*p1land(j) }
  + a1(j) + (a1prim(j) + a1primgen) + a1cap(j)
  - SIGMA1CAP(j)*{a1cap(j) -SSTAR1FAC("labour",j)*[a1lab(j) + a1labgen]
    -  SSTAR1FAC("capital",j)*a1cap(j) - SSTAR1FAC("land",j)*a1land(j) }
    - SOURCE_SHRLK("labour",j)*twistlk(j);

Equation E_p1land # Industry demands for agricultural land #
  (All,j,IND) n(j) = z(j)
  - SIGMA1LND(j)*{p1land(j) - SSTAR1FAC("labour",j)*p1lab(j)
    -  SSTAR1FAC("capital",j)*p1cap(j) - SSTAR1FAC("land",j)*p1land(j)}
  + a1(j) + (a1prim(j) + a1primgen) + a1land(j)
  - SIGMA1LND(j) *{a1land(j) -SSTAR1FAC("labour",j) *[a1lab(j) +a1labgen]
    -  SSTAR1FAC("capital",j)*a1cap(j) - SSTAR1FAC("land",j)*a1land(j)};

Equation E_x1laboi # Demand for labour by industry and broad occupation #
  (All,qq,OCC)(All,j,IND)  x1laboi(qq,j) = labind(j)
  -SIGMA1LABOCC(qq,j)*{p1laboi(qq,j)-
                                 Sum(m,OCC,SSTAR1LABOCC(m,j)*p1laboi(m,j))}
  + a1laboi(qq,j)  -SIGMA1LABOCC(qq,j)*{a1laboi(qq,j)-
                                  Sum(m,OCC,SSTAR1LABOCC(m,j)*a1laboi(m,j))}
  -Sum(m,OCC,(LABOCCIND(m,j)/(TINY+LABIND_J(j)))*a1laboi(m,j));

Equation E_lambda # Total demands for labour (hours) by broad occupation #
  (All,m,OCC) LABOCC(m)*lambda(m) = Sum(j,IND,HOURS(m,j)*x1laboi(m,j));

Equation E_p1lab # Prices to industries of  composite units of labour #
  (All,j,IND) [LABIND_J(j)
           + If(LABIND_J(j) EQ 0, Sum(m,OCC,TINY) )]*p1lab(j) =
  Sum(m,OCC, [LABOCCIND(m,j) + If(LABIND_J(j) EQ 0, TINY)]*p1laboi(m,j));

Equation E_p1laboi # Unit cost of labour of type m in industry j #
  (All,m,OCC)(All,j,IND) p1laboi(m,j) = wlaboi(m,j) + powpayroll(j);

Equation E_wlaboi
  # Allows flexibility in setting of money wages by industry and occupation #
  (All,j,IND)(All,m,OCC) wlaboi(m,j) = xi3 + fwage + fwagei(j) + fwageo(m)
  + fwageoi(m,j) + Sum(rr,INDSAGGTO20:INDTOIA20(j)=$POS(rr),fwage_s(rr) );

Equation E_wlab_s # Sectoral wage rates, excludes payroll taxes #
  (All,jj,INDSAGGTO20) (Sum[j,IND:
  INDTOIA20(j)=$POS(jj),LABIND_J(j)/POW_PAYROLL(j)] + TINY)*wlab_s(jj) =
    Sum(j,IND:INDTOIA20(j)=$POS(jj),
      (LABIND_J(j)/POW_PAYROLL(j) + TINY)*(p1lab(j)-powpayroll(j)) );

Equation E_labind_s # Sectoral employment #
  (All,jj,INDSAGGTO20)
  (Sum[j,IND:INDTOIA20(j)=$POS(jj),LABIND_J(j)/POW_PAYROLL(j)] + TINY)
  *labind_s(jj) =Sum(j,IND:INDTOIA20(j)=$POS(jj),
                                                       (LABIND_J(j)/POW_PAYROLL(j))*labind(j) );

Equation E_wagebill_s # Sectoral wagebills, excludes payroll taxes #
  (All,jj,INDSAGGTO20) wagebill_s(jj) = wlab_s(jj) +labind_s(jj);

Equation E_ave_wage # Average nominal wage rate, excludes payroll tax #
   (Sum[j,IND,LABIND_J(j)/POW_PAYROLL(j)] + TINY)*ave_wage =
  Sum(j,IND, (LABIND_J(j)/POW_PAYROLL(j) + TINY)*(p1lab(j)-powpayroll(j)) );

! 18.8d Demands for inputs to capital creation !

Equation E_x2csi # Demands for inputs to capital creation #
  (All,i,COM)(All,s,SOURCE)(All,j,IND) x2csi(i,s,j) = y(j)
  - SIGMA2(i)*{p2csi(i,s,j) - Sum(t,SOURCE,SOURCE_SHR2(i,t,j)*p2csi(i,t,j))}
  + a2ind(j) + a2ci(i,j) +  a2csi(i,s,j)
  - SIGMA2(i)*{a2csi(i,s,j) - Sum(t,SOURCE,SOURCE_SHR2(i,t,j)*a2csi(i,t,j))}
  - {SOURCEDOM(s) - SOURCE_SHR2(i,"dom",j)}*twist_src(i);

Equation E_pi
  # Asset prices of units of capital by industry: costs of capital creation #
  (All,j,IND)
  (VINVEST(j) +TINY)*pi(j) = Sum(i,COM,Sum(s,SOURCE, PURCHVAL2(i,s,j)
  *[p2csi(i,s,j) + a2ind(j) + a2ci(i,j) + a2csi(i,s,j)]));

! 18.8e Household demands for commodities !

Equation  E_x3 # Household demands for composite commodities #
  (All,i,COM)
  x3(i) - q = EPS(i)*(c-q) + Sum(j,COM,ETA(i,j)*p3(j)) + a3com(i) - ave_a3com;

Equation E_ave_a3com # Average value of a3com #
  ave_a3com = Sum(k,COM, S3COM(k)*a3com(k));

Equation E_deltapc
  # Movements in marginal budget shares in the linear expenditure system #
  (All,i,COM) deltapc(i) = a3com(i) - Sum(k,COM, DELTA(k)*a3com(k));

Equation E_d_gamma
  # Movements in subsistence variables in the linear expenditure system #
  (All,i,COM) d_gamma(i) = [1 + EPS(i)/FRISCH]*[a3com(i)-ave_a3com];

Equation E_aa3com # Cont. of change in hhld pref for i to sales of domestic i #
  (All,i,COM) aa3com(i) = {BAS3(i,"dom")/(SALES(i)+TINY)}*a3com(i);

Equation E_x3cs # Household demands by commodity and source #
  (All,i,COM)(All,s,SOURCE) x3cs(i,s) = x3(i)
  - SIGMA3(i)*[p3cs(i,s) - Sum(t,SOURCE,SOURCE_SHR3(i,t)*p3cs(i,t))]
  + a3cs(i,s)
  - SIGMA3(i)*[a3cs(i,s) - Sum(t,SOURCE,SOURCE_SHR3(i,t)*a3cs(i,t))]
  - Sum(t,SOURCE,SOURCE_SHR3(i,t)*a3cs(i,t))
  - {SOURCEDOM(s) - SOURCE_SHR3(i,"dom")}*twist_src(i);

Equation E_p3 # General prices of consumer goods #
  (All,i,COM) p3(i) = Sum(s,SOURCE,SOURCE_SHR3(i,s)*p3cs(i,s));

! 18.8f Demands for exports, export aggregates and foreign-currency
  price indexes !

! Traditional exports: commodity demands, aggregate volume & price index !

Equation E_x4_TRADEXP # Demand functions for exports #
  (All,i,TRADEXP) x4(i) = EXP_ELAST(i)*[pe(i) - fep(i)] + feq(i) + feq_general;

Equation E_tradexpvol # Volume of traditional exports #
  AGGTRADEXP*tradexpvol = Sum(i,TRADEXP, PURCHVAL4(i)*x4(i));

Equation E_xi4trad # Traditional exports, index of foreign currency prices #
  AGGTRADEXP*xi4trad = Sum(i,TRADEXP, PURCHVAL4(i)*pe(i));

! Non-traditional exports: commodity demands, aggregate volume & price index !

Equation E_x4_NTRADEXP # Demands for non-traditional exports #
  (All,i,NTRADEXP) x4(i) = EXP_ELAST_N*[xi4ntrad - fep_ntradi(i)-fep_ntrad]
        + f_ntrad(i) + feq_ntrad + feq_general;

Equation E_ntradexpvol # Volume of non-traditional exports #
  AGGNTRADEXP*ntradexpvol =  Sum(i,NTRADEXP, PURCHVAL4(i)*x4(i));

Equation E_xi4ntrad
  # Non-traditional exports, index of foreign currency prices #
  AGGNTRADEXP*xi4ntrad = Sum(i,NTRADEXP, PURCHVAL4(i)*pe(i));

Equation E_hs_ntrad # Horizontal shift in demand for non-traditional exports #
  hs_ntrad = Sum(i,NTRADEXP,[PURCHVAL4(i)/AGGNTRADEXP]*f_ntrad(i))
                                  + feq_ntrad;

! Tourism exports: commodity demands, aggregate volume and price index !

Equation E_x4_TOURISM # Composition of tourism exports #
  (All,i,TOURISM) x4(i) = EXP_ELAST_T*[xi4tour - fep_touri(i)- fep_tour]
        + f_tour(i) + feq_tour + feq_general;

Equation E_agg_tour # Volume of tourism exports #
  AGGTOURISM*agg_tour = Sum(i,TOURISM, PURCHVAL4(i)*x4(i));

Equation E_xi4tour # Tourism exports, index of foreign currency prices #
  AGGTOURISM*xi4tour = Sum(i,TOURISM,PURCHVAL4(i)*pe(i));

! Special modelling of  export demands for: water transport (C97), transport
services (C99) and communication (C100) !

Equation E_x4_WATERTRAN
  # Water transport exports (C97) related to traditional export volumes #
  (All,i,WATERTRAN) x4(i) = tradexpvol + fx4_wattran;

Equation E_x4_TRANSERV
  # Transp. serv. exports related to air imports, tourism & merchandise trade #
  (All,i,TRANSERV)(All,j,AIRTRAN)
  x4(i) = 0.25*x0imp(j) + 0.25*agg_tour + 0.5*merch_trad  + fx4_transerv;
Equation E_merch_trad # Volume of merchandise trade #
  Sum(r,MERCHANDISE,TOT_T(r))*merch_trad =
                                                                          Sum(i,MERCHANDISE, TOT_T(i)*tt(i));

Equation E_tt # Total trade at the commodity level #
  (All,i,COM) [TOT_T(i) +TINY]*tt(i) = PURCHVAL4(i)*x4(i) + IMPCOST(i)*x0imp(i);

Equation E_x4_COMMUNICAT # Communication exports related to communic. imports #
  (All,i,COMMUNICAT) x4(i) = x0imp(i) + fx4_commun;

! 18.8g Other demands for commodities !

Equation E_x5cs # Government demands for commodities #
  (All,i,COM)(All,s,SOURCE)
  x5cs(i,s) =  SOURCEDOM(s)*f5dom(i) + (1-SOURCEDOM(s))*f5imp(i) + f5gen
  - {SOURCEDOM(s) - SOURCE_SHR5(i,"dom")}*twist_src(i);

! 18.8h Margins usage of commodities !

Equation E_x1marg # Margins on inputs  to producers #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
  x1marg(i,s,j,r) = x1csi(i,s,j) + a1marg(i,s,j,r);

Equation E_x2marg # Margins on inputs to capital creators #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
  x2marg(i,s,j,r) = x2csi(i,s,j) + a2marg(i,s,j,r);

Equation E_x3marg # Margins on flows to households #
  (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
  x3marg(i,s,r) = x3cs(i,s) + a3marg(i,s,r);

Equation E_x4marg # Margins in transferring exports to ports of exit #
  (All,i,COM)(All,r,MARGCOM) x4marg(i,r) = x4(i) + a4marg(i,r);

Equation E_x5marg # Margins on flows to government users #
  (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
  x5marg(i,s,r) = x5cs(i,s) + a5marg(i,s,r);

! 18.8i Supply equals demand for domestic and imported commodities !

Equation E_p0dom_NM # Demand equals supply for non-margin commodities #
  (All,r,NONMARGCOM) SALES(r) *x0dom(r) =
  Sum(j,IND:BAS1(r,"dom",j) ne 0, BAS1(r,"dom",j)*x1csi(r,"dom",j))
  + Sum(j,IND:BAS2(r,"dom",j) ne 0, BAS2(r,"dom",j)*x2csi(r,"dom",j))
  + BAS3(r,"dom")*x3cs(r,"dom") + BAS4(r)*x4(r)
  + BAS5(r,"dom")*x5cs(r,"dom") + 100*P0LEV(r,"dom")*d_x6cs(r,"dom");

Equation E_p0dom_M # Demand equals supply for margin commodities #
  (All,r,MARGCOM) SALES(r)*x0dom(r) =
  Sum(j,IND:BAS1(r,"dom",j) ne 0, BAS1(r,"dom",j)*x1csi(r,"dom",j))
  + Sum(j,IND:BAS2(r,"dom",j) ne 0, BAS2(r,"dom",j)*x2csi(r,"dom",j))
  + BAS3(r,"dom")*x3cs(r,"dom") + BAS4(r)*x4(r)
  + BAS5(r,"dom")*x5cs(r,"dom") + 100*P0LEV(r,"dom")*d_x6cs(r,"dom")
  + Sum(i,COM,Sum(s,SOURCE,
  { Sum(j,IND:MAR1(i,s,j,r) ne 0, MAR1(i,s,j,r)*x1marg(i,s,j,r)) +
    Sum(j,IND:MAR2(i,s,j,r) ne 0, MAR2(i,s,j,r)*x2marg(i,s,j,r)) }))
  + Sum(i,COM,Sum(s,SOURCE, MAR3(i,s,r)*x3marg(i,s,r)))
  + Sum(i,COM, MAR4(i,r)*x4marg(i,r))
  + Sum(i,COM,Sum(s,SOURCE, MAR5(i,s,r)*x5marg(i,s,r)));

Equation E_x0imp # Import volumes by commodity #
  (All,i,COM) [IMPORTS(i)+If(IMPORTS(i) eq 0, TINY)]*x0imp(i) =
  Sum(j,IND:BAS1(i,"imp",j) ne 0, BAS1(i,"imp",j)*x1csi(i,"imp",j))
  + Sum(j,IND:BAS2(i,"imp",j) ne 0, BAS2(i,"imp",j)*x2csi(i,"imp",j))
  + [BAS3(i,"imp")]*x3cs(i,"imp") + BAS5(i,"imp")*x5cs(i,"imp")
  + 100*P0LEV(i,"imp")*d_x6cs(i,"imp");

! 18.8j Zero pure profits in production, importing, exporting and distribution,
and equations for total technical change by industry !

Equation E_p0ind # Price of industry output #
  (All,j,IND) p0ind(j) = Sum(i,COM,H0CI(i,j)*p0dom(i));

Equation E_z # Zero pure profits in production #
  (All,j,IND) p0ind(j) =  a(j) +
  [1/{COSTS(j)-TAX0(j)}]*
  {Sum(i,COM, Sum(s,SOURCE, PURCHVAL1(i,s,j)*p1csi(i,s,j)))
   + Sum(m,OCC, LABOCCIND(m,j)*p1laboi(m,j)) + CAPITAL(j)*p1cap(j)
   + LAND(j)*p1land(j) + 100*ROTHCOST(j)*del_p1oct(j)}
            +[POW_TAX0(j)/(2-POW_TAX0(j))]*powtax0(j);

Equation E_a_JP
  # Total output-augmenting & input-saving tech change, joint-product inds #
  (All,j,IND_JP) a(j) = a0ind(j) + a1(j)
    + Sum(cc,COMPCOM, H0CC(cc,j)*a0ccom(cc,j))
    + Sum(i,COM_JP, H0CI(i,j)*[a0com(i) +a0ci(i,j)])
    + [1.0/{COSTS(j)-TAX0(j)}]*{ Sum(i,COM, TPURCHVAL1(i,j)*a1ci(i,j))
          + Sum(i,COM, Sum(s,SOURCE, PURCHVAL1(i,s,j)*a1csi(i,s,j)))
     + TOTFACIND(j)*[a1prim(j) + a1primgen] + LABIND_J(j)*[a1lab(j) + a1labgen]
         + CAPITAL(j)*a1cap(j) + LAND(j)*a1land(j) + OTHCOST(j)*a1oct(j)
                   + Sum(i,TRANSERV,PURCHVAL1(i,"imp",j)*f1_trans(j) )
                   + Sum(i,COMMUNICAT,PURCHVAL1(i,"imp",j)*f1_commun(j))};

Equation E_a_UP
  # Total output-augmenting & input-saving tech change, unique-product inds #
  (All,j,IND_UP)  a(j) = a0ind(j) + a1(j)
  + Sum(i,COM_UP, H0CI(i,j)*a0com(i))
  + [1.0/{COSTS(j)-TAX0(j)}]*{ Sum(i,COM, TPURCHVAL1(i,j)*a1ci(i,j))
      + Sum(i,COM, Sum(s,SOURCE, PURCHVAL1(i,s,j)*a1csi(i,s,j)))
      + TOTFACIND(j)*[a1prim(j) +a1primgen] + LABIND_J(j)*[a1lab(j) + a1labgen]
      + CAPITAL(j)*a1cap(j) + LAND(j)*a1land(j) + OTHCOST(j)*a1oct(j)
                   + Sum(i,TRANSERV,PURCHVAL1(i,"imp",j)*f1_trans(j))
                   + Sum(i,COMMUNICAT,PURCHVAL1(i,"imp",j)*f1_commun(j))};

Equation E_p0imp # Zero pure profits in importing #
  (All,i,COM) p0imp(i) = pm(i) - phi + powtaxm(i);

Equation E_p0
  # Basic prices for domestic and imported goods in a single matrix variable #
  (All,i,COM)(All,s,SOURCE)
  p0(i,s) = SOURCEDOM(s)*p0dom(i) + (1-SOURCEDOM(s))*p0imp(i);

Equation E_fpdm # Ratio of basic prices: domestic to import #
  (All,i,COM) fpdm(i) = p0dom(i) - p0imp(i);

! Zero pure profits in distributing: purchasers' prices of commodities sold for
  intermediate use, capital creation, consumption, export, and government use !

Equation E_p1csi # Purchasers' prices of intermediate inputs #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  [PURCHVAL1(i,s,j) + TINY]*p1csi(i,s,j) =
  [BAS1(i,s,j) + TAX1(i,s,j)]*[p0(i,s) + powtax1(i,s,j)]
  + Sum(r,MARGCOM, MAR1(i,s,j,r)*[p0dom(r) + a1marg(i,s,j,r)]);

Equation E_p2csi # Purchasers' prices of inputs to capital creation #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  [PURCHVAL2(i,s,j) + TINY]*p2csi(i,s,j) =



212 Dynamic General Equilibrium Modelling for Forecasting and Policy

  [BAS2(i,s,j) + TAX2(i,s,j)]*[p0(i,s) + powtax2(i,s,j)]
  + Sum(r,MARGCOM, MAR2(i,s,j,r)*[p0dom(r) + a2marg(i,s,j,r)]);

Equation E_p3cs   # Purchasers' prices of consumer goods #
  (All,i,COM)(All,s,SOURCE) [PURCHVAL3(i,s) + 10*TINY]*p3cs(i,s) =
  [BAS3(i,s) + TAX3(i,s)+TINY]*[p0(i,s) + powtax3(i,s)]
  + Sum(r,MARGCOM, [MAR3(i,s,r)+TINY]*[p0dom(r) + a3marg(i,s,r)]);

Equation E_p5cs # Purchasers' prices of commodities sold to government #
  (All,i,COM)(All,s,SOURCE) [PURCHVAL5(i,s) + TINY]*p5cs(i,s) =
  [BAS5(i,s) + TAX5(i,s)]*[p0(i,s) + powtax5(i,s)]
  + Sum(r,MARGCOM, MAR5(i,s,r)*[p0dom(r) + a5marg(i,s,r)]);

Equation E_pe # Purchasers' prices (i.e., f.o.b. prices) of exports #
  (All,i,COM) [PURCHVAL4(i) + TINY]*[pe(i) - phi] =
  [BAS4(i) + TAX4(i)]*[p0dom(i) + powtax4(i)]
  + Sum(r,MARGCOM, MAR4(i,r)*[p0dom(r) + a4marg(i,r)]);

! 18.8k Indirect taxes !

! Indirect tax revenue by commodity, source and user and by industry !

Equation E_del_tax1
  # Revenue from genuine & phantom indirect taxes on intermediate input flows #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  100*del_tax1(i,s,j) = TAX1(i,s,j)*[p0(i,s) + x1csi(i,s,j)]
  + [TAX1(i,s,j) + BAS1(i,s,j)]*powtax1(i,s,j);

Equation E_del_tax2
  # Revenue from genuine & phantom indirect taxes on flows to capital creation #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  100*del_tax2(i,s,j) = TAX2(i,s,j)*[p0(i,s) + x2csi(i,s,j)]
  + [TAX2(i,s,j) + BAS2(i,s,j)]*powtax2(i,s,j);

Equation E_del_tax3
  # Revenue from genuine and phantom indirect taxes on flows to households #
  (All,i,COM)(All,s,SOURCE) 100*del_tax3(i,s) = TAX3(i,s)*[p0(i,s) + x3cs(i,s)]
  + [TAX3(i,s) + BAS3(i,s)]*powtax3(i,s);

Equation E_del_tax4
  # Revenue from genuine and phantom indirect taxes on exports #
  (All,i,COM) 100*del_tax4(i) = TAX4(i)*[p0dom(i) + x4(i)]
  + [TAX4(i) + BAS4(i)]*powtax4(i);

Equation E_del_tax5
  # Revenue from genuine and phantom indirect taxes on flows to government #
  (All,i,COM)(All,s,SOURCE) 100*del_tax5(i,s) = TAX5(i,s)*[p0(i,s) + x5cs(i,s)]

  + [TAX5(i,s) + BAS5(i,s)]*powtax5(i,s);

Equation E_del_tax0 # Revenue from genuine and phantom taxes on production #
  (All,j,IND) 100*del_tax0(j) = TAX0(j)*(p0ind(j)+x0ind(j))
  + [TAX0(j) + COSTS(j)]*powtax0(j);

! Specification of powers of taxes by commodity, source and user !

Equation E_powtax1 # Power of tax on sales to intermediate users #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  powtax1(i,s,j) = powtaxgg(i,s) + fpowtax1gg(i,s)
  + SOURCEDOM(s)*powtax4sph(i) + powtaxphph(i,s) + fpowtax1phph(i,s);

Equation E_powtax2 # Power of tax on sales to capital creators #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  powtax2(i,s,j) = powtaxgg(i,s) + fpowtax2gg(i,s)
  + SOURCEDOM(s)*powtax4sph(i) + powtaxphph(i,s) + fpowtax2phph(i,s);

Equation E_powtax3 # Power of tax on sales to consumers #
  (All,i,COM)(All,s,SOURCE)
  powtax3(i,s) = powtaxgg(i,s) + fpowtax3g(i,s) + powtax3vg(i) +fpowtax3gu
  + SOURCEDOM(s)*powtax4sph(i) + powtaxphph(i,s) + fpowtax3ph(i,s);

Equation E_powtax4 # Power of tax on exports #
  (All,i,COM) powtax4(i) = powtax4g(i) + powtax4sph(i) + powtax4ph(i);

Equation E_powtax5 # Power of tax on sales to government users #
  (All,i,COM)(All,s,SOURCE) powtax5(i,s) = powtaxgg(i,s) + fpowtax5g(i,s)
  + SOURCEDOM(s)*powtax4sph(i) + powtaxphph(i,s) + fpowtax5ph(i,s);

Equation E_powtax0 # Power of tax on production #
  (All,j,IND) powtax0(j) = powtax0g(j) + powtax0ph(j);

! Collections of genuine and phantom indirect taxes on commodity sales by
  souce and user and on industry production !

Equation E_del_tax1g
  # Revenue from genuine indirect taxes on intermediate input flows #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  100*del_tax1g(i,s,j) = TAX1G(i,s,j)*[p0(i,s) + x1csi(i,s,j)]
  + [TAX1G(i,s,j) +BAS1(i,s,j)]*[powtaxgg(i,s) + fpowtax1gg(i,s)];

Equation E_del_tax1ph
  # Revenue from phantom indirect taxes on intermediate input flows #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  del_tax1ph(i,s,j) = del_tax1(i,s,j) - del_tax1g(i,s,j);

Equation E_del_tax2g
  # Revenue from genuine indirect taxes on flows to capital creation #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  100*del_tax2g(i,s,j) = TAX2G(i,s,j)*[p0(i,s) + x2csi(i,s,j)]
  + [TAX2G(i,s,j) + BAS2(i,s,j)]*[powtaxgg(i,s) + fpowtax2gg(i,s)];

Equation E_del_tax2ph
  # Revenue from phantom indirect taxes on flows to capital creation #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)
  del_tax2ph(i,s,j) = del_tax2(i,s,j) - del_tax2g(i,s,j);

Equation E_del_tax3g
  # Revenue from genuine indirect taxes on flows to households #
  (All,i,COM)(All,s,SOURCE)
  100*del_tax3g(i,s) = TAX3G(i,s)*[p0(i,s) + x3cs(i,s)]
  + [TAX3G(i,s) + BAS3(i,s)]
  *[powtaxgg(i,s) + fpowtax3g(i,s) + powtax3vg(i) + fpowtax3gu];

Equation E_del_tax3ph # Revenue from phantom indirect taxes on flows to hhlds #
  (All,i,COM)(All,s,SOURCE) del_tax3ph(i,s) = del_tax3(i,s) - del_tax3g(i,s);

Equation E_del_tax4g # Revenue from genuine indirect taxes on exports #
  (All,i,COM) 100*del_tax4g(i) = TAX4G(i)*[p0dom(i) + x4(i)]
                + [TAX4G(i) + BAS4(i)]*powtax4g(i);

Equation E_del_tax4ph # Revenue from phantom indirect taxes on exports #
  (All,i,COM) del_tax4ph(i) = del_tax4(i) - del_tax4g(i);

Equation E_del_tax5g
  # Revenue from genuine indirect taxes on flows to government#
  (All,i,COM)(All,s,SOURCE)
  100*del_tax5g(i,s) = TAX5G(i,s)*[p0(i,s) + x5cs(i,s)]
  + [TAX5G(i,s) + BAS5(i,s)]*[powtaxgg(i,s) +fpowtax5g(i,s)];

Equation E_del_tax5ph
  # Revenue from phantom indirect taxes on flows to government #
  (All,i,COM)(All,s,SOURCE) del_tax5ph(i,s) = del_tax5(i,s) - del_tax5g(i,s);

Equation E_del_tax0g # Revenue from genuine taxes on production #
  (All,j,IND) 100*del_tax0g(j) = TAX0G(j)*(p0ind(j)+x0ind(j))
                                        + [TAX0G(j) + COSTS(j)]*powtax0g(j);

Equation E_del_tax0ph # Revenue from phantom taxes on production #
  (All,j,IND) del_tax0ph(j) = del_tax0(j) - del_tax0g(j);

! Collection of tariff revenue and specification of tariff rates !

Equation E_del_taxm # Revenue from tariffs on imports #
  (All,i,COM)
  100*del_taxm(i) = TARIFF(i)*[pm(i) - phi + x0imp(i)] + IMPORTS(i)*powtaxm(i);

Equation E_powtaxm # Power of tariffs on imports #
  (All,i,COM) powtaxm(i) = powtaxmo(i) +fpowtaxm(i) + ffpowtaxm;

Equation E_powtaxmo
  # Power of tariffs, used in transferring forecasts to policy simulations #
  (All,i,COM) powtaxmo(i) = fpowtaxmo(i)
                        + {[LEV_PTAXM(i)-1.0]/[LEV_PTAXM(i)]}*iacrate(i);

Equation E_ft2 # Tariff rates #
  (All,i,COM) iacrate(i) = t2_star + ft2(i);

! Miscellaneous tax aggregates !

Equation E_taxrev1
  # Aggregate revenue from indirect taxes on flows to intermediate users #
  AGGTAX1*taxrev1 = 100*Sum(i,COM, Sum(s,SOURCE, Sum(j,IND, del_tax1(i,s,j))));

Equation E_taxrev1g
  #Aggregate revenue from genuine indirect taxes on flows to intermediate users#
  AGGTAX1G* taxrev1g =
  100*Sum(i,COM, Sum(s,SOURCE, Sum(j,IND, del_tax1g(i,s,j))));

Equation E_taxrev2
  # Aggregate revenue from indirect taxes on flows to capital creation #
  AGGTAX2*taxrev2 =  100*Sum(i,COM, Sum(s,SOURCE, Sum(j,IND, del_tax2(i,s,j))));

Equation E_taxrev2g
  # Aggregate revenue from genuine indirect taxes on flows to capital creation #
  AGGTAX2G*taxrev2g =
  100*Sum(i,COM, Sum(s,SOURCE, Sum(j,IND, del_tax2g(i,s,j))));

Equation E_taxrev3 # Aggregate revenue from indirect taxes on flows to hholds #
  AGGTAX3*taxrev3 = 100*Sum(i,COM, Sum(s,SOURCE, del_tax3(i,s)));

Equation E_taxrev3g
  # Aggregate revenue from genuine indirect taxes on flows to households #
  AGGTAX3G*taxrev3g = 100* Sum(i,COM,Sum(s,SOURCE,del_tax3g(i,s)));

Equation E_taxrev3gc
  # Collection of genuine consumption taxes by commodity #
  (All,i,COM) Sum(s,SOURCE, TAX3G(i,s) + TINY)*taxrev3gc(i) =
                100*Sum(s,SOURCE, del_tax3g(i,s));

Equation E_del_tot_tax4
  # Change in aggregate revenue from indirect taxes on exports #
  del_tot_tax4 =  Sum(i,COM, del_tax4(i));

Equation E_del_tot_tax4g
  # Change in aggregate revenue from genuine indirect taxes on exports #
  del_tot_tax4g = Sum(i,COM, del_tax4g(i));

Equation E_del_tot_tax4ph
  # Change in aggregate revenue from phantom indirect taxes on exports #
  del_tot_tax4ph = del_tot_tax4 - del_tot_tax4g;

Equation E_del_tot_tax5
  # Change in aggregate revenue from indirect taxes on flows to government #
  del_tot_tax5 =  Sum(i,COM, Sum(s,SOURCE, del_tax5(i,s)));

Equation E_del_tot_tax5g
  # Change in agg revenue from genuine indirect taxes on flows to government #
  del_tot_tax5g =  Sum(i,COM, Sum(s,SOURCE, del_tax5g(i,s)));

Equation E_del_tot_tax0
  # Change in aggregate revenue from taxes on production #
  del_tot_tax0 =  Sum(j,IND, del_tax0(j));

Equation E_del_tot_tax0g
  # Change in aggregate revenue from genuine taxes on production #
  del_tot_tax0g =  Sum(j,IND, del_tax0g(j));

Equation E_taxrevm # Aggregate tariff revenue #
  AGGTAXM*taxrevm = 100*Sum(i,COM, del_taxm(i));

Equation E_taxindg # Aggregate value of genuine indirect taxes #
  AGGTAXG*taxindg = AGGTAX1G*taxrev1g + AGGTAX2G*taxrev2g
                 + AGGTAX3G*taxrev3g  + 100*del_tot_tax4g + 100*del_tot_tax5g
                 + 100*del_tot_tax0g + AGGTAXM*taxrevm;

! 18.8l Macro variables !

! Expenditure components of GDP and related price indexes !

Equation E_c # Household expenditure #
  c = xi3 + cr;

Equation E_phi # Index of prices of consumption goods #
  AGGCON*xi3 = Sum(i,COM, Sum(s,SOURCE, PURCHVAL3(i,s)*p3cs(i,s)));

Equation E_in # Total nominal investment #
  in = xi2 + ir;

Equation E_xi2 # Investment price index #
  AGGINV*xi2 = Sum(j,IND, VINVEST(j)*pi(j));

Equation E_r_inv_cap_u # Total real investment #
  AGGINV*ir = Sum(j,IND, VINVEST(j)*y(j));

Equation E_othnom # Aggregate nominal value of government demands #
  othnom = xi5 + othreal;

Equation E_xi5 # Index of prices of goods sold to government #
  AGGOTH*xi5 = Sum(i,COM, Sum(s,SOURCE, PURCHVAL5(i,s)*p5cs(i,s)));

Equation E_f5gen # Aggregate real government demands #
  AGGOTH*othreal = Sum(i,COM, Sum(s,SOURCE, PURCHVAL5(i,s)*x5cs(i,s)));

Equation E_expvald # Exports (fob) in domestic currency #
  expvald = xi4 + expvol;

Equation E_xi4 # Index of domestic currency fob export prices #
  AGGEXP*[xi4+phi] = Sum(i,COM, PURCHVAL4(i)*(pe(i)));

Equation E_expvol # Export volume index #
  AGGEXP*expvol = Sum(i,COM, PURCHVAL4(i)*x4(i));

Equation E_impvald # Imports (cif) in domestic currency #
  impvald = xim + impvol;

Equation E_xim # Index of cif prices of imports, domestic currency #
  AGGIMP*xim = Sum(i,COM, IMPCOST(i)*(pm(i) - phi));

Equation E_impvol # Import volume index #
  AGGIMP*impvol = Sum(i,COM, IMPCOST(i)*x0imp(i));

Equation E_d_xi6
  # Effect on the value of inventory accumulation of price changes #
  100*d_xi6 = Sum(i,COM, Sum(s,SOURCE, BAS6(i,s)*p0(i,s)));

Equation E_d_inventreal # Aggregate real inventory accumulations #
  d_inventreal = Sum(i,COM, Sum(s,SOURCE, P0LEV(i,s)*d_x6cs(i,s)));

Equation E_gne # Nominal GNE #
  gne = xigne + gner;

Equation E_xigne # Price index for GNE #
  [AGGCON + AGGINV + AGGOTH + AGGINVENT]*xigne =
    AGGCON*xi3 + AGGINV*xi2 + AGGOTH*xi5 + 100*d_xi6;

Equation E_gner # Real GNE #
  [AGGCON + AGGINV + AGGOTH + AGGINVENT]*gner =
                  AGGCON*cr + AGGINV*ir + AGGOTH*othreal + 100*d_inventreal;

Equation E_gdpexp # Nominal GDP, expenditure side #
  gdpexp = xigdp + gdpreal;

Equation E_xigdp # Price index for GDP, expenditure side #
  GDPEX*xigdp = AGGCON*xi3 + AGGINV*xi2 + AGGOTH*xi5 + 100*d_xi6
                                        + AGGEXP*xi4 - AGGIMP*xim;

Equation E_gdpreal # Real GDP, expenditure side #
  GDPEX*gdpreal = AGGCON*cr + AGGINV*ir + AGGOTH*othreal + 100*d_inventreal
                                        + AGGEXP*expvol - AGGIMP*impvol;

! The income measure of GDP and its components !

Equation E_labrev # Aggregate costs of labour #
  AGGLAB*labrev =
        Sum(j,IND, Sum(m,OCC,LABOCCIND(m,j)*[p1laboi(m,j)+x1laboi(m,j)]));

Equation E_caprev # Aggregate payments to capital #
  AGGCAP*caprev = Sum(j,IND, CAPITAL(j)*[p1cap(j) + cap_at_t(j)]);

Equation E_lndrev # Aggregate payments to land #
  AGGLND*lndrev = Sum(j,IND, LAND(j)*[p1land(j) + n(j)]);

Equation E_octrev # Aggregate other cost ticket payments #
  AGGOCT*octrev =
        Sum(j,IND, OTHCOST(j)*x1oct(j) + ROTHCOST(j)*100.0*del_p1oct(j));

Equation E_taxind # Aggregate value of indirect taxes #
  AGGTAX*taxind = AGGTAX1*taxrev1 + AGGTAX2*taxrev2 + AGGTAX3*taxrev3
  + 100*del_tot_tax4 + 100*del_tot_tax5 +100*del_tot_tax0 + AGGTAXM*taxrevm;

Equation E_gdpinc # Nominal GDP from income side #
  GDPIN*gdpinc = AGGLND*lndrev + AGGCAP*caprev + AGGLAB*labrev
                         + AGGOCT*octrev + AGGTAX*taxind;

Equation E_gdprealinc # Real GDP from the income side #
  GDPIN*gdprealinc = -Sum(j,IND_JP,{COSTS(j)-TAX0(j)}*a(j))
                - Sum(j,IND_UP,{COSTS(j)-TAX0(j)}*a(j))
    - Sum(i,MARGCOM,Sum(ii,COM,Sum(s,SOURCE,
          { Sum(j,IND:MAR1(ii,s,j,i) ne 0, MAR1(ii,s,j,i)*a1marg(ii,s,j,i)) +
            Sum(j,IND:MAR2(ii,s,j,i) ne 0, MAR2(ii,s,j,i)*a2marg(ii,s,j,i)) }
          +  MAR3(ii,s,i)*a3marg(ii,s,i) +  MAR5(ii,s,i)*a5marg(ii,s,i)  )
          +  MAR4(ii,i)*a4marg(ii,i))  )
   - Sum(i,COM,Sum[j,IND,TPURCHVAL2(i,j)*

   { a2ind(j) + a2ci(i,j) + Sum(s,SOURCE,SOURCE_SHR2(i,s,j)*a2csi(i,s,j))}])
         + Sum(m,OCC,Sum(j,IND,LABOCCIND(m,j)*x1laboi(m,j)))
         + Sum(j,IND, CAPITAL(j)*cap_at_t(j))
         +  Sum(j,IND,LAND(j)*n(j))
  + Sum(j,IND, OTHCOST(j)*x1oct(j))
  + Sum(i,COM,TARIFF(i)*x0imp(i))
  + Sum(j,IND,TAX0(j)*x0ind(j))
  + Sum(i,COM,Sum(s,SOURCE, Sum(j,IND,TAX1(i,s,j)*x1csi(i,s,j))))
  + Sum(i,COM,Sum(s,SOURCE, Sum(j,IND,TAX2(i,s,j)*x2csi(i,s,j))))
  + Sum(i,COM,Sum(s,SOURCE, TAX3(i,s)*x3cs(i,s)))
  + Sum(i,COM, TAX4(i)*x4(i))
  + Sum(i,COM,Sum(s,SOURCE, TAX5(i,s)*x5cs(i,s)));

Equation E_a_gdp # Contribution of technical change to GDP growth #
  GDPIN*a_gdp = -Sum(j,IND_JP,{COSTS(j)-TAX0(j)}*a(j))
                - Sum(j,IND_UP,{COSTS(j)-TAX0(j)}*a(j))
    - Sum(i,MARGCOM,Sum(ii,COM,Sum(s,SOURCE,
          { Sum(j,IND:MAR1(ii,s,j,i) ne 0, MAR1(ii,s,j,i)*a1marg(ii,s,j,i)) +
            Sum(j,IND:MAR2(ii,s,j,i) ne 0, MAR2(ii,s,j,i)*a2marg(ii,s,j,i)) }
          +  MAR3(ii,s,i)*a3marg(ii,s,i) +  MAR5(ii,s,i)*a5marg(ii,s,i)  )
          +  MAR4(ii,i)*a4marg(ii,i))  )
   - Sum(i,COM,Sum[j,IND,TPURCHVAL2(i,j)*
   { a2ind(j) + a2ci(i,j) + Sum(s,SOURCE,SOURCE_SHR2(i,s,j)*a2csi(i,s,j))}]);

Equation E_ave_aprim
  # Average primary-factor-saving technical change across industries #
  Sum(j,IND,TOTFACIND(j))*ave_aprim
                = Sum(j,IND,TOTFACIND(j)*(a1prim(j)+a1primgen));

! Miscellaneous trade-related macro variables !

Equation E_del_b # Balance of trade in billions of domestic dollars #
  100000*del_b = AGGEXP*expvald - AGGIMP*impvald;

Equation E_del_bt_gdp # Change in bal of trade as percent of GDP #
  del_bt_gdp = (100000/GDPEX)*del_b - [(AGGEXP - AGGIMP)/GDPEX]*gdpexp;

Equation E_realdev # Real devaluation #
  realdev =  xim - xigdp;

Equation E_toft # Terms of trade #
  toft = xi4 - xim;

Equation E_ximp0
  # Index of landed-duty-paid prices of imports, domestic currency #
  [AGGIMP + AGGTAXM]*ximp0 = Sum(i,COM, IMPORTS(i)*p0imp(i));

Equation E_expvalf # Exports fob in foreign currency #
  expvalf = expvald + phi;

Equation E_tradexpvalf # Foreign currency value of traditional exports #
  AGGTRADEXP*tradexpvalf = Sum(i,TRADEXP, PURCHVAL4(i)*[pe(i) + x4(i)]);

Equation E_ntradexpvalf # Foreign currency value of non-traditional exports #
  AGGNTRADEXP*ntradexpvalf = Sum(i,NTRADEXP, PURCHVAL4(i)*[pe(i) +x4(i)]);

Equation E_impvalf # Imports cif in foreign currency #
  impvalf = impvald + phi;

! Wage rates and measures of labour, capital and land inputs !

Equation E_real_wage_c # Economy-wide real wage rate for consumers #
  (AGGLAB-COL_PAYRTOT)*[real_wage_c + xi3] =
  Sum(m,OCC, Sum(j,IND, (LABOCCIND(m,j)/POW_PAYROLL(j))*wlaboi(m,j)));

Equation E_real_wage_e # Economy-wide wage rate, deflator is the price for GDP #
   real_wage_e =
  (1/AGGLAB)*Sum(m,OCC, Sum(j,IND, LABOCCIND(m,j)*p1laboi(m,j))) - xigdp;

Equation E_real_wage_e2
  # Economy-wide wage rate, deflator is the price of primary factors #
  real_wage_e2 =
  (1/AGGLAB)*Sum(m,OCC, Sum(j,IND, LABOCCIND(m,j)*p1laboi(m,j))) - xifac;

Equation E_xifac # Price of primary factors #
  [AGGLAB + AGGCAP + AGGLND]*xifac = AGGLAB*(labrev - emp_c_wgts)
                       + AGGCAP*(caprev - k_r_wgts) + AGGLND*(lndrev - lndr);

Equation E_emp_c_wgts # Aggregate employment, labour cost weights #
  (AGGLAB)*emp_c_wgts = Sum(j,IND, (LABIND_J(j))*labind(j));

Equation E_fwage # Aggregate employment, hours #
  HOURSTOT*emp_hours = Sum(m,OCC,Sum(j,IND, HOURS(m,j)*x1laboi(m,j)));

Equation E_k_r_wgts # Aggregate capital, rental weights #
  AGGCAP*k_r_wgts = Sum(j,IND, CAPITAL(j)*cap_at_t(j));

Equation E_k_a_wgts # Aggregate capital, asset value weights #
  Sum(j,IND,VCAP_AT_TM(j))* k_a_wgts = Sum(j,IND, VCAP_AT_TM(j)*cap_at_t(j));

Equation E_lndr # Aggregate land, rental weights #
  AGGLND*lndr = Sum(j,IND,LAND(j)*n(j));

! Ratios of components of GDP !

Equation  E_othreal
  # Ratio of real private comsumption to real government consumption #

  r_cr_othreal = cr - othreal;

Equation E_r_cr_gner # Ratio of real consumption to real GNE #
  r_cr_gner = cr - gner;

Equation E_ir # Ratio of real investment to real GNE #
  r_ir_gner = ir - gner;

Equation E_r_othr_gner # Ratio of real government consumption to real GNE #
  r_othr_gner = othreal - gner;

Equation E_r_cr_ir # Ratio of real private consumption to real investment #
  r_cr_ir = cr - ir;

Equation E_r_c_inc
  # Ratio of nominal consumption to returns to land, capital and labour #
  r_c_inc =  c - (1/[AGGLND+AGGCAP+AGGLAB])
                          *(AGGLND*lndrev + AGGCAP*caprev + AGGLAB*labrev);

Equation E_r_c_gdpinc # Ratio of nominal consumption to nominal GDP #
  r_c_gdpinc = c - gdpinc;

! 18.8m Capital stocks, investment and the inverse logistic !

Equation E_cap_at_tplus1
  # Capital accum thru the fcst year (t) related to investment in the year #
  (All,j,IND) [QCAPATTPLUS1(j) + TINY]*cap_at_tplus1(j)
                = [1-DEP(j)]*QCAPATT(j)*cap_at_t(j) + QINVEST(j)*y(j);

Equation E_del_f_ac_p_y
  # Gives shock in yr-to-yr forecasting to capital at begining of year t #
  (All,j,IND) [QCAPATT(j) + TINY]*cap_at_t(j)
     = 100*{QINV_BASE(j) - DEP(j)*QCAPATT_B(j)}*del_unity + 100*del_f_ac_p_y(j);

Equation E_y # Investment/capital ratios by industry #
  (All,j,IND) y(j) = cap_at_t(j) + r_inv_cap(j) + r_inv_cap_u;

Equation E_del_k_gr # Capital growth thru forecast year  #
  (All,j,IND) del_k_gr(j) =
     [{QCAPATTPLUS1(j)/QCAPATT(j)}/100]*[cap_at_tplus1(j) - cap_at_t(j)];

Equation E_d_eeqror
  # Expected ror equals equil. expec. ror plus disequilibrium in expec. ror #
  (All,j,IND) d_eror(j) = d_eeqror(j) + d_diseq(j);

Equation E_d_f_eeqror_j
  # Change in equilibrium expected rate of return in forecast year #
  (All,j,IND) d_eeqror(j) =  (1/COEFF_SL(j))*
         [1/(K_GR(j)-K_GR_MIN(j))+1/(K_GR_MAX(j)-K_GR(j))]*del_k_gr(j)
                   + d_f_eeqror_j(j) + d_f_eeqror;

Equation E_d_diseq
  # Gives shock to disequil. in s.e. rors, moves them towards zero #
  (All,j,IND) d_diseq(j) = - ADJ_COEFF(j)*DISEQSE_B(j)*del_unity + d_f_diseq(j);

Equation E_d_f_diseqre
  # Gives shock to disequil. in r.e. rors, moves them towards zero #
  (All,j,IND) d_diseq(j) = - ADJ_COEFF(j)*DISEQRE_B(j)*del_unity+d_f_diseqre(j);

Equation E_ch_kgr1
  # Provides convenient method for viewing the values of CHKGR1 #
  (All,j,IND) ch_kgr1(j) = CHKGR1(j)*del_unity;

Equation E_ch_kgr2
  # Provides convenient method for viewing the values of CHKGR2 #
  (All,j,IND) ch_kgr2(j) = CHKGR2(j)*del_unity;

! 18.8n Expected and actual rates of return !

Equation E_d_eror # Expected rate of return, static expectations #
  (All,j,IND) d_eror(j) =  del_ror_se(j) + d_ff(j);

Equation E_d_f # Expected rate of return, rational expectations #
  (All,j,IND) d_eror(j) =  ONE_ITER1*del_ror_se(j)
  + ONE_IT1_REP*COEFF_NYEAR*( d_eror_o(j) + del_ror_se(j) - del_ror_se_o(j))
  + (1-ONE_IT1_REP)*(1-ONE_ITER1)*COEFF_NYEAR*
                        {EROR_F(j) - EROR_B(j)}*del_unity + d_f(j);

Equation E_d_eror_ave # Average expected rate of return #
  d_eror_ave= (1/{Sum(j,IND,CAPITAL(j))+TINY})*Sum(j,IND,CAPITAL(j)*d_eror(j));

Equation E_p1cap # Changes in expected rors by industry: static exp. #
  (All,j,IND) 100*del_ror_se(j) = (1/(1 + RINT_PT_SE))*
  {    [CAPITAL(j)*(1 - TAX_K_RATE)/VCAP_AT_TM(j)]*[p1cap(j) - pi(j)]
           - TAX_K_RATE*{[CAPITAL(j)/VCAP_AT_TM(j)] - RALPH*DEP(j)}*tax_k_r
    - [CAPITAL(j)*(1-TAX_K_RATE)/VCAP_AT_TM(j)+1-DEP(j)
  + RALPH*TAX_K_RATE*DEP(j)  ]* (1/[(1 + RINT_PT_SE)])*100*d_rint_pt_se    };

Equation E_d_rint_pt_se
  # Changes in real post-tax rate of interest, static expectations #
  100*d_rint_pt_se = (1/(1+INF))*{100*(1-TAX_K_RATE)*d_int
                 -INT*TAX_K_RATE*tax_k_r
          - 100*(1/(1+INF))*(1+INT*(1-TAX_K_RATE))*d_inf };

Equation E_d_int # Nominal rate of interest #
  d_int = (1+INF)*d_rint + (1+RINT)*d_inf;

Equation E_d_inf # Rate of inflation #
  100*d_inf =(1+INF)*(xi3 - xi3_l);

Equation  E_d_f_xi3_l
  # Lagged value of the CPI if initial sol for year t is sol for year t-1 #
  LEV_CPI_L*xi3_l = 100*(LEV_CPI_B - LEV_CPI_L_B)*del_unity + 100*d_f_xi3_l;

Equation E_del_ror_se
  # Allows for equalization of static expectations of changes in rors #
  (All,j,IND) del_ror_se(j) = del_r_tot + del_f_rate(j);

Equation E_del_f_rate
  # Expec. static rors related to ind. growth deviat. & init. expec. rors #
  (All,j,IND) del_f_rate(j) = {BETA_CH/100}*[cap_at_t(j) - k_r_wgts]
  - {ROR_SE_BASE(j) - AV_ROR_SE_B}*del_r + del_ff_rate(j)
  + Sum(jj,INDSAGGTO20:INDTOIA20(j) EQ $POS(jj), del_ff_rate_s(jj));

Equation E_f_r_inv_cap
  # Allows equalization of changes in investment/capital ratios within sectors #
  (All,j,IND) r_inv_cap(j) = f_r_inv_cap(j)
       + Sum(s,INDSAGGTO26:INDTOIA26(j) EQ $POS(s), r_inv_cap_s(s));

Equation E_d_ror_act_l # Actual rate of return for year t-1 #
  (All,j,IND) 100*d_ror_act_l(j) =
        (1/{[1 + INT_L*(1-TAX_K_RATE)]*PCAP_J_L(j)})*
               {  - (CAPITAL(j)/QCAPATT(j))*TAX_K_RATE*tax_k_r
                + (1-TAX_K_RATE)*(CAPITAL(j)/QCAPATT(j))*(p1cap(j) - pi_l(j))
                + (1-DEP(j))*PCAP_J(j)*(pi(j) - pi_l(j))
           + RALPH*TAX_K_RATE*DEP(j)*PCAP_J(j)*(tax_k_r + pi(j) - pi_l(j)) }
       - {(1+ROR_ACT_L(j))/[1 + INT_L*(1-TAX_K_RATE)]}*
                 { 100*(1-TAX_K_RATE)*d_int_l - INT_L*TAX_K_RATE*tax_k_r };

Equation  E_d_f_pi_l
  # Lag value of cap. asset price if init. sol for year t is sol for year t-1 #
  (All,j,IND) PCAP_J_L(j)*pi_l(j) = 100*(PCAP_J_B(j) - PCAP_J_L_B(j))*del_unity
                                      + 100*d_f_pi_l(j);

Equation E_d_int_l # Lagged nominal interest rate #
  d_int_l =  (1+RINT_L)*d_inf_l +(1+INF_L)*d_rint_l;

Equation E_d_inf_l # Lagged rate of inflation #
  100*d_inf_l =(1+INF_L)*(xi3_l - xi3_2l);

Equation E_d_f_rint_l
  # Lagged real interest rate, if init. sol for year t is sol for year t-1 #
  d_rint_l =(RINT_B - RINT_L_B)*del_unity + d_f_rint_l;

Equation  E_d_f_xi3_2l
  # Double lagged value of the CPI if init. sol for year t is sol for year t-1 #
  LEV_CPI_2L*xi3_2l = 100*(LEV_CPI_L_B - LEV_CPI_2L_B)*del_unity
                                      + 100*d_f_xi3_2l;

Equation E_lev_eror # Level of the expected ror in year t #
  (All,j,IND) lev_eror(j) = EROR_B(j)*del_unity + d_eror(j);

Equation E_lev_eror_l # Level of the expected ror in year t-1 #
  (All,j,IND) lev_eror_l(j) = EROR_B(j)*del_unity;

Equation E_lev_ror_act_l # Actual ror in year t-1 #
  (All,j,IND) lev_ror_act_l(j) = ROR_ACT_L_B(j)*del_unity + d_ror_act_l(j);

! 18.8o Miscellaneous equations to facilitate historical & forecast simulations !

! Equations to facilitate the use of price information in historical and  forecast
simulations !

Equation E_peobs
  # Allows uniform shift from obs'd or f'cst foreign-curr export prices #
  (All,i,COM) pe(i) = peobs(i) + DUMMY_NT(i)*f_pe_u_nt + f_pe_u;

Equation E_pmobs
  # Allows uniform shift from obs'd or f'cst foreign-curr import prices #
  (All,i,COM) pm(i) = pmobs(i) + f_pm;

Equation E_p3_obs # Allows adjustment to observed consumer prices #
  (All,i,COM) p3(i) = p3_obs(i) + p3_adj;

Equation E_pow_phph3_ave # Average powtaxphph applying to consumption #
  AGGCON*pow_phph3_ave
                = Sum(i,COM,Sum(s,SOURCE, PURCHVAL3(i,s)*powtaxphph(i,s)));

Equation E_pm_f  # Foreign currency price of imports #
  pm_f = xim + phi;

Equation E_pmrel # Price of imports of i relative to overall price of imports #
  (All,i,COM) pmrel(i) = pm(i) - pm_f + pm_ff;

Equation E_p_va # Price of value added by industry #
  (All,j,IND) VALUEADD(j)*p_va(j) =     (COSTS(j)-TAX0F(j))*p0ind(j) -
                [Sum(i,COM, Sum(s,SOURCE, PURCHVAL1(i,s,j)*p1csi(i,s,j)))]
      -(COSTS(j)+TAX0(j))*powtax0ph(j) - TAX0F(j)*powtax0g(j);

Equation E_p_va_s # Value added prices for industries aggregated to 20 groups #
  (All,jj,INDSAGGTO20)
  Sum(j,IND:INDTOIA20(j) EQ $POS(jj),VALUEADD(j))*(p_va_s(jj) + fvas)
          = Sum(j,IND:INDTOIA20(j) EQ $POS(jj), VALUEADD(j)*p_va(j));

Equation E_del_f1oct
  # Shifts in prices of other costs equalized within 20 sectors #
  (All,j,IND) [LEV_CPI*ROTHCOST(j)+TINY]*del_f1oct(j)
       = Sum(qq,INDSAGGTO20:INDTOIA20(j) EQ $POS(qq),
           [VALUEADD(j)/100]*f_1octss(qq))  + del_ff1oct(j);

Equation E_ff_p_va_s # Sectoral shifts in rates of return #
  (All,qq,INDSAGGTO20) Sum(j,IND:$POS(qq)=INDTOIA20(j),
        VCAP_AT_TM(j))*((1+RINT_PT_SE)/(1-TAX_K_RATE))*del_ff_rate_s(qq) =
              0.02*Sum(j,IND:$POS(qq)=INDTOIA20(j),VALUEADD(j))*f_1octss(qq)
                                + ff_p_va_s(qq);

! Equations to facilitate the use in historical simulations of consumption  data
cross-classified by MONASH and national accounts categories !

Equation E_x3_m_na
  # Consumption of cross-classified MONASH/national accounts commodities #
  (All,i,COM)(All,na,NCOM)
  x3_m_na(i,na) = x3_imputed(i) + ac(i) + a3shift(i) + a3ncom(na);

Equation E_x3_imputed # Consumption of i in the absence of taste changes #
  (All,i,COM) x3_imputed(i) =q + EPS(i)*[c - q] + Sum(j,COM, ETA(i,j)*p3(j));

Equation E_a3shift
  # Cons of MONASH comms related to cons of MONASH/national accounts comms #
  (All,i,COM) MMI(i)*x3(i) = Sum(na,NCOM, MM(i,na)*x3_m_na(i,na));

Equation E_x3ncom
  # Cons of national acc. comms related to cons of MONASH/nat. acc. comms #
  (All,na,NCOM) MMN(na)*x3ncom(na) = Sum(i,COM, MM(i,na)*x3_m_na(i,na));

! Equations to facilitate the use in historical simulations of data on  employment
in 87 ABARE and ABS industries !

Equation E_labind_d
  # Employment: base-period-wage units by 87 labour-data industries #
  (All,d,LAB_DATA_IND)
  Sum(i,IND:INDTOLABDIND(i) EQ $POS(d),(TINY+LABIND_J(i)))*labind_d(d) =
   Sum(i,IND:INDTOLABDIND(i) EQ $POS(d), (TINY+LABIND_J(i))*labind(i));

Equation E_flabprod
  # Allows equaliz. of lab productivity in MONASH inds in same lab-data ind #
  (All,i,IND) labprod(i) =
  Sum(d,LAB_DATA_IND:INDTOLABDIND(i) EQ $POS(d), labprod_d(d)) + flabprod(i);

Equation E_labprod # Labour productivity in industries #
  (All,j,IND) labprod(j) = x0ind(j) -  labind(j);

Equation E_x0ind_UP # Output of unique-production industries #
  (All,j,IND_UP)
  Sum(i,COM, MAKE(i,j))*x0ind(j) = Sum(i,COM, MAKE(i,j)*x0dom(i));

Equation E_x0ind_JP # Output of joint-production industries #
  (All,j,IND_JP)
  Sum(i,COM_JP, MAKE(i,j))*x0ind(j) = Sum(i,COM_JP, MAKE(i,j)*x0ci(i,j));

! Equations to facilitate the use in historical simulations of data on  output of 104
ABS and ABARE commodities !

Equation E_x0dom_abs # Commodity outputs in ABS & ABARE classifications #
  (All,abs,ABS_COM)
  Sum(i,COM:COMTOABSCOM(i) EQ $POS(abs), TOTFACCOM(i))*x0dom_abs(abs) =
  Sum(i,COM:COMTOABSCOM(i) EQ $POS(abs), TOTFACCOM(i)*x0dom(i));

Equation E_ffac
  # Allows equaliz. of input-saving tc in use of comms from same ABS class'n #
  (All,i,COM_UP)
  ac(i) = Sum(abs,ABS_COM:COMTOABSCOM(i) EQ $POS(abs), aq(abs) + ffac(i));

Equation E_a1ci
  # Allows equalization of tech change in all intermediate uses of commodity i #
  (All,i,COM)(All,j,IND) a1ci(i,j) = ac(i) + fa1c(i) + fa1ci(i,j);

Equation E_a2ci
  # Allows equalization of tech change in all investment uses of commodity i #
  (All,i,COM)(All,j,IND) a2ci(i,j) = ac(i) + fa2c(i) + fa2ci(i,j);

Equation E_a1marg
  # Allows equalization of tech change in all intermediate margin use of com r #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
  a1marg(i,s,j,r) = ac(r) + fa1margc(r) + fa1marg(i,s,j,r);

Equation E_a2marg
  # Allows equalization of tech change in all investment margin use of com r #
  (All,i,COM)(All,s,SOURCE)(All,j,IND)(All,r,MARGCOM)
  a2marg(i,s,j,r) = ac(r) + fa2margc(r) + fa2marg(i,s,j,r);

Equation E_a3marg
  # Allows equalization of tech change in all consumption margin use of com r #
  (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
  a3marg(i,s,r) = ac(r) + fa3margc(r) + fa3marg(i,s,r);

Equation E_a4marg
  # Allows equalization of tech change in all export margin use of com r #
  (All,i,COM)(All,r,MARGCOM) a4marg(i,r) = ac(r) + fa4margc(r) + fa4marg(i,r);

Equation E_a5marg
  # Allows equalization of tech change in all government margin use of com r #
  (All,i,COM)(All,s,SOURCE)(All,r,MARGCOM)
  a5marg(i,s,r) = ac(r) + fa5margc(r) + fa5marg(i,s,r);

Equation E_del_f_a1
  # All input-saving tech ch. to offset cost effects of ac(i)s, current prod. #
  (All,j,IND)
  - {COSTS(j)-TAX0(j)}*a1(j) = Sum(i,COM,Sum(s,SOURCE,PURCHVAL1(i,s,j)*ac(i)))
                  + 100*del_f_a1(j);

Equation E_del_f_a2
  # All input-saving tech ch. to offset cost effects of ac(i)s, investment #
  (All,j,IND)
  - VINVEST(j)*a2ind(j) = Sum(i,COM,Sum(s,SOURCE,PURCHVAL2(i,s,j)*ac(i)))
                  + 100*del_f_a2(j);

Equation E_ac1_tot # Average i-using tech change, intermediate #
  (All,i,COM) { Sum(j,IND,Sum(s,SOURCE,BAS1(i,s,j))) + TINY }*ac1_tot(i) =
     Sum[j,IND, Sum(s,SOURCE,BAS1(i,s,j))*
       { a1(j) + a1ci(i,j) + Sum(s,SOURCE,SOURCE_SHR1(i,s,j)*a1csi(i,s,j) )
           + SOURCE_SHR1(i,"imp",j)*
                [TRANSERVDUM(i)*f1_trans(j)+COMMUNICDUM(i)*f1_commun(j)] }  ];

Equation E_ac2_tot # Average i-using tech change, investment #
  (All,i,COM) { Sum(j,IND,Sum(s,SOURCE,BAS2(i,s,j))) + TINY }*ac2_tot(i) =
     Sum[j,IND, Sum(s,SOURCE,BAS2(i,s,j))*
     { a2ind(j) + a2ci(i,j) + Sum(s,SOURCE,SOURCE_SHR2(i,s,j)*a2csi(i,s,j))}];

Equation E_ac12mar_tot_NM
  # Average i-using tech change in intermed. & invest. usage of non-marg. coms #
  (All,i,NONMARGCOM)
  [Sum(j,IND,Sum(s,SOURCE,BAS1(i,s,j)+BAS2(i,s,j))) + TINY]*ac12mar_tot(i)
                       =   Sum(j,IND,Sum(s,SOURCE,BAS1(i,s,j)))*ac1_tot(i)
                         + Sum(j,IND,Sum(s,SOURCE,BAS2(i,s,j)))*ac2_tot(i);

Equation E_ac12mar_tot_M
  # Ave i-using tech change in inter., invest. & marg. usage of margin coms #
  (All,i,MARGCOM)
  [Sum(j,IND,Sum(s,SOURCE,BAS1(i,s,j)+BAS2(i,s,j)))+TOTMARGINS(i) + TINY]
        *ac12mar_tot(i) =   Sum(j,IND,Sum(s,SOURCE,BAS1(i,s,j)))*ac1_tot(i)
                          + Sum(j,IND,Sum(s,SOURCE,BAS2(i,s,j)))*ac2_tot(i)
       + Sum(ii,COM,Sum(s,SOURCE,
          { Sum(j,IND:MAR1(ii,s,j,i) ne 0, MAR1(ii,s,j,i)*a1marg(ii,s,j,i)) +
            Sum(j,IND:MAR2(ii,s,j,i) ne 0, MAR2(ii,s,j,i)*a2marg(ii,s,j,i)) }
          +  MAR3(ii,s,i)*a3marg(ii,s,i) +  MAR5(ii,s,i)*a5marg(ii,s,i)  )
          +  MAR4(ii,i)*a4marg(ii,i)  );

Equation E_int_input # Total use of intermediate inputs by industry #
  (All,j,IND) Sum(i,COM, TPURCHVAL1(i,j))*int_input(j) =
              Sum(i,COM,Sum(s,SOURCE, PURCHVAL1(i,s,j)*x1csi(i,s,j) ));

Equation E_adj # Adjustment to be applied to MONASH commodity quantities #
  (All,i,COM) adj(i) = Sum(ab,ABS_COM:COMTOABSCOM(i) EQ
$POS(ab),adj_abs(ab));

Equation E_x5cs_obs # Adjustment of government demands #
  (All,i,COM)(All,s,SOURCE) x5cs(i,s) = x5cs_obs(i,s)  - adj(i);

Equation E_x4_obs # Adjustment of exports #
  (All,i,COM) x4(i) = x4_obs(i) -adj(i);

Equation E_x3_obs # Adjustment of consumption #
  (All,i,COM) x3(i) = x3_obs(i) -adj(i);

Equation E_x0imp_obs # Adjustment of imports #
  (All,i,COM) x0imp(i) = x0imp_obs(i) +adj(i);

Equation E_x0dom_absobs # Adjustment of outputs of ABS commodities #
  (All,abs,ABS_COM) x0dom_abs(abs) = x0dom_absobs(abs) + adj_abs(abs);

Equation E_x3ncom_obs
  # Adjustment of consumption in National Account categories #
  (All,na,NCOM)
  MMN(na)*(x3ncom(na)-x3ncom_obs(na)) = - Sum(i,COM, MM(i,na)*adj(i));

! Equations to facilitate the use in historical simulations of data on  capital and
investment growth in 26 sectors !

Equation E_cap_at_t_s # Capital stocks, asset weights, for 26 sectors #
  (All,s,INDSAGGTO26)
  {Sum(j,IND:INDTOIA26(j) EQ $POS(s), VCAP_AT_TM(j))+TINY}*cap_at_t_s(s)
    = Sum(j,IND:INDTOIA26(j) EQ $POS(s), VCAP_AT_TM(j) *cap_at_t(j));

Equation E_y_s # Real investment for 26 sectors #
  (All,s,INDSAGGTO26)
  {Sum(j,IND:INDTOIA26(j) EQ $POS(s), VINVEST(j))+TINY}*y_s(s)
             = Sum(j,IND:INDTOIA26(j) EQ $POS(s), VINVEST(j) *y(j));

Equation E_twistlk # Allows equaliz. of labour/capital twists within sectors #
  (All,j,IND)
  twistlk(j) = Sum(s,INDSAGGTO26:INDTOIA26(j) EQ $POS(s), twistlk_s(s))
                   + ff_twistlk(j) + f_twistlk;

Equation E_x0ind_26s # Output of 26 sectors #
  (All,s,INDSAGGTO26)
  {Sum(j,IND:INDTOIA26(j) EQ $POS(s),TOTFACIND(j))+TINY}*x0ind_26s(s)=
       Sum(j,IND:INDTOIA26(j) EQ $POS(s),{TOTFACIND(j)+TINY}*x0ind(j));

! Equations to facilitate the use of information on agriculture !

Equation E_x0ci_obs
  # Allows uniform diff across j between x0ci(i,j) and obs of x0ci(i,j) #
  (All,i,COM_JP)(All,j,IND_JP) x0ci(i,j)=x0ci_obs(i,j)+f_x0ci(i);

Equation E_avea0ci_j # Average output-contracting tech change in industry j #
  (All,j,IND_JP)
  MAKE_COSTS(j)*avea0ci_j(j) =  Sum(i,COM_JP,MAKE(i,j)*a0ci(i,j));

! Equations to facilitate forecasting of the structure of imports!

Equation E_twist_src # Import/domestic twist #
  (All,i,COM) twist_src(i)= twist_src_bar + ftwist_src(i) + twist_eff(i);

Equation E_twist_eff
  # Import/domestic twist for i related to growth in output of i #
  (All,i,COM)
  twist_eff(i) =  C_TWIST_SRC(i)*[x0dom(i) - gdpreal] + ftwist_eff(i);

Equation E_impftwist
  # Impact of twist trends on dom. non-mar non-invent demands for dom. prod #
  (All,i,COM) { Sum(j,IND, BAS1(i,"dom",j) + BAS2(i,"dom",j))
             + BAS3(i,"dom") + BAS5(i,"dom") }*impftwist(i) =
                  - { Sum(j,IND, SOURCE_SHR1(i,"imp",j)*BAS1(i,"dom",j)
                               + SOURCE_SHR2(i,"imp",j)*BAS2(i,"dom",j))
                     +     SOURCE_SHR3(i,"imp")*BAS3(i,"dom")
                     +     SOURCE_SHR5(i,"imp")*BAS5(i,"dom")  }*ftwist_src(i)
                  + d_f_tw_hist(i);

Equation E_d_f_tw_forc # Imposes twist trends in forecasting #
  (All,i,COM) DUM_TW(i)*{ Sum(j,IND, BAS1(i,"dom",j) + BAS2(i,"dom",j))
             + BAS3(i,"dom") + BAS5(i,"dom") }*impftwist(i) =

           - { Sum(j,IND, SOURCE_SHR1(i,"imp",j)*BAS1(i,"dom",j)
                  + SOURCE_SHR2(i,"imp",j)*BAS2(i,"dom",j))
              +         SOURCE_SHR3(i,"imp")*BAS3(i,"dom")
              +         SOURCE_SHR5(i,"imp")*BAS5(i,"dom")+TINY}*ftwist_src(i)
              + d_f_tw_forc(i);

! Equations used to give structure to non-traditional export forecasts !

Equation E_hist_cont
  # Contrib.to growth in output of i of export growth above non-trad average #
  (All,i,NTRADEXP)
  hist_cont(i) = (BAS4(i)/SALES(i))*(x4(i)-ntradexpvol)+ f_hist(i);

Equation E_f_forc
  # Imposes on forecast & dev. simulations hist_cont(i)s from hist. sims #
  (All,i,NTRADEXP) DUMMY_HC(i)*hist_cont(i)
          = (BAS4(i)/SALES(i))*(x4(i)-ntradexpvol + fhist_cont) + f_forc(i);

! Equations allowing use of ABARE forecasts !

Equation E_x4_abare_RUR # Allows use of ABARE forecasts for rural commodities #
  (All,i,ABARE_RUR) x4(i) = x4_abare(i) + fx4_abare_ru;

Equation E_x4_abare_MIN # Allows use of ABARE forecasts for mining comms #
  (All,i,ABARE_MIN) x4(i) = x4_abare(i) + fx4_abare_mi;

Equation E_gx4_abare_ru
  # Allows use of overall ABARE forecast for rural commodities #
  (Sum(i,ABARE_RUR,PURCHVAL4(i)))*gx4_abare_ru =
                Sum(i,ABARE_RUR,PURCHVAL4(i)*x4(i));

Equation E_gx4_abare_mi
  # Allows use of overall ABARE forecast for mining commodities #
  (Sum(i,ABARE_MIN,PURCHVAL4(i)))*gx4_abare_mi =
                Sum(i,ABARE_MIN,PURCHVAL4(i)*x4(i));

! Other miscellaneous equations used in forecasting !

Equation E_r_elec_cpi # Allows electricity price to move with CPI #
  (All,i,CELECSET)(All,s,SOURCE) p3cs(i,s) = xi3 + r_elec_cpi(s);

Equation E_capprod # Capital productivity in industries #
  (All,j,IND) capprod(j) = x0ind(j) -  cap_at_t(j);

Equation E_f_gdpreal
  # Allows comparison of exog f'cst of GDP with sum of exog C,I,G,X-M #
  gdpreal_true = gdpreal + f_gdpreal;

Equation E_f_oil
  # Allows exogenisation of ratio of domestic prod'n of oil to oil imports #
  (All,i,COILSET) f_oil = x0dom(i) - x0imp(i);

Equation E_irtrue
  # Allows demand for buildings to be met by take-up of under-used bldgs #
  (All,i,CONSTRUCTION)
[AGGINV+BAS6(i,"dom")]*irtrue = AGGINV*ir +
                                                            100*P0LEV(i,"dom")*d_x6cs(i,"dom");

! 18.8p Equations to assist in policy simulations !

! Employment and wages in policy simulations !

Equation E_del_f_wage_c
  # Relates deviation in CPI-deflated pre-tax wage to deviation in employment #
  (RWAGE/RWAGE_OLD)*(real_wage_c - real_wage_c_o) =
        100*((RWAGE_B/RWAGE_OLD_B) - (RWAGE_L_B/RWAGE_O_L_B))*del_unity
     + ALPHA1*(EMPLOY/EMPLOY_OLD)*(emp_hours - emp_hours_o)
  -  100*ALPHA1*((RWAGE_B/RWAGE_OLD_B)^ALPHA2 –
                                 (RWAGE_L_B/RWAGE_O_L_B)^ALPHA2)*del_unity

      +del_f_wage_c;

Equation E_del_f_wage_pt
  # Relates deviation in CPI-deflated post-tax wage to deviat. in employment #
  (RWAGE_PT/RWAGE_PT_OLD)*(real_wage_pt - real_wage_pt_o) =
   100*((RWAGE_PT_B/RWAGE_PT_O_B)-
(RWAGE_PT_L_B/RW_PT_O_L_B))*del_unity
    + ALPHA1*(EMPLOY/EMPLOY_OLD)*(emp_hours - emp_hours_o)
  - 100*ALPHA1*((RWAGE_PT_B/RWAGE_PT_O_B)^ALPHA2 -
                              (RWAGE_PT_L_B/RW_PT_O_L_B)^ALPHA2)*del_unity
     + del_f_wage_pt;

Equation E_real_wage_pt
  # Economy-wide CPI-deflated wage rate, post tax #
  real_wage_pt = real_wage_c - TAX_L_RATE/(1 - TAX_L_RATE)*tax_l_r;

Equation E_d_f_empadj
  # Direct adjustment of employment back to basecase forecast #
  (EMPLOY/EMPLOY_OLD)*(emp_hours - emp_hours_o) =
  100*{(EMPLOY_B/EMPLOY_O_B) -
(EMPLOY_L_B/EMPLOY_O_L_B)}*d_empadj+d_f_empadj;

Equation E_d_ff_empadj
  # Equation for moving shift variable in E_d_f_empadj back to its forecast #
  d_f_empadj = {-FEMPADJ_B+ FEMPADJ_O}*d_emp_sh + d_ff_empadj;

! Equations to facilitate the transfer of forecasts into policy simulations !

Equation E_real_wage_c_o
  # Introduces forecast CPI-deflated pre-tax wage into policy simulation #
  real_wage_c_o = real_wage_c + f_rwage_o;

Equation E_emp_hours_o # Introduces forecast employment into policy simulation #
  emp_hours_o  = emp_hours + f_emp_o;

Equation E_real_wage_pt_o
  # Forecast post-tax CPI-deflated wage used in policy simulations #
  real_wage_pt_o = real_wage_pt + f_rwage_pt_o;

Equation E_tax_l_r_o # Forecast tax rate on wages used in policy simulations #
  tax_l_r_o = tax_l_r + ftax_l_r_o;

Equation E_labsup_o # Forecast labour supply used in policy simulations #
  labsup_o = labsup + f_labsup_o;

Equation E_x3cs_o
  # Introduces forecasts of consumption by com/source into policy sims. #
  (All,i,COM)(All,s,SOURCE) x3cs_o(i,s) = x3cs(i,s) + f_x3cs_o(i,s);

Equation E_b3sh_o
  # Introduces forecasts of budget shares by com/source into policy sims. #
  (All,i,COM)(All,s,SOURCE) b3sh_o(i,s) = b3shr(i,s)+ f_b3sh_o(i,s);

Equation E_del_ror_se_o
  # Used in 1st rat. expect. policy iteration to introduce forecast s.e. rors #
  (All,j,IND) del_ror_se_o(j) = del_ror_se(j) + d_f_ror_se_o(j);

Equation E_d_eror_o
  # Used in 1st rat. expect. policy iteration to introduce forecast r.e. rors #
  (All,j,IND) d_eror_o(j) = d_eror(j) + d_f_eror_o(j);

! Calculation of welfare effects !

Equation E_ev_over
  # Upper bound on equivalent variation as % of h'hold expenditure #
  ev_over = 100*Sum(i,COM,Sum(s,SOURCE,
     B3SHBO(i,s)*((LEVX3B(i,s)-LEVX3B_O(i,s))/LEVX3B_O(i,s))*del_unity));

Equation E_cv_under
  # Lower bound on compensating variation as % of h'hold expenditure #
  cv_under = 100*Sum(i,COM,Sum(s,SOURCE,
      B3SHB(i,s)*((LEVX3B(i,s)-LEVX3B_O(i,s))/LEVX3B(i,s))*del_unity  ));

Equation E_b3shr # Budget shares by commodity/source  #
  (All,i,COM)(All,s,SOURCE) b3shr(i,s) = x3cs(i,s)+p3cs(i,s) - c;

Equation E_d_dum_year1 # One in year one, zero in later years #
  d_dum_year1 = ADJDUMYEAR1*del_unity;

! Equations to facilitate simulation of cuts in motor vehicle tariffs !

Equation E_powtax0g # Power of genuine tax on production #
  (All,j,IND) powtax0g(j) = 100*(COSTS(j)/[COSTS(j)+TAX0G(j)])*d_fpowtax0g(j)
   - {0.095*DUMMYCAR(j)*COSTS(j)/[COSTS(j)+TAX0G(j)]}*
                  Sum(i,MOTORVEH, LEV_PTAXM(i)*powtaxm(i));

Equation  E_d_add_rev
  # Used in calculating rate of revenue replacement tax in car tariff cut sim #
  d_add_rev = Sum(i,MOTORVEH, IMPORTS(i)*fpowtaxm(i))
   - Sum(j,MOTORIND,0.095*COSTS(j)* Sum(i,MOTORVEH,
LEV_PTAXM(i)*fpowtaxm(i)))
         +  (AGGTAX3+Sum(i,COM,Sum(s,SOURCE,BAS3(i,s))))*fpowtax3gu
   -100*([RWEAL_T_O -RWEAL_T_L_O]-[RWEAL_T_B -RWEAL_T_L ])*del_wealth;

Equation E_x0dom_dom # Domestic sales of domestic goods #
  (All,i,COM)
  (SALES(i) - BAS4(i))*x0dom_dom(i) = SALES(i)*x0dom(i) - BAS4(i)*x4(i);

Equation E_x0 # Domestic sales of domestic and imported commodity i #
  (All,i,COM) TTPURCHVAL(i)*x0(i)= Sum(j,IND,Sum(s,SOURCE,
       PURCHVAL1(i,s,j)*x1csi(i,s,j)+PURCHVAL2(i,s,j)*x2csi(i,s,j)))
      +TPURCHVAL3(i)*x3(i) + Sum(s,SOURCE,PURCHVAL5(i,s)*x5cs(i,s)+
         + 100*P0LEV(i,s)*d_x6cs(i,s) );

! 18.8q The government accounts !

! Measures of the budget deficit !

Equation E_d_gov_def
  # Public sector deficit, or public sector financing transactions #
  100*d_gov_def = AGGOTH*othnom + AGGINVG*agginv_g + 100*d_othcapgov
   - NET_TAXTOTG*net_tax_tot - OTHGOVREV*oth_gov_rev + TRANS*transfers;

Equation E_d_def_gdp_r
  # Change in the ratio of the government deficit to GDP #
  100*d_def_gdp_r = (100/GDPEX)*d_gov_def - (GOV_DEF/GDPEX)*gdpexp;

Equation E_lev_def_gdp_r
  # Level of the ratio of the government deficit to GDP #
  lev_def_gdp_r = R_DEFGDP_B*del_unity + d_def_gdp_r;

! Government investment !

Equation E_agginv_g # Government investment expenditure #
  AGGINVG*agginv_g =  Sum(j,IND,G_VINVEST(j)*(pi(j)+y_g(j)));

Equation E_agginv_rg # Aggregate real public investment #
  AGGINVG*agginv_rg = Sum(j,IND,G_VINVEST(j)*y_g(j));

Equation E_xi2g # Price index for government investment expenditure #
  AGGINVG*xi2g =  Sum(j,IND,G_VINVEST(j)*pi(j));

Equation E_y_g
  # Allows public investment by industry to move with investment by industry #
  (All,j,IND) y_g(j) = y(j) + f_y_g(j) + ff_y_g;

Equation E_f_gj # Allows common movement in public investment by industry #
  (All,j,IND) y_g(j) = f_g + f_gj(j);

Equation E_d_othcapgov # Purchase or sale of 2nd hand assets by government #
  (1/GDPEX)*d_othcapgov = 0.01*(OTHCAPGOV/GDPEX)*gdpexp + d_f_othcapgov;

! Government revenue net of subsidies !

Equation E_net_tax_tot
  # Net collection of genuine indirect taxes and income taxes #
  NET_TAXTOTG*net_tax_tot
        = AGGTAXG*taxindg + INCTAX*taxrev_inc + 100*d_col_payr;

Equation E_d_net_tax_gdp # Net tax to GDP ratio #
  100*d_net_tax_gdp = (NET_TAXTOTG/GDPEX)*(net_tax_tot - gdpexp);

Equation E_taxrev_inc # Income tax revenue #
  INCTAX*taxrev_inc = TAX_LAB*(labrevinc + tax_l_r )+
              + TAX_CAP*(caprev + tax_k_r) + TAX_LND*(lndrev + tax_k_r);

Equation E_labrevinc # Labour income net of payroll taxes #
  (AGGLAB-COL_PAYRTOT)*labrevinc = AGGLAB*labrev -100*d_col_payr;

Equation E_tax_k_r
  # Rate of tax on capital income related to rate of tax on labour income #
  tax_k_r = tax_l_r + f_tax_r;

Equation E_d_col_payr # Total collection of payroll taxes #
  100*d_col_payr =
    Sum(j,IND,((POW_PAYROLL(j)-1)/POW_PAYROLL(j))*LABIND_J(j)*labrev_j(j)
    +(LABIND_J(j)/POW_PAYROLL(j))*powpayroll(j));

Equation E_labrev_j # Costs of labour by industry #
  (All,j,IND) ( LABIND_J(j)+TINY)*labrev_j(j) =
         Sum(m,OCC,LABOCCIND(m,j)*[p1laboi(m,j)+x1laboi(m,j)]);

Equation E_oth_gov_rev
  # Other government revenue, e.g. income from public enterprises #
  oth_gov_rev = gdpexp + f_oth_g_rev;

! Transfers from Government !

Equation E_transfers # Transfers from the government #
  TRANS*transfers = UNEMPBEN*unemp_ben  + AGEBEN*age_ben
            + OTHBEN*oth_ben + GRANT*grants + 100*d_net_int_g;

Equation E_unemp_ben # Unemployment benefits #
  unemp_ben = eligsh + unempben_rat
    + (1/(LAB_SUP - EMPLOY))*(LAB_SUP*labsup - EMPLOY*emp_hours);

Equation E_unempben_rat # Unemployment benefit rate #
  unempben_rat = xi3 + real_wage_c + f_unempben;

Equation E_d_f_labsup
  # Deviation in labour supply related to deviation in employment #
  (LAB_SUP/LAB_SUP_O)*(labsup - labsup_o) =
         LS_COEFF*(EMPLOY/EMPLOY_OLD)*(emp_hours - emp_hours_o) + d_f_labsup;

Equation E_age_ben # Old age benefits paid by government #
  age_ben = pop_aged + xi3 + real_wage_c + f_age_ben;

Equation E_oth_ben # Other personal benefits paid by government #
  oth_ben = pop + xi3 +real_wage_c + f_oth_ben;

Equation E_grants # Grants & transf. other than unemp, age & oth person. bens.#
  grants = gdpexp + f_grants;

Equation E_d_net_int_g # Net interest payments from government #
  d_net_int_g = [(PSDATT+PSDATTPLUS1)/2]*d_int_psd
             + 0.5*INT_PSD*(d_psd_t + d_psd_t1);

Equation E_d_int_psd  # Nominal rate of interest on public sector debt #
  d_int_psd = (1+INF)*d_rint_psd + (1+RINT_PSD)*d_inf;

Equation E_d_rint_psd
  # Link between real rates of interest on PSD and business borrowing #
d_rint_psd = d_rint + d_f_rint_psd;

Equation E_d_psd_t1 # Public sector debt, end of year #
  d_psd_t1 = d_psd_t + d_gov_def + d_f_psd_t1;

Equation E_d_f_psd_t
  # Gives shock to start-of-year public sector debt, yr-to-yr sims #
  d_psd_t = (PSDATT_1_B - PSDATT_B)*del_unity + d_f_psd_t;

Equation E_d_r_psdgdp
  # Ratio of st-of-yr public sector debt to GDP #
  d_r_psdgdp = (1/GDPEX)*d_psd_t - 0.01*R_PSDGDP*gdpexp;

Equation E_lev_r_psdgdp
  # Level of ratio of st-of-yr public sector debt to GDP #
  lev_r_psdgdp = R_PSDGDP_B*del_unity + d_r_psdgdp;

! 18.8r Equations to facilitate decomposition simulations: definitions of Gross
National Product, net foreign liabilities & consumption/savings propensities !

Equation E_gnpnom # Gross national product #
  GNP*gnpnom =GDPEX*gdpexp + [ROIFOREIGN*NET_FLTF/EXCH]*phi
                         - [100*ROIFOREIGN/EXCH]*d_netfltf;

Equation E_cr # Relates total public and private consumption to GNP #
  (1/(AGGOTH+AGGCON))*(AGGOTH*othnom + AGGCON*c) = gnpnom + apc_gnp;

Equation E_d_netfltf # Foreign currency value of net foreign liabilities #
  d_netfltf = { NET_FLTF_B*{(1+ROIFOREIGN)^TAU}
  + EXCH_B*Sum(j,IND,QCAPATT_B(j)*PCAP_J_B(j)*DEP(j))*
                  [(1+ROIFOREIGN)^TAU-1]/ROIFOREIGN
  - EXCH_B*[APSGNP_B*GNP_B+TRNFGNAUS_B]*
[(1+ROIFOREIGN)^TAU - 1]/ROIFOREIGN - NET_FLTF_B}*del_unity
  - EXCH_B*[APSGNP_B*GNP_B+TRNFGNAUS_B]*((1+ROIFOREIGN)^(TAU - 1))*
  Sum(s,COUNTSET,DUM_COUNT(s)*(COUNT(s)/TAU)*
                                                                       [(1/(1+ROIFOREIGN))^COUNT(s)]*
  {((EXCH*[APSGNP*GNP+TRNFGNAUS])
                     /(EXCH_B*[APSGNP_B*GNP_B+TRNFGNAUS_B]))
^(COUNT(s)/TAU)})
  *(0.01*phi+[1/(APSGNP*GNP+TRNFGNAUS)]*
            [0.01*APSGNP*GNP*(gnpnom +aps_gnp)+ d_trn_fgn_a])
  +0.01*Sum(s,COUNTSET,
  DUM_COUNT(s)*[(1+ROIFOREIGN)^(TAU-1-COUNT(s))]
*EXCH_B*(COUNT(s)/TAU)*Sum(j,IND, PCAP_J_B(j)*QCAPATT_B(j)*
                                                [(QCAPATT(j)/QCAPATT_B(j))^(1/TAU)-(1-DEP(j))]*
{[EXCH*PCAP_J(j)*QCAPATT(j)/(EXCH_B*PCAP_J_B(j)*QCAPATT_B(j))]
                                                                    ^(COUNT(s)/TAU)}*(phi+pi(j)+cap_at_t(j))))
  +0.01*Sum(s,COUNTSET,
  DUM_COUNT(s)*[(1+ROIFOREIGN)^(TAU-1-COUNT(s))]*EXCH_B*(1/TAU)*
  Sum(j,IND, PCAP_J_B(j)*QCAPATT_B(j)*
{[EXCH*PCAP_J(j)*QCAPATT(j)/(EXCH_B*PCAP_J_B(j)*QCAPATT_B(j))]
                 ^(COUNT(s)/TAU)}
*[(QCAPATT(j)/QCAPATT_B(j))^(1/TAU)]*cap_at_t(j)))
  + d_f_netfltf;

Equation E_aps_gnp
  # Relates the average propensities to consume and save out of GNP #
  APSGNP*aps_gnp +APCGNP*apc_gnp = 0;

Equation E_d_netflt # $A value of net foreign liabilities #
  d_netflt = (1/EXCH_T)*d_netfltf - 0.01*(NET_FLTF/EXCH_T)*excht;

Equation E_wealth_at_t # National wealth at start of year #
  WEALTH_T*wealth_at_t = AGGVCAP_T*aggvcap - 100*d_netflt;

Equation E_rwealth_at_t # Real national wealth at start of year #
  rwealth_at_t = wealth_at_t - xi2_t;

Equation E_xi2_t # Index of start-of-year prices of investment goods #
  AGGVCAP_T*xi2_t = Sum(j,IND, VCAP_AT_T(j)*pcapatt(j));

! 18.8s The balance of payments and Australia's foreign assets and liabilities !

! Current account of the balance of payments !

Equation E_d_cad  # Current account of the balance of payments #
  100*d_cad = AGGIMP*impvald - [AGGEXP+(1-DUMF)*AGGTAX4F]*expvald
     + INTFDFC*int_fdfc + INTFDDC*int_fddc + DIV_FE_A*div_fea
     - INTFCFC*int_fcfc - INTFCDC*int_fcdc - DIV_AE_F*div_aef- 100*d_trn_fgn_a;

Equation E_d_newfl
  # $A value of f'gn borrowing &  equity investment in Australia during yr #
  d_newfl = d_cad + d_newfa - d_balitem;

Equation E_d_balitem # Determines the balancing item in balance of payments  #
  d_balitem  = GDPEX*d_r_balgdp + (BAL_ITEM/100)*gdpexp;

! Interest/dividend payments and receipts in the balance of payments !

Equation E_int_fdfc
  # $A value of interest on f'gn debt repayable in f'gn curr. #
  int_fdfc = roi_fdfc + (FDFCATT_1/(FDFCATT_1 + FDFCATT))*fdfc_t1
        + (FDFCATT/(FDFCATT_1 + FDFCATT))*fdfc_t;

Equation E_int_fddc
  # $A value of interest on f'gn debt repayable in dom. curr. #
  int_fddc = roi_fddc + (FDDCATT_1/(FDDCATT_1 + FDDCATT))*fddc_t1
        + (FDDCATT/(FDDCATT_1 + FDDCATT))*fddc_t;

Equation E_int_fcfc
  # $A value of interest earned on f'gn credit repayable in f'gn curr. #
  int_fcfc = roi_fcfc + (FCFCATT_1/(FCFCATT_1 + FCFCATT))*fcfc_t1
        + (FCFCATT/(FCFCATT_1 + FCFCATT))*fcfc_t;

Equation E_int_fcdc
  # $A value of interest earned on f'gn credit repayable in dom. curr. #
  int_fcdc = roi_fcdc + (FCDCATT_1/(FCDCATT_1 + FCDCATT))*fcdc_t1
        + (FCDCATT/(FCDCATT_1 + FCDCATT))*fcdc_t;

Equation E_div_fea # Dividend payments on f'gn-owned equity #
  DIV_FE_A*div_fea = Sum(j,IND, DIV_FE_A_J(j)*div_fea_j(j));

Equation E_div_fea_j # Dividend payments on f'gn-owned equity #
  (All,j,IND) div_fea_j(j) = p1cap(j) - (TAX_K_RATE/(1-TAX_K_RATE))*tax_k_r
             + fea_t_j(j) - pcapatt(j)  + fgnshd_j(j) +fgnshd;

Equation E_div_aef # Dividend receipts on f'gn equity held by Australians #
  div_aef = rodiv_aef + aef_t;

Equation E_p1cap_ave # Economy-wide average rental rate of capital #
  p1cap_ave = (1/AGGCAP)*Sum(j,IND, CAPITAL(j)*p1cap(j));

Equation E_p1cap_ave_f # Average rental rate of capital owned by foreigners #
  p1cap_ave_f = (1/AGGCAPF)*Sum(j,IND, SHFEA_T_J(j)*CAPITAL(j)*p1cap(j));

! Start-of-year foreign debt, credit and equity values !

Equation E_d_f_fdfc_t
  # Gives shock to start-of-year f'gn debt f'gn currency in year-to-year sims #
  FDFCATT*fdfc_t = 100*(FDFCATT_1_B - FDFCATT_B)*del_unity + d_f_fdfc_t;

Equation E_d_f_fddc_t
  # Gives shock to start-of-year f'gn debt dom. currency in year-to-year sims #
  FDDCATT*fddc_t=100*(FDDCATT_1_B - FDDCATT_B)*del_unity + d_f_fddc_t;

Equation E_d_f_fea_t_j
  # $A value of foreign equity in Australia, by industry, start of year #
  (All,j,IND)
  FE_A_T_J(j)*fea_t_j(j) = 100*(FE_A_T1_J_B(j)-FE_A_T_J_B(j))*del_unity
             + d_f_fea_t_j(j);

Equation E_d_f_fcfc_t
  # Gives shock to start-of-year f'gn credit f'gn curr. in year-to-year sims #
  FCFCATT*fcfc_t = 100*(FCFCATT_1_B - FCFCATT_B)*del_unity + d_f_fcfc_t;

Equation E_d_f_fcdc_t
  # Gives shock to start-of-year f'gn credit dom. currency in yr-to-yr sims #
  FCDCATT*fcdc_t=100*(FCDCATT_1_B - FCDCATT_B)*del_unity + d_f_fcdc_t;

Equation E_d_f_aef_t
  # Gives shock to start-of-yr f'gn equity held by Australians, yr-to-yr sims #
  AE_F_T*aef_t =100*(AE_F_T_1_B - AE_F_T_B)*del_unity + d_f_aef_t;

Equation E_ff_shfeat
  # Aggregate foreign equity in Australian industries, start of year #
  FE_A_T*fea_t = Sum(j,IND,FE_A_T_J(j)*fea_t_j(j));

Equation E_fea_t
  # Total foreign liabilities, sum of debt and equity, start of year #
  FL_T*flt = FDFCATT*fdfc_t +FDDCATT*fddc_t + FE_A_T*fea_t;

Equation E_aef_t
  # F'gn assets held by Australians, sum of credit and equity, start of year #
  FA_T*fat = FCFCATT*fcfc_t +FCDCATT*fcdc_t + AE_F_T*aef_t;

Equation E_flt # $A value of net foreign liabilities, start of year #
  100*d_netflt = FL_T*flt -FA_T*fat;

Equation E_fdfc_t
  # Relates f'gn currency debt to total f'gn liabilities, start of year #
  fdfc_t = shtfdfc + flt;

Equation E_fddc_t
  # Relates domestic currency debt to total f'gn liabilities, start of year #
  fddc_t = shtfddc +flt;

Equation E_fea_t_j # Relates f'gn equity in j to capital in j, start of year #
  (All,j,IND) fea_t_j(j) = cap_at_t(j) + pcapatt(j) + shfeat_j(j) + ff_shfeat;

Equation E_fcfc_t
  # Relates f'gn currency credits to total f'gn assets, start of year #
  fcfc_t = shtfcfc +fat;

Equation E_fcdc_t
  # Relates domestic currency credits to total f'gn assets, start of year #
  fcdc_t = shtfcdc +fat;

Equation E_fat
  # Relates Australia's foreign assets to GDP, start of year #
  fat = gdpexp + f_fat;

! End-of-year foreign debt and equity values !

Equation E_fdfc_t1 # F'gn debt incurred in f'gn currency, end of yr #
  fdfc_t1 = sht1fdfc + flt1;

Equation E_fddc_t1 # F'gn debt incurred in domestic currency, end of yr #
  fddc_t1 = sht1fddc + flt1;

 Equation E_fea_t1
  # $A value of f'gn liabilities as sum of debt and equity, end of year #
  FL_T1*flt1 = FDFCATT_1*fdfc_t1 +FDDCATT_1*fddc_t1 + FE_A_T_1*fea_t1;

Equation E_fea_t1_j
  # $A value f'gn-owned equity in Australia by industry, end of year #
  (All,j,IND) fea_t1_j(j) = shfeat1_j(j) + cap_at_tplus1(j) + pcapatt1(j);

Equation E_ff_shfeat1 # Aggregate f'gn-owned equity, end of year #
  FE_A_T_1*fea_t1 = Sum(j,IND,FE_A_T1_J(j)*fea_t1_j(j));

Equation E_shfeat1_j # F'gn equity shares in Aust. industries, end of year #
  (All,j,IND)
  shfeat1_j(j) = (shfeat_j(j) + ff_shfeat) + f_shfeat1_j(j) + ff_shfeat1;

! New foreign debt and equity !

Equation E_d_new_fdfc
  # $A value of new foreign debt incurred in f'gn currency #
  FDFCATT_1*fdfc_t1 = {FDFCATT*(EXCH_T/EXCH_T_1)}*{fdfc_t + excht - excht1}
    + 100*(EXCH/EXCH_T_1)*d_new_fdfc + NEWFDFC*(EXCH/EXCH_T_1)*
                                                                                                                      (phi - excht1);

Equation E_d_new_fddc
  # $A value of new foreign debt incurred in domestic currency #
  FDDCATT_1*fddc_t1= FDDCATT*fddc_t + 100*d_new_fddc;

Equation E_d_new_fea_j
  # $A value of new foreign equity in Australian industries #
  (All,j,IND) FE_A_T1_J(j)*fea_t1_j(j) =
          [FE_A_T_J(j)*(1-DEP(j))*PCAP_AT_T1(j)/PCAP_AT_T(j)]*
    [fea_t_j(j) + pcapatt1(j) - pcapatt(j)]
       + 100*(PCAP_AT_T1(j)/PCAP_J(j))*d_new_fea_j(j)
            + NEW_FE_A_J(j)*(PCAP_AT_T1(j)/PCAP_J(j))*(pcapatt1(j) -  pi(j));

Equation E_flt1 # $A value of new foreign equity in Australia #
  d_new_fea = Sum(j,IND, d_new_fea_j(j));

Equation E_d_new_fea
  # $A value of new f'gn liabilities as sum of new borrowing & equity #
  d_newfl = d_new_fdfc + d_new_fddc + d_new_fea;

Equation E_fat1
  # Relates end-of-year foreign assets owned by Australians to GDP #
  fat1 = f_fat1  + gdpexp;

Equation E_fcfc_t1 # $A value of f'gn credit given in f'gn currency, end of yr #
  fcfc_t1 = sht1fcfc + fat1;

Equation E_fcdc_t1 # $A value of f'gn credit given in dom. currency, end of yr #
  fcdc_t1 = sht1fcdc + fat1;

Equation E_aef_t1
  # $A value of f'gn assets as sum of credit and equity, end of year #
  FA_T1*fat1 = FCFCATT_1*fcfc_t1 +FCDCATT_1*fcdc_t1 + AE_F_T_1*aef_t1;

Equation E_d_new_fcfc
  # $A value of new foreign credit given in foreign currency #
  FCFCATT_1*fcfc_t1 = {FCFCATT*(EXCH_T/EXCH_T_1)}*{fcfc_t + excht - excht1}
  + 100*(EXCH/EXCH_T_1)*d_new_fcfc + NEWFCFC*(EXCH/EXCH_T_1)
                                                                                                               *{phi - excht1};

Equation E_d_new_fcdc
  # $A value of new foreign credit given in domestic currency #
  FCDCATT_1*fcdc_t1= FCDCATT*fcdc_t + 100*d_new_fcdc;

Equation E_d_new_aef
  # $A value of new foreign equity obtained by Australians #
  AE_F_T_1*aef_t1 = {AE_F_T*(WPTPLUS1/WPT)*(EXCH_T/EXCH_T_1)}*
            {aef_t + xiworld_t1 - xiworld_t + excht - excht1}
             + 100*(WPTPLUS1/WP)*(EXCH/EXCH_T_1)*d_new_aef
             + NEW_AE_F*(WPTPLUS1/WP)*(EXCH/EXCH_T_1)*
                                 {xiworld_t1 -  xiworld + phi - excht1};

Equation E_d_newfa # $A value of new f'gn assets #
  d_newfa = d_new_fcfc + d_new_fcdc + d_new_aef;

! Start, mid and end-year asset prices, exchange rates and world price indexes !

Equation E_d_f_pcapatt
  # Gives shock to start-of-year asset prices in year-to-year simulations #
  (All,j,IND) PCAP_AT_T(j)*pcapatt(j)
     = 100*[PCAP_AT_T1_B(j) - PCAP_AT_T_B(j)]*del_unity + d_f_pcapatt(j);

Equation E_pcapatt
  # Used in historical/decomp sims for movements in start-of-year asset prices #
  (All,j,IND) PCAP_AT_T(j)*pcapatt(j)=
  (1 - 0.5/TAU)*[PCAP_J(j)*(PCAP_J(j)/PCAP_J_B(j))^(-0.5/TAU)]*pi(j)
      +100*(PCAP_J_B(j) - PCAP_AT_T_B(j))*del_unity + d_ff_pcapatt(j);

Equation E_pcapatt1
  # Used in historical/decomp sims for movements in end-of-year asset prices #
  (All,j,IND) PCAP_AT_T1(j)*pcapatt1(j) =
     (1 + 0.5/TAU)*[PCAP_J(j)*(PCAP_J(j)/PCAP_J_B(j))^(0.5/TAU)]*pi(j)
      +100*(PCAP_J_B(j) - PCAP_AT_T1_B(j))*del_unity + d_f_pcapatt1(j);

Equation E_d_ff_pcapatt1 # End-of-year asset prices in year-to-year sims #
  (All,j,IND) PCAP_AT_T1(j)*pcapatt1(j)=
          (1 + 0.5)*[PCAP_J(j)*(PCAP_J(j)/PCAP_J_B(j))^(0.5)]*pi(j)
        + 100*(PCAP_J_B(j) - PCAP_AT_T1_B(j))*del_unity + d_ff_pcapatt1(j);

Equation E_d_f_excht
  # Gives shock to start-of-year exchange rate in year-to-year simulations #
  EXCH_T*excht = 100*[EXCH_T_1_B - EXCH_T_B]*del_unity + d_f_excht;

Equation E_excht
  # Used in historical/decomp sims for movements in st-of-year exchange rate #
  EXCH_T*excht= (1 - 0.5/TAU)*[EXCH*(EXCH/EXCH_B)^(-0.5/TAU)]*phi
      +100*(EXCH_B - EXCH_T_B)*del_unity + d_ff_excht;

Equation E_excht1
  # Used in historical/decomp sims for movements in end-of-year exchange rate #
  EXCH_T_1*excht1= (1 + 0.5/TAU)*[EXCH*(EXCH/EXCH_B)^(0.5/TAU)]*phi
      +100*(EXCH_B - EXCH_T_1_B)*del_unity + d_f_excht1;

Equation E_d_ff_excht1
  # End-of-year exchange rate in year-to-year simulations #
  EXCH_T_1*excht1= (1 + 0.5)*[EXCH*(EXCH/EXCH_B)^(0.5)]*phi
      +100*(EXCH_B - EXCH_T_1_B)*del_unity + d_ff_excht1;

Equation E_d_f_xiwldt
  # Gives shock to start-of-year world price index in yr-to-yr simulations #
  WPT*xiworld_t = 100*[WPT_1_B - WPT_B]*del_unity + d_f_xiwldt;

Equation E_xiworld_t
  # Used in hist./decomp sims for movements in st-of-year world price index #
  WPT*xiworld_t= (1 - 0.5/TAU)*[WP*(WP/WP_B)^(-0.5/TAU)]*xiworld
      +100*(WP_B - WPT_B)*del_unity + d_ff_xiwldt;

Equation E_xiworld_t1
  # Used in hist./decomp sims for movements in end-of-year world price index #
  WPTPLUS1*xiworld_t1= (1 + 0.5/TAU)*[WP*(WP/WP_B)^(0.5/TAU)]*xiworld
      + 100*(WP_B - WPT_1_B)*del_unity + d_f_xiwldt1;

Equation E_d_ff_xiwldt1
  # End-of-year world price index in year-to-year simulations #
  WPTPLUS1*xiworld_t1= (1 + 0.5)*[WP*(WP/WP_B)^(0.5)]*xiworld
      +100*(WP_B - WPT_1_B)*del_unity + d_ff_xiwldt1;

! Transfer payments and receipts in the balance of payments !

Equation E_d_trn_fgn_a # Total net transfers from foreigners to Australians #
  100*d_trn_fgn_a =
        TRNFGNHH*trn_fgn_hh + TAXFGNGOV*tax_fgn_gov - TRNOTHFGN*trnoth_fgn;

Equation E_trn_fgn_hh
  # $A value of unrequited transfers to Australian residents from foreigners #
  trn_fgn_hh = ftrn_fgn_hh + pop + world_gdp + xiworld - phi;

Equation E_tax_fgn_gov
  # $A value of income taxes paid by foreigners to Australia #
  tax_fgn_gov = ftax_fgn_gov +taxrev_inc;

Equation E_trnoth_fgn # $A value of other net transfers to foreigners #
  trnoth_fgn = ftrnoth_fgn + gdpexp;

! Macro variables introduced in BOP equations and measures of net foreign
liabilities and the current account deficit !

Equation E_ximf # Index of cif prices of imports, foreign currency #
  AGGIMP*ximf = Sum(i,COM, IMPCOST(i)*pm(i));

Equation E_xiworld # World inflation #
  xiworld = ximf + f_xiworld;

Equation E_d_nfl_gdp # Change in ratio of st-of-yr net f'gn liabilities to GDP #
  d_nfl_gdp = (1/GDPEX)*d_netflt - 0.01*R_NFL_GDP*gdpexp;

Equation  E_d_nfl_vcap
  # Change in ratio of st-of-yr NFL to aggregate capital stock #
  d_nfl_vcap = [1/Sum(j,IND,VCAP_AT_T(j))]*d_netflt - 0.01*R_NFL_VCAP*aggvcap;

Equation E_aggvcap # Aggregate value of capital #
  Sum(j,IND,VCAP_AT_T(j))*aggvcap =
                     Sum(j,IND,VCAP_AT_T(j)*(cap_at_t(j)+pcapatt(j)));

Equation E_lev_nfl_gdp # Level of the ratio of net foreign liabilities to GDP #
  lev_nfl_gdp = R_NFL_GDP_B*del_unity + d_nfl_gdp;

Equation E_lev_nfl_vcap
  # Level of the ratio of net foreign liabilities to capital stock #
  lev_nfl_vcap = R_NFL_VCAP_B*del_unity + d_nfl_vcap;

Equation E_d_r_cadgdp # Change in the ratio of CAD to GDP #
  d_r_cadgdp = (1/GDPEX)*d_cad - (1/100)*R_CADGDP*gdpexp;

Equation E_lev_r_cadgdp # Level of the ratio of CAD to GDP #
  lev_r_cadgdp = R_CADGDP_B*del_unity + d_r_cadgdp;

! 18.8t Household disposable income, household saving and national saving !

Equation E_hdy # Household disposable income #
  HOUS_DIS_INC*hdy = GDPEX*gdpexp + TRANS*transfers
  -{NET_TAXTOTG*net_tax_tot + OTHGOVREV*oth_gov_rev
                                                                   -   TAXFGNGOV*tax_fgn_gov}
+ TRNFGNHH*trn_fgn_hh + INTFCFC*int_fcfc + INTFCDC*int_fcdc +
DIV_AE_F*div_aef - INTFDFC*int_fdfc - INTFDDC*int_fddc - DIV_FE_A*div_fea;

Equation E_housav # Household saving #
  HOUS_SAV*housav = HOUS_DIS_INC*hdy - AGGCON*c;

Equation E_real_housav # Real household saving #
  real_housav = housav - xi2;

Equation E_natsav
  # National saving: household saving plus public sector surplus #
  NAT_SAV*natsav = HOUS_SAV*housav +GOV_SAV*govsav;

Equation E_govsav # Government saving #
  GOV_SAV*govsav = - 100*d_gov_def
                    + AGGINVG*agginv_g + 100*d_othcapgov;

Equation E_natsav_r # Real national saving #
  natsav_r = natsav - xi2;

Equation E_govsav_r # Real government saving #
  govsav_r = govsav - xi2;

Equation E_apc # Consumption related to disposable income #
  AGGCON*c = AV_PROP_CON*HOUS_DIS_INC*(apc + hdy);

! 18.9  Displays and Writes!

Display COSTS;
OTHCOST;
SALES;
AGGCON;
INCTAX;

Write GDPEX to file WRITFILE;
GDPIN to file WRITFILE;
LOST_GOODS to file WRITFILE;
PURE_PROFITS to file WRITFILE;


